
مقدمة الكتاب

 بسم الله الرحمن الرحيم، والصلة  والسلم  على أشرف الأنبياء والمرسلين، نبينا محمد وعلى آله
وصحبه أجمعين. 

 ) لإستخدمها مع بيئةObject Pascalأما بعد فهذا الكتاب موجه لمن يريد أن يتعلم لغة باسكال الكائنية (
ُلح  للمبرمج الجديد الذي ليست لديه خبرة  فيDelphi أو بيئة ديلفي Free Pascalفري باسكال   . فهو يص

ُلح  لمن كانت لديه  البرمجة ولكن شرط أن يكون لديه خبرة  ومعرفة في علوم  الحاسوب. كذلك فهو يص
خبرة  في لغة برمجة أخرى ويريد تعلم هذه اللغة.

ًا حيث أنه يساعد ًا على فهم وتعلم البرمجة عموم  وبالنسبة للمبرمج الجديد فهذا الكتاب يساعد أيض
المبرمج على فهم كيفية حل المشاكل البرمجية والحيل التي يقوم  بها المبرمج في التطبيقات المختلفة.

لغة أوبجكت باسكال 

 ) كان في عام Object Oriented Programmingأول ظهور للغة باسكال تدعم البرمجة الكائنية الموجهة (
  الشهيرة  التابعة لسلسة تيربو باسكال5.5 في شركة أبل للكمبيوتر. بعد ذلك تلتها تيربو باسكال 1983

 التي كانت تنتجتها شركة بورلند، وقد قام  المبرمج أندرس هجلسبيرغ بإضافة البرمجة الكائنية لخط
.1989تيربو باسكال في عام  

 ) كما وتدعم البرمجةStructured Programmingلغة الباسكال الكائنية هي لغة تدعم البرمجة الهيكلية (
). Object Oriented Programmingالكائنية(

ُلملغة الباسكال هي لغة متعددية  الأغرراض تصلح  لكافة أنواع التطبيقات والإحتياجات،   فبدايةً من تع
ًا بالألعاب ، والبرامج الحسابية، والبرامج التعليمية، وتطبيقات  البرمجة لسهولتها ووضوحها، مرور

ًل بيئة تطوير ديلفي مطورة  بالدلفي نفسها، كذلك  الإنترنت، وبرامج الإتصالت ،  ولغات البرمجة، فمث
ًء بنظم التشغيل مثل الإصدارات الأولى من  فري باسكال ولزااراس تم تطويرهما بلغة الباسكال، وإنتها

و StreamOSنظام  تشغيل ماكنتوش كان يستخدم  فيها لغة باسكال الكائنية و كذلك نظامي التشغيل 
Toro OS المستخدم  فيه مترجم Free Pascal.

ُتستخدم  لإنتاج برامج تعمل في نظام   بعد النجاح التي حققته أدياة  التطوير تيربو باسكال التي كانت 
  لتستهدف نظام  التشغيل وندوزا1995الدوزا، قامت شركة بورلنت بإنتاج أدياة  التطوير ديلفي في عام  

 . وقد لقت ديلفي نجاح كبير حيث أنها95 ليعمل مع وندوزا 2 بت في إصدارة  ديلفي 32 بت ثم وندوزا 16
 كانت تنتج برامج أسرع بأضعاف المرات التي كانت تنتجها نظيرتها الفيجوال بيسك، وكان البرنامج الذي

 حين إنزال البرامج في Run-time librariesينتج عن الدلفي ليحتاج لمكتبات إضافية أو مايعرف بالـ 
أجهزة  أخرى لتحتوي على ديلفي.

Delphiديلفي 

  التي أنتجتها شركة بورلند كما سبق ذكره هي عبارة  عن أدياة  تطوير سريعة للبرامجديلفي أدياة  التطوير 
)Rapid Application Development Tool.أما اللغة المستخدمة في هذه الأدياة  فهي لغة باسكال الكائنية ، (

 حدث تطوير كبير للغة باسكال الكائنية من قِبل فريق ديلفي حتى أصبحت لغة منافسة لمثيلتها.
ّية. المشكلة الرئيسية الموجودية  في الدلفي هي أنها  وأصبحت لغة ذات إمكانات عالية ومكتبات غرن
مرتبطة بنظام  التشغيل وندوزا، ولم يحدث تطوير لمترجمها ليعمل على أنظمة أخرى بشكل جدي.

  بت وليعمل64الن يعمل فريق الدلفي لإعادية  صياغرة مترجم الدلفي حتى يتمكن من إنتاج برامج وندوزا 



.2010كذلك مع منصات أخرى غرير الوندوزا. و يتوقع إنتاج هذا المترجم الجديد في أواسط عام  
ًا بهذه المنافسة،  بعد المنافسة بين المصادير المفتوحة والمصادير المغلقة، تأثرت شركة بورلند كثير

ًا القسم المتخصص في إنتاج أديوات البرمجة الذي تم فصله في شركة تسمى    . بعدCodeGearخصوص
.2008 في عام  Embarcaderoذلك تم بيع هذه الشركة المتخصصة في أديوات البرمجة لشركة 

  نسخ مجانية في أديوات تطوير ديلفي وسي بلدر تسمى تيربو2006أنتجت شركة بورلند في عام  
ًاTurbo Explorerإكسبلورر ( ًا وخاص  ).  طوال هذه الفترة  كان الكودي المصدري للغة وأديوات التطوير مغلق

.Embarcaderoبشركة بورلند ثم 
 بلغ عددي المبرمجين الذين يستخدمون الدلفي أكثر من مليون ونصف مبرمج حسب إحصائية الشركة في

   وغريرها منSkype, Morfik, Age of wonder. وتم تصميم برامج كثيرة  بإستخدامها، نذكر منها 2008عام  
البرامج المهمة.

Free Pascalفري باسكال 

 ،1993بعد توقف شركة بورلند عن إنتاج خط تيربو باسكال الذي كان يستخدم  نظام  الدوزا إلى عام  
 قام  فريق فري باسكال بإنتاج نسخة شبيهة بتيربو باسكال ليكون بديل حر مفتوح المصدر. لكن هذه

 المرة  مع إضافة مهمة وهي إستهداف منصات جديدة  مثل: لينكس، ماكنتوش، آرم ، والي فون، وغريرها
 بت.64 بت والوندوزا 32بالإضافة إلى الوندوزا 

فريق فري باسكال كانت إحدى أهدافه هي التوافقية مع لغة باسكال الكائنية المستخدمة في الدلفي.
 ، وآخر نسخة رسمية متوفرة  الن2000النسخة الأولى من مترجم فري باسكال صدرت في يوليو عام  

2.6 هي نسخة رقم 2012في عام  

Lazarusلزااراس 

 بعد نجاح مترجم فري باسكال وتفوقه على مترجم تيربو باسكال، وإنتاج نسخة تعمل في عددي من
 المنصات التشغيلية، كانت الحلقة الناقصة هي أدياة  التطوير المتكاملة. لزااراس هي أدياة  التطوير

 المستخدمة مع فري باسكال، أو هي أدياة  التطوير التي تستخدم  فري باسكال كمترجم. وهي عبارة  عن
 ، وبهذه الطريقة نكودي قد حولنا أدياة  باسكال إلى أدياة  معتمدة  علىclass libraryمكتبة ضخمة للكائنات 

  مماثلة للدلفي بالإضافة لكونهاcomponent driven development التطوير بإستخدام  المكونات أو الكائنات
 RAD Rapid Applicationمحرر للكودي ومصمم للبرنامج. فهي بذلك تحقق كونها أدياة  تطوير سريعة 

Development.
 ،2012 منه في أواخر أغرسطس من العام  1 وتم إصدار النسخة رقم 1999بدأ مشروع لزااراس عام  

  كما تمت كتابة عددي من1لكن تمت كتابة عددي كبير من البرامج بواسطة النسخ السابقة للنسخة رقم 
الكتب حوله.

 في هذا الكتاب سوف نستخدم  لزااراس وفري باسكال في كتابة وشرح البرامج، إل أن نفس البرامج
يمكن تطبيقها بإستخدام  الدلفي مع بعض التعدليات.



ميزات لغة باسكال

 تتميز لغة باسكال الهدفية بسهولة تعلمها، وإمكاناتها العالية، وسرعة مترجماتها والبرامج التي تنتج عنها.
 لذلك فهي تعطي المبرمج فرصة إنتاج برامج ذات كفاءة  وإعتماديية عاليتين في وقت وجيز، بإستخدام 

 بيئة تطوير متكاملة وواضحة ديون الدخول في تعقيدات اللغات وأديوات التطوير الصعبة. وهذا يحقق
الإنتاجية العالية.

المؤلف: معتز عبدالعظيم الطاهر

  هجرية. وقد بدأت بدراسة1420م  الموافق 1999تخرجت في جامعة السوديان للعلوم  والتكنلوجيا عام  
  كلغة ثانية بعد تعلم البيسك الذي بدأت تعلمه في1995لغة باسكال في العام  الأول في الجامعة عام  

 المرحلة المتوسطة. ومنذ بداياتي مع لغة باسكال مازالت أستخدمها ووجدها أسهل وأفضل لغة
ًا بعد أن ديرست السي ++. وقد بدأت بإستخدام  الدلفي في عام    م  إلى الن1997للستخدام  خصوص

 ) حيث استخدمها بالإضافة إلى لزااراس وفري باسكال في كافة البرامج التي تلبي كافة2011(عام  
ًا في أكثر من   % من الوقت الذي قضيته في البرمجة متمثلة في90الأغرراض. وقد إستخدمتها تقريب

 PHPعشرات البرامج المختلفة التي قمت بكتابتها والتي تعمل الن. وماتبقى يتمثل في الجافا و 
.Software Developerوغريرها. والن أعمل كمطور برامج 

ترخيص الكتاب:

 " حيث أن المقصودي به فائدة  المسملين وزايادية  قوتهم وعلمهم."وقفيرخيص ترخيص هذا الكتاب هو ت
ويمكن لأي شخص طباعته ونسخه وتوزايعه بدون أي شروط.

بيئة التعليم المستخدمة مع هذا الكتاب:

 سوف نستخدم  في هذا الكتاب إن شاء الله بيئة لزااراس وفري باسكال كما سبق ذكره، ويمكن
.lazarus.freepascal.orgالحصول عليه من هذا الموقع: 

 أو يمكن الحصول عليه من دياخل نظام  لينكس عن طريق برامج إضافة التطبيقات، حيث نجده في قسم
  في فيدورا أو مايشابهها من توزايعاتyum install lazarusأديوات التطوير، أو عن طريق إستخدام  

 أو مايشابهها.Ubuntu في توزايعة apt-get install lazarusلينكس، أو بإستخدام  

 واللزااراس هو عبارة  عن برنامج حر ومفتوح المصدر كما ذكرنا، ويوجد في أكثر من منصة نظام 
 ) فيCompilation and linkingتشغيل. الكودي والبرامج المكتوبة به يمكن نقلها لإعادية  ترجمتها وربطها (

 ) ليمكن نقله، فهو مربتطexecutablesأي منصة يريدها المبرمج، إل أن البرنامج الناتج (الملف الثنائي 
 فقط بالمنصة التي قام  المبرمج بترجمة وربط برنامجه فيها حيث أن اللزااراس وفري باسكال ينتج عنه

 ملفات ثنائية تنفيذية تعمل مباشرة  على نواة  نظام  التشغيل ومكتباتها ولتحتاج لوسيط مثل برامج الجافا
 والدوت نت، لذلك فهو يتفوق على هذه اللغات بالسرعة في التنفيذ وعدم  إحتياج مكتبات إضافية في

الأجهزة  التي سوف يتم تثبيت برامج لزااراس بها.

http://lazarus.freepascal.org/


إستخدام  البيئة النصية 

 الفصول الأولى من هذا الكتاب تستخدم  إمكانات الإديخال والإخراج البسيطة والأساسية التي تسمى
 ، وذلك لبساطتها وتوفرها في كل أنظمة التشغيل وسهولةText mode أو البيئة النصية consoleبالطرفية 

فهمها.
 ربما يمل الطالب منها ويتمنى أن يقوم  بعمل برامج تستخدم  الشاشة الرسومية بما تحتويه من أزارار
 ونوافذ وقوائم ومربعات نصية وغريرها، إل أننا أحببنا أن لينشغل الطالب بشيء ويركز على المفاهيم

 الأساسية التي سوف يستخدمها في البرامج النصية البسيطة وسوف يستخدمها في البرامج الرسومية
ًا لأساسيات الباسكال والبرمجة. ًل وسريع المعقدة  إن شاء الله. بهذه الطريقة يكون الفهم سه

 وفي الفصول المتقدمة سوف نستخدم  إن شاء الله البيئة الرسومية ذات الواجهة المحببة للمستخدم 
العاديي.

الأمثلة

  في الصفحة التي تحتوي علىhttp://code.sdمعظم الأمثلة التي قمنا بكتباتبها  سوف نجدها في الموقع 
 هذا الكتاب. وعلى الدارس محاولة تصميم الأمثلة خطوة  بخطوة ، حتى تترسخ له البرمجة وإستخدام 
 هذه الأديوات. فإذا تعذر له ذلك أو حدثت مشكلة يمكن مقارنة البرنامج الذي قام  بتصميمه مع المثال

الموجودي في الصفحة.

http://code.sd/

الأول  الأول البرنامج البرنامج
بعد تثبيت اللزااراس وتشغيله، نقوم  بإنشاء برنامج جديد بإستخدام  الأمر التالي من القائمة الرئيسية:

Project/New Project/Program

:Source Editorلنحصل على الكودي التالي في الـ 

program Project1;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

{$IFDEF WINDOWS}{$R project1.rc}{$ENDIF}

begin
end. 

 وأحفظه في مجلد معروف بالنسبة لك،من القائمة الرئيسية.  File/Saveقم بحفظ البرنامج عن طريق 
first.lpiثم سمه 

 قم بكتابة الأوامر التالية:.begin endبين عبارتي 

  Writeln('This is Free Pascal and Lazarus');
  Writeln('Press enter key to close');
  Readln;

ليصبح  البرنامج الكامل كالتي:

program first;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

{$IFDEF WINDOWS}{$R first.rc}{$ENDIF}

begin



  Writeln('This is Free Pascal and Lazarus');
  Writeln('Press enter key to close');
  Readln;
end.

  تقوم  بتعليق البرنامج حتى يقوم  المتسخدم Readln تقوم  بكتابة نص في الشاشة. وعبارة  Writelnعبارة  
 بقراءة  الأسطر السابقة وإنتظار المستخدم  حتى يقوم  بالضغط على مفتاح إديخال. بعد ذلك يتنهي

البرنامج ويتم إغرلقه.

 أو بالضغط على الزر: F9 البرنامج عن طريق المفتاح بتنفيذبعد ذلك نقوم  

لتحصل على المخرجات التالية:

This is Free Pascal and Lazarus
Press enter key to close

 بعد تشغيل البرنامج والضغط على مفتاح الإديخال، ينتهي البرنامج ونرجع إلى الحالة الأولى، وهي حالة
 ، وإذا كنا نستخدم  وندوزا نجدfirstكتابة الكودي. إذا كنا نستخدم  نظام  لينكس نجد على القرص ملف بإسم 

  وهي الملفات الثنائية التنفيذية التي يمكن توزايعها في أي أجهزة  أخرى لتحتويfirst.exeملف بإسم 
 Nativeبالضرورة  على لزااراس أو فري باسكال. هذه البرامج تكون عبارة  عن برامج تطبيقية طبيعية (

Applications.(

ملحظة

 كالتي:Debuggerإذا لم تظهر مخرجات البرنامج كالسابق (الشاشة السودياء) قم بتعطيل الـ 
 من القائمة الرئيسية في لزااراس.Environment/Options/Debuggerقم بإختيار 
)None قم بإختيار (Debugger type and pathفي مربع 



تجارب أخرى:

في نفس البرنامج السابق قم بتغيير هذا السطر:

  Writeln('This is Free Pascal and Lazarus');

بالسطر التالي:

  Writeln('This is a number: ', 15);

قم بتشغيل البرنامج وشاهد المخرجات التالية:

This is a number: 15

قم بإستبدال السطر السابق بالأسطر التالية ونفذ البرنامج في كل مرة :

الكودي:

  Writeln('This is a number: ', 3 + 2);

المخرجات:

This is a number: 5

الكودي:

  Writeln('5 * 2 = ', 5 * 2);

المخرجات:

5 * 2 = 10

الكودي:

  Writeln('This is a real number: ', 7.2);

المخرجات:

7.2000000000000E+0000



الكودي:

  Writeln('One, Two, Three : ', 1, 2, 3);

المخرجات:

One, Two, Three : 123

الكودي:

  Writeln(10, ' * ',  3 , ' = ', 10 * 3);

المخرجات:

10 * 3 = 30

 لنرى ماهي النتائج.Writelnيمكن كتابة أي قيم بأشكال مختلفة في عبارة  



Variablesالمتغيرات 

ًل في الطريقة الرياضية عندما نقول أن    فهذا يعني5س = المتغيرات هي عبارة  حاويات للبيانات. فمث
 . كذلك يمكن إديخالها في عبارات رياضية، حيث5 هي متغير وهي في هذه اللحظة تحمل القيمة سأن 

:10 ينتج عنها 2 مضروبة في سأن قيمة 

5س = 
10 ينتج عنها 2س * 

 ، فحسب البيانات التي سوف نضعهاStrong Typed languageتتميز لغة باسكال بأنها تلتزم  بنوع المتغير
 لبد من تحديد نوع المتغير. وهذا المتغير سوف يحمل نوع واحد فقط من البيانات طوال تشغيل

ًل إذا قمنا بتعريف متغير من النوع الصحيح ، فإنه يمكننا فقط إسنادي ارقام  صحيحة له،  البرنامج، فمث
ًل. وليمكننا إسنادي عددي كسري مث

كذلك يجب التعريف عن المتغير قبل إستخدامه كما في المثال التالي:

program FirstVar;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

var
  x: Integer;
begin
  x:= 5;
  Writeln(x * 2);
  Writeln('Press enter key to close');
  Readln;
end. 

.10فعند تنفيذه نحصل على القيمة 
  والتي تفيد بأن الأسطر القاديمة عبارة  عن تعريفVarفي المثال أعله قمنا بكتابة الكلمة المحجوزاة  

للمتغيرات. وعبارة :

  x: Integer;

  وهو نوع العدديInteger وأن نوعه xتفيد شيئين: أولهما أن إسم المتغير الذي سوف نستخدمه هو 
الصحيح  الذي يقبل فقط أرقام  صحيحة لتحتوي على كسور ويمكن أن تكون موجبة أو سالبة.

أما عبارة  

  x:= 5;

x في المتغير الصحيح   5فهي تعني وضع القيمة 

ًل كما في المثال التالي:yيمكن إضافة متغير آخر للبرنامج نسميه   مث



var
  x, y: Integer;
begin
  x:= 5;
  y:= 10;
  Writeln(x * y);
  Writeln('Press enter key to close');
  Readln;
end.

نجد أن مخرجات البرنامج السابق هي:
50
Press enter key to close

characterفي المثال التالي نقوم  بإختبار نوع جديد من المتغيرات، وهو متغير يحتوي على حرف 

var
  c: Char;
begin
  c:= 'M';
  Writeln('My first letter is: ', c);
  Writeln('Press enter key to close');
  Readln;
end.

أما المثال التالي فهو لنوع الأعدادي الحقيقية التي يمكن أن تحتوي على كسور:

var
  x: Single;
begin
  x:= 1.8;
  Writeln('My Car engine capacity is ', x, ' litters');
  Writeln('Press enter key to close');
  Readln;
end.

ًل من ًل لبد من ذكر طريقة إديخال قيمة المتغيرات من المستخدم  بد  لنتمكن من كتابة برامج أكثر تفاع
ّكن المستخدم  من إديخالReadlnكتابتها في البرنامج. والطريقة البسيطة هي إستخدام  عبارة   ُتم   التي 

مدخلت حسب نوع المتغيرات كما في المثال التالي:

var
  x: Integer;
begin
  Write('Please input any number:');
  Readln(x);
  Writeln('You have entered: ', x);
  Writeln('Press enter key to close');
  Readln;
end.  

 هو عن طريق لوحة المفاتيح .xفي هذه الحالة أصبح  تخصيص القيمة للمتغير 



البرنامج التالي يقوم  بحساب جدول الضرب لرقم يقوم  بإديخاله المستخدم :

program MultTable;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

var
  x: Integer;
begin
  Write('Please input any number:');
  Readln(x);
  Writeln(x, ' * 1 = ', x * 1);
  Writeln(x, ' * 2 = ', x * 2);
  Writeln(x, ' * 3 = ', x * 3);
  Writeln(x, ' * 4 = ', x * 4);
  Writeln(x, ' * 5 = ', x * 5);
  Writeln(x, ' * 6 = ', x * 6);
  Writeln(x, ' * 7 = ', x * 7);
  Writeln(x, ' * 8 = ', x * 8);
  Writeln(x, ' * 9 = ', x * 9);
  Writeln(x, ' * 10 = ', x * 10);
  Writeln(x, ' * 11 = ', x * 11);
  Writeln(x, ' * 12 = ', x * 12);
  Writeln('Press enter key to close');
  Readln;
end.

 الملحظة المهمة في المثال السابق أن أي عبارة  تتم كتابتها بين علمتي تنصيص أحاديية تكتب كما هي
ًل: مث

  ' * 1 = '

ُيكتب بدون تنصيص فإن قيمته هي التي تظهر. أما ما
يمكن تجربة العبارتين التاليتين حتى تكون الفرق أوضح  بإستخدام  علمة التنصيص وعدم  إستخدامها:

  Writeln('5 * 3');
  Writeln(5 * 3); 

فنتيجة العبارة  الأولي يكون:

5 * 3

أما ناتج العبارة  الثانية فيكون حاصل العملية الحسابية:

15



 في المثال التالي سوف نقوم  بإجراء عملية حسابية ووضع الناتج في متغير ثالث ثم إظهار قيمة هذا
المتغير:

var
  x, y: Integer;
  Res: Single;
begin
  Write('Input a number: ');
  Readln(x);
  Write('Input another number: ');
  Readln(y);
  Res:= x / y;
  Writeln(x, ' / ', y, ' = ', Res);
  Writeln('Press enter key to close');
  Readln;
end. 

 فبما أن العملية الحسابية هي قسمة وربما ينتج عنها عددي كسري (حقيقي) لذلك يرفض مترجم
  عددي حقيقي.Resالباسكال وضع النتيجة في متغير صحيح ، لذلك لبد أن يكون هذا المتغير 

ُأحاديية.Singleونوع المتغير  ُيستخدم  لتعريف متغيرات كسرية ذات ديقية عشرية   



الأنواع الفرعية

ًل الأعدادي الصحيحة توجد منها   .Byte, SmallInt, Integer, LongInt, Wordتوجد أنواع كثيرة  للمتغيرات، فمث
 وتختلف عن بعضها في مدى الأرقام  وهي أصغر وأكبر عددي يمكن إسناديه لها. كذلك تختلف في إحتاجها

لعددي خانات الذاكرة  (البايت).

أكبر قيمةأصغر قيمةالحجم بالبايتالنوع

Byte10255

ShortInt1-128127

SmallInt2-3276832767

Word2065535

Integer4-21474836482147483647

LongInt4-21474836482147483647

Cardinal404294967295

Int648-9223372036854780000922337203685477580
7

 .Low, High, SizeOfيمكن معرفة مدى هذه الأنواع وعددي خانات الذاكرة  التي تحتاجها بإستخدام  الدوال: 
كما في المثال التالي:

program Types;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes;

begin
  Writeln('Byte: Size = ', SizeOf(Byte), ', Minimum value = ', Low(Byte), 
   ', Maximum value = ', High(Byte));

  Writeln('Integer: Size = ', SizeOf(Integer), 
    ', Minimum value = ', Low(Integer), ', Maximum value = ', High(Integer));
 
  Write('Press enter key to close');
  Readln;
end.



Conditional Branchingالتفرعات المشروطة 

 من أهم مايميز برامج الحاسوب أو أي جهازا إلكتروني آخر، هو إمكانية تنفيذ إجراء معين عند حدوث
ًل نجد أن في بعض السيارات يتم تأمين الأبواب عند بلوغرها سرعة معينة وهي غرير  شرط معين. مث

 مؤمنة. ففي هذه الحالة الشرط مركب هو وصول سرعة معينة في وجودي أبواب غرير مؤمنة، أما التفرع
 بلوغ سرعة معينة مع وجوديأو الإجراء فهو عملية تأمين الأبواب. أما في حالة عدم  توفر الشرط وهو (

) ل يتم تنفيذه.تأمين الأبواب) فإن الإجراء (حالة الأبواب غرير مؤمنة

 معظم السيارات والمصانع والغسالت اللية وماشابهها من الأجهزة  تعمل بمعالج صغير أو مايسمى
micro controller)وهي ديائرة  مدمجة ICيمكن برمجتها بواسطة الأسمبلي أو لغة سي. كذلك فإن بعض ( 

  مثل المعالج آرم embedded systemsأنواع هذه المعالجات التي تعمل في مثل هذه الأنظمة المدمجة 
ARMيمكن برمجتها بواسطة فري باسكال، وهي توجد في الموبايلت وبعض أجهزة  كمبيوترات  

الإنترنت  أو الألعاب وبعض الأجهزة  الأخرى.

If conditionعبارة  الشرط 

  الشرطية في لغة باسكال هي عبارة  سهلة وواضحة. في المثال التالي سوف نستقبل منifعبارة  
المستخدم  ديرجة الحرارة  ثم نحكم هل نقوم  بتشغيل جهازا التكيف أم  ل:

: الهواء   مكيف :برنامج الهواء   مكيف برنامج
var
  Temp: Single;
begin
  Write('Please enter Temperature of this room :');
  Readln(Temp);
  if Temp > 28 then
    Writeln('Please turn on air­condition')
  else
    Writeln('Please turn off air­condition');
  
  Write('Press enter key to close');
  Readln;
end.

  قم بكتابة28وهي تعني: إذا كانت ديرجة الحرارة  أكبر من  if then elseنجد أن العبارات الجديدة  هي: 
 ) في هذه28) وإذا لم يتحقق الشرط (رقم أصغر أو يساوي Please turn on air-conditionالسطر الأول (

).Please turn off air-condition(الحالة قم بكتابة السطر الثاني 

يمكن كتابة شروط متتالية كما في المثال التالي: 

var
  Temp: Single;
begin



  Write('Please enter Temperature of this room :');
  Readln(Temp);
  if Temp > 28 then
    Writeln('Please turn on air­condition')
  else
  if Temp < 25 then
    Writeln('Please turn off air­condition')
  else
    Writeln('Do nothing');

ًل قيم تتراوح بين   وشاهد النتائج.30 و 20قم بتنفيذ البرنامج أعله عدة  مرات مُدخِ

يمكن تعقيد البرنامج السابق ليصبح  أكثر واقعية كما في المثال أديناه:

var
  Temp: Single;
  ACIsOn: Byte;
begin
  Write('Please enter Temperature of this room : ');
  Readln(Temp);
  Write('Is air condition on? if it is (On) write 1, if it is (Off) write 0 : ');
  Readln(ACIsOn);

  if (ACIsOn = 1) and (Temp > 28) then
    Writeln('Do no thing, we still need cooling')
  else
  if (ACIsOn = 1) and (Temp < 25) then
    Writeln('Please turn off air­condition')
  else
  if (ACIsOn = 0) and (Temp < 25) then
    Writeln('Do nothing, it is still cold')
  else
  if (ACIsOn = 0) and (Temp > 28) then
    Writeln('Please turn on air­condition')
  else
    Writeln('Please enter a valid values');
  
  Write('Press enter key to close');
  Readln;
end.  

  وهي تعني إذا تحقق الشرطين الأول والثاني قم بتفيذ العبارة andنجد في المثال السابق وجودي عبارة  
التالية.

وهذا هو تفسير البرنامج السابق:

 ، أماDo no thing, we still need cooling قم بكتابة:  28إذا كان المكيف يعمل وديرجة الحرارة  أكبر من –
) فقم بالذهاب إلى الإجراء التالي:elseإذا لم ينطبق هذا الشرط (

 . أما إذا لمPlease turn off air-condition  قم بكتابة: 25إذا كان المكيف يعمل وديرجة الحرارة  أقل من –
ينطبق الشرط فقم بالإجراء التالي:

 . أما إذا لمDo nothing, it is still cold قم بكتابة:  25إذا كان المكيف مغلق وديرجة الحرارة  أقل من –



ينطبق الشرط فقم بالإجراء التالي:

.Please turn on air-condition قم بكتابة:  28إذا كان المكيف مغلق وديرجة الحرارة  أكبر من –

  في المتغير1 أو 0- أما إذا لم ينطبق هذا الشرط فهوي يعني أن المستخدم  قام  بإديخال قيمة غرير الـ 
ACIsOn :وفي هذه الحالة نقوم  بتنبيهه بإديخال قيم صحيحة ،Please enter a valid values.

 إذا إفترضنا أن جهازا الحاسوب أو أي جهازا آخر يمكن تشغيل الباسكال به موصل بالمكيف وأن هناك 
  بإجراءات حقيقية تقوم Writelnإجراء لفتح  المكيف وآخر لغلقه في هذه الحال يمكننا إستبدال إجراء 

ًا للمدة  الزمنية التي ًا تحسب  بتنفيذ الفتح  والإغرلق. في هذه الحالة يجب زايادية  الشروط لتصبح  أكثر تعقيد
عمل بها المكيف، كذلك يمكن الأخذ في الإعتبار أن الوقت هو في الصباح أم  المساء.

الأوزان  الأوزان برنامج برنامج
 في المثال التالي يقوم  المستخدم  بإديخال طوله بالأمتار ووزانه بالكيلوجرام . يقوم  البرنامج بحساب
ًء على طول المستخدم . ويقارن الوزان المثالي بالوزان الحالي للمستخدم ، ويطبع  الوزان المثالي بنا

ًا لفرق الوزان: النتائج حسب

program Weight;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  Height: Double;
  Weight: Double;
  IdealWeight: Double;
begin
  Write('What is your height in meters (e.g. 1.8 meter) : ');
  Readln(Height);
  Write('What is your weight in kilos : ');
  Readln(Weight);
  if Height >= 1.4 then
    IdealWeight:= (Height ­ 1) * 100
  else
    IdealWeight:= Height * 20;

  if (Height < 0.4) or (Height > 2.5) or (Weight < 3) or 
  (Weight > 200) then
  begin
    Writeln('Invalid values');
    Writeln('Please enter proper values');
  end
  else
  if IdealWeight = Weight then
    Writeln('Your wight is suitable')



  else
  if IdealWeight > Weight then
    Writeln('You are under weight, you need more ',
     Format('%.2f', [IdealWeight ­ Weight]), ' Kilos')
  else
    Writeln('You are over weight, you need to lose ',
     Format('%.2f', [Weight ­ IdealWeight]), ' Kilos');

  Write('Press enter key to close');  
  Readln;
end.

في المثال السابق نجد عدة  أشياء جديدة :

 فكلهما عددي حقيقي، أو مايسمى بالرقم ذو الفاصلة العائمة (Single: وهو مشابه للنوع Doubleالنوع 
floating point number والنوع .(Doubleهو رقم حقيقي ذو ديقة مضاعفة، وهو يحتل مساحة ضعف النوع  

single بايت من الذاكرة  أما الـ 4 حيث أن الأخير يحتل مساحة Double بايت من8 فهو يحتاج إلى  
الذاكرة  لتخزينه.

ًل إذاAnd، وهي بخلف Orالشي الثاني هو إستخدامنا للكلمة    ومن معناها تفيد تحقيق أحد الشروط. مث
 ) فإن البرنامج يقوم  بتنفيذ إجراء الشرط، ولو لم يتحقق سوف يتمHeight < 0.4تحقق الشرط الأول (

 إختبار الشرط الثاني، فإذ تحقق تم تنفيذ إجراء الشرط، وإذا لم يتحقق الشرط الثاني تم إختبار الشرط
الثالث وهكذا..

  وذلك لأن إجراء الشرط لبد أن يكونif بعد عبارة  end و beginالشيء الثالث هو إستخدامنا لعبارتي 
عبارة  واحدة ، وفي هذه الحالة إحتجنا لأن نقوم  بتنفيذ عبارتين وهما:

    Writeln('Invalid values');
    Writeln('Please enter proper values'); 

  لتحويل العبارتين إلى كتلة واحدة  أو عبارة  واحدة  تصلح  لأن تكون إجراءbegin endلذلك إستخدمنا 
الشرط:

  if (Height < 0.4) or (Height > 2.5) or (Weight < 3) or 
    (Weight > 200) then
    begin
      Writeln('Invalid values');
      Writeln('Please enter proper values');
    end

SysUtils. ول يمكن إستخدام  هذه الدالة إل بعد إضافة الوحدة  Formatالشيء الرابع هو إستخدام  الدالة 
 ، وسوف نقوم Format هي عبارة  عن مكتبة تحتوي على الإجراء SysUtils، والوحدة  Uses في عبارة  

 بتفصيل الوحدات أو المكتبات والإجراءأت في فصل البرمجة الهيكلية إن شاء الله. هذه الدالة تقوم 
ًل في المثال السابق تقوم  بإظهار العددي الحقيقي في شكل رقم  بإظهار المتغيرات بشكل معين، مث

ًل يحتوي على فاصلة عشرية ذات خانتين كما يظهر في التنفيذ. مث

What is your height in meters (e.g. 1.8 meter) : 1.8
What is your weight in kilos : 60.2
You are under weight, you need more 19.80 Kilos

 ملحوظة: حساب الوزان المثالي في البرنامج السابق ربما تكون غرير ديقيقة. ولمزيد من الدقة يمكن
 البحث عن الموضوع في الإنترنت. والمقصودي بهذا البرنامج كيفية وضع حلول برمجية لمثل هذه الأمثلة.



 فلبد للمبرمج أن يعرف كيف يضع حل لمسألة معينة، لأن المعرفة بالبرمجة وأديواتها لتكفي لأن يصبح 
 المبرمج قادير على تحليل وتصميم برامج يعتمد عليها، لذلك لبد من الدخول في تفاصيل الموضوع الذي

يتم تحليله قبل وضع البرنامج المناسب.

Case .. ofعبارة  الشرط 

.Case .. ofتوجد طريقة أخرى للتفرع المشروط وهو إستخدام  عبارة  
ًل نريد إستقبال طلب للزبون في مطعم، والوجبات مرقمة في قائمة رئيسية، ونريد من الزبون إختيار  مث

رقم الطلب كما في المثال التالي:

المطعم  المطعم برنامج برنامج
var
  Meal: Byte;
begin

  Writeln('Welcome to Pascal Restaurant. Please select your order');
  Writeln('1 ­ Chicken      (10 Geneh)');
  Writeln('2 ­ Fish         (7 Geneh)');
  Writeln('3 ­ Meat         (8 Geneh)');
  Writeln('4 – Salad        (2 Geneh)');
  Writeln('5 ­ Orange Juice (1 Geneh)');
  Writeln('6 ­ Milk         (1 Geneh)');
  Writeln;
  Write('Please enter your selection: ');
  Readln(Meal);

  case Meal of
    1: Writeln('You have ordered Chicken, ', 
        'this will take 15 minutes');
    2: Writeln('You have ordered Fish, this will take 12 minutes');
    3: Writeln('You have ordered meat, this will take 18 minutes');
    4: Writeln('You have ordered Salad, this will take 5 minutes');
    5: Writeln('You have ordered Orange juice, ', 
        'this will take 2 minutes');
    6: Writeln('You have ordered Milk, this will take 1 minute');
  else
    Writeln('Wrong entry');
  end;
  Write('Press enter key to close');
  Readln;
end. 

ًا ويحتوي علىIfإذا أردينا عمل نفس البرنامج بإستخدام  عبارة     فإن البرنامج سوف يكون أكثر تعقيد
تكرار.



عبارة     بإستخدام المطعم عبارة    برنامج بإستخدام المطعم ififبرنامج

var
  Meal: Byte;
begin
  Writeln('Welcome to Pascal restaurant, please select your meal');
  Writeln('1 ­ Chicken      (10 Geneh)');
  Writeln('2 ­ Fish         (7 Geneh)');
  Writeln('3 ­ Meat         (8 Geneh)');
  Writeln('4 ­ Salad        (2 Geneh)');
  Writeln('5 ­ Orange Juice (1 Geneh)');
  Writeln('6 ­ Milk         (1 Geneh)');
  Writeln;
  Write('Please enter your selection: ');
  Readln(Meal);

  if Meal = 1 then
    Writeln('You have ordered Chicken, this will take 15 minutes')
  else
  if Meal = 2 then
    Writeln('You have ordered Fish, this will take 12 minutes')
  else
  if Meal = 3 then
     Writeln('You have ordered meat, this will take 18 minutes')
  else
  if Meal = 4 then
    Writeln('You have ordered Salad, this will take 5 minutes')
  else
  if Meal = 5 then
    Writeln('You have ordered Orange juice, ',
    'this will take 2 minutes')
  else
  if Meal = 6 then
    Writeln('You have ordered Milk, this will take 1 minute')
  else
    Writeln('Wrong entry');

  Write('Press enter key to close');
  Readln;
end.

. في المثال التالي يقوم  البرنامج بتقييم ديرجة الطالب:caseيمكن كذلك إستخدام  مدى للرقام  بعبارة  



الطالب   درجة الطالب  برنامج درجة برنامج

var
  Mark: Integer;
begin
  Write('Please enter student mark: ');
  Readln(Mark);
  Writeln;

  case Mark of
    0 .. 39 : Writeln('Student grade is: F');
    40 .. 49: Writeln('Student grade is: E');
    50 .. 59: Writeln('Student grade is: D');
    60 .. 69: Writeln('Student grade is: C');
    70 .. 84: Writeln('Student grade is: B');
    85 .. 100: Writeln('Student grade is: A');
  else
    Writeln('Wrong mark');
  end;

  Write('Press enter key to close');
  Readln;
end.

  أي أن الشرط هو أن تكون الدرجة بين100 .. 85نلحظ في البرنامج السابق أننا إستخدما مدى مثل 
هاتين القيمتين.

  لتعمل إل مع المتغيرات المعدودية  مثل الأعدادي الصحيحة أو الحروف، لكنها لتعمل مع بعضcaseعبارة  
الأنواع الأخرى مثل الأعدادي الحقيقية والمقاطع.

المفاتيح   لوحة المفاتيح  برنامج لوحة برنامج
  وذلك لإستقبال حرف من لوحة المفاتيح charفي البرنامج التالي نقوم  بإستخدام  نوع المتغيرات الحرفية 

ومعرفة موقعه:

var
  Key: Char;
begin
  Write('Please enter any English letter: ');
  Readln(Key);
  Writeln;

  case Key of
    'q', 'w', 'e', 'r', 't', 'y', 'u', 'i', 'o', 'p':
      Writeln('This is in the second row in keyboard');
    'a', 's', 'd', 'f', 'g', 'h', 'j', 'k', 'l':
      Writeln('This is in the third row in keyboard');
    'z', 'x', 'c', 'v', 'b', 'n', 'm':
      Writeln('This is in the fourth row in keyboard');
  else
    Writeln('Unknown character');
  end;



  Write('Press enter key to close');
  Readln;
end. 

ًل إذا  إخترنا العبارة   نلحظ في المثال السابق أنا لم نستخدم  المدى لكن إستخدمنا تعددي الخيارات، مث
الأخيرة : 

    'z', 'x', 'c', 'v', 'b', 'n', 'm':

 قم بتنفيذ إجراء الشرط.m أو n أو b أو v أو c أو x أو z هي keyفهي تعني إذا كانت قيمة 

يمكن كذلك المزج بين صيغة المدى وتعددي الخيارات مثل:

    'a' .. 'd', 'x', 'y', 'z':

 z إو y أو x أو كان هو الحرف d إلى الحرف aوهي تعني إختبار الحرف إذا كان في المدى من الحرف 
قم بتنفيذ الشرط.



loopsالحلقات 

 الحلقات هي من المواضيع المهمة والعملية في البرمجة، فهي تعني الإستمرار في تنفيذ جزء معين من
العبارات بوجودي شرط معين. وعندما ينتهي أو ينتفي هذا الشرط تتوقف الحلقة عن الدوران.

forحلقة 

 كما في المثال التالي:forيمكن تكرار عبارة  معينة بعددي معين بإستخدام  

var
  i: Integer;
  Count: Integer;
begin
  Write('How many times? ');
  Readln(Count);
  for i:= 1 to Count do
    Writeln('Hello there');

  Write('Press enter key to close');
  Readln;
end.

 ، وقيمته تبدأi متغير صحيح  يسمى متغير الحلقة، وهو في هذا المثال المتغير forنستخدم  في حلقات 
  ثم يزيد هذا1في الدورة  الأولى بالقيمة الإبتدائية التي حدديها المبرمج، في هذه الحالة هو الرقم 

 المتغير في كل ديورة  حتى يتنهي في الدورة  الأخيرة  بالقيمة الأخيرة ، وهي التي يحدديها المستخدم 
Countبإديخالة لقيمة 

يمكن إظهار قيمة متغير الحلقة كما في التعديل التالي للمثال السابق:

var
  i: Integer;
  Count: Integer;
begin
  Write('How many times? ');
  Readln(Count);
  for i:= 1 to Count do
  begin
    Writeln('Cycle number: ', i);
    Writeln('Hello there');
  end;

  Write('Press enter key to close');
  Readln;
end. 

 نلحظ هذه المرة  أننا قمنا تنفيذ عبارتين، لذلك إحتجنا لأن نوحدهما في شكل عبارة  واحدة  بإستخدام 
begin end.



for loopجدول الضرب بإستخدام  

  سوف نجد أنforلو قارنا بين برنامج جدول الضرب السابق والتالي الذي سوف نستخدم  فيه عبارة  
الأخير ليس فيه تكرار كالأول:

program MultTableWithForLoop;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

{$IFDEF WINDOWS}{$R first.rc}{$ENDIF}

var
  x, i: Integer;
begin
  Write('Please input any number: ');
  Readln(x);
  for i:= 1 to 12 do
    Writeln(x, ' * ', i, ' = ', x * i);

  Writeln('Press enter key to close');
  Readln;
end.

ًل عن كتابة إجراء إظهار حاصل الضرب    مرة  فقد تمت كتابته مرة  واحدة  فقط وتولت12نجد أننا بد
 مرة .12 تكرار هذا الإجراء forحلقة 

  وذلكdowntoيمكن جعل الحلقة تدور بالعكس، من القيمة الكبرى إلى القيمة الصغرى وذلك بإستخدام  
بتغيير سطر واحد في المثال السابق لجدول الضرب:

  for i:= 12 downto 1 do

المضروب   المضروب  برنامج FactorialFactorialبرنامج

المضروب هو مجموع حاصل ضرب الرقم مع الرقم الذي يسبقه إلى الرقم واحد:

6 وتنتج عنه القيمة:  1 * 2 * 3 يساوي 3مضروب 

var
  Fac, Num, i: Integer;
begin
  Write('Please input any number: ');
  Readln(Num);
  Fac:= 1;
  for i:= Num downto 1 do



    Fac:= Fac * i;
  Writeln('Factorial of ', Num ,' is ', Fac);

  Writeln('Press enter key to close');
  Readln;
end. 



Repeat Untilحلقة 

  تدور إلى أن يتحقق شرط معين،Repeat loop التي تتميز بأنها تدور بعددي مين فإن for loopبخلف حلقة 
 فماديام  هذا الشرط غرير مُحقق فهي تعيد الدوران، فإذا تحقق الشرط فسوف يخرج مؤشر التنفيذ من
 هذه الحلقة ويتم تنفيذ مابعدها. لذلك نجد أن هذه الحلقة غرير محددية  بعددي معين من الدورات، فعدديها

يعتمد على أن الشرط تحقق بعد كم من الدورات.

var
  Num : Integer;
begin
  repeat
    Write('Please input a number: ');
    Readln(Num);
  until Num <= 0;
  Writeln('Finished, please press enter key to close');
  Readln;
end. 

ًا، ثم يقوم  ًل ثم سؤال المستخدم  أن يدخل عددي  في البرنامج السابق يقوم  البرنامج بالدخول للحلقة أو
 في نهاية الحلقة بفحص قيمة هذا العددي فإذا ساوي الصفر أو عددي أقل منه فإن الشرط يكون قد تحقق

وتتنتهي الحلقة، أما إذا أديخلنا فيها أرقام  أكبر من الصفر فإن الحلقة تستمر في الدوران.

بإستخدام    المطعم بإستخدام   برنامج المطعم     Repeat UntilRepeat Untilبرنامج

var
  Selection: Char;
  Price: Integer;
  Total: Integer;
begin
  Total:= 0;
  repeat
    Writeln('Welcome to Pascal Restaurant. Please select your order');
    Writeln('1 ­ Chicken      (10 Geneh)');
    Writeln('2 ­ Fish         (7 Geneh)');
    Writeln('3 ­ Meat         (8 Geneh)');
    Writeln('4 – Salad        (2 Geneh)');
    Writeln('5 ­ Orange Juice (1 Geneh)');
    Writeln('6 ­ Milk         (1 Geneh)');
    Writeln('X ­ nothing');
    Writeln;
    Write('Please enter your selection: ');
    Readln(Selection);

    case Selection of
      '1': begin
             Writeln('You have ordered Chicken, ',
               'this will take 15 minutes');
             Price:= 10;
           end;
      '2': begin
             Writeln('You have ordered Fish, ',



              ' this will take 12 minutes');
             Price:= 7;
           end;
      '3': begin
             Writeln('You have ordered meat, ',
              ' this will take 18 minutes');
             Price:= 8;
           end;
      '4': begin
             Writeln('You have ordered Salad, '
              ' this will take 5 minutes');
             Price:= 2;
           end;
      '5': begin
             Writeln('You have ordered Orange juice, ',
              'this will take 2 minutes');
             Price:= 1;
           end;
      '6': begin
             Writeln('You have ordered Milk, ',
              'this will take 1 minute');
             Price:= 1;
           end;
      else
      begin
        Writeln('Wrong entry');
        Price:= 0;
      end;
    end;

    Total:= Total + Price;

  until (Selection = 'x') or (Selection = 'X');
  Writeln('Total price            = ', Total);
  Write('Press enter key to close');
  Readln;
end. 

في البرنامج السابق هناك  عدة  أشياء جديدة  وهي:

  وذلك لأن تنفيذ الشرط لبد أن يكون عبارة  واحدة ، وكما سبق ذكره فإنcase مع begin endإضافة .1
begin end  تقوم  بتحويل عدة  عبارات إلى عبارة  واحدة . فعبارة Writeln('You have ordered..تمت إضافة  

ًا لمعرفة الحساب الكلي.price:= 2عبارة  جديدة  معها وهي   وهي لمعرفة قيمة الطلب وإضافته لحق

  وهو يتميز بإمكانيته في إستقبال أي رمز منchar من النوع الحرفي selectionتم إستخدام  المتغير .2
ًل يمكن أن يستقبل رقم (خانة واحدة  فقط) أو حرف. وقد إستخدمنا الحرف   xلوحة المفاتيح ، فمث

للدللة على نهاية الطلبات.

  والذي سوف نقوم  بتجميع قيمة الطلبات فيه ثم نعرضه فيTotalقمنا بوضع صفر في  المتغير .3
النهاية. وقد إستخدمنا هذه العبارة  للتجميع: 

Total:= Total + Price;

  فإن الحلقةx أو صغيرة  X كبيرة  x فإذا قام  المستخدم  بإديخال xوضعنا خيارين لإديخال الحرف .4



ًا بأن قيمة الحرف  .X تختلف عن قيمته بالصيغة الكبيرة  xتتوقف. علم

while doحلقة 

  تختلفwhile بإرتباط الإستمرار في الدوران مع شرط معين، إل أن repeat مع حلقة whileتتشابه حلقة 
إختلفين:

ًل ثم الدخول إلى الحلقة.1 يتم فحص الشرط أو
  فهيwhile يتم تنفيذها مرة  واحدة  على الأقل، وبعد الدورة  الأولى يتم فحص الشرط، أما repeatحلقة .2

تمنع البرنامج من ديخول الحلقة إذا كان الشرط غرير صحيح  من البداية.
  فإنهاrepeat. أما حلقة begin endإذا كانت الحلقة تحتوي على أكثر من عبارة  فلبد من إستخدام  .3

.until وتنتهي بـ repeatلتحتاج، لأن الحلقة حدوديها تبدأ من كلمة 

مثال:

var
  Num: Integer;
begin
  Write('Input a number: ');
  Readln(Num);
  while Num > 0 do
  begin
    Write('From inside loop: Input a number : ');
    Readln(Num);
  end;
  Write('Press enter key to close');
  Readln;
end.  

while doيمكننا إعادية  كتابة برنامج المضروب بإستخدام  

حلقة     بإستخدام المضروب حلقة    برنامج بإستخدام المضروب while dowhile doبرنامج
var
  Fac, Num, i: Integer;
begin
  Write('Please input any number: ');
  Readln(Num);
  Fac:= 1;
  i:= Num;
  while i > 1 do
  begin
    Fac:= Fac * i;
    i:= i ­ 1;
  end;
  Writeln('Factorial of ', Num ,' is ', Fac);

  Writeln('Press enter key to close');
  Readln;
end. 
 



  ليكون متغيرi فقد قمنا بإستخدام  المتغير for loop ليس لديها متغير حلقة مثل whileبما أن حلقة 
  حتى يصلى إلى واحد وهو شرط عدم  الدخول مرة  أخرى فيiالحلقة. لذلك قمنا بإنقاص قيمة المتغير 

الحلقة.



stringsالمقاطع 

 ) أو مجموعة من الحروف والأرقام  والرموزا، حيث أنه يمكن للمتغيرtextالمقاطع هي عبارة  عن نص (
ًل أو رقم جوازا أو رقم لوحة سيارة  تحتوي على حروف وأرقام . المقطعي أن يستقبل إسم مستخدم  مث

المثال التالي يوضح  كيفية إستقبال وطباعة قيمة متغير مقطعي:

var
  Name: string;
begin
  Write('Please enter your name : ');
  Readln(Name);
  Writeln('Hello ', Name);

  Writeln('Press enter key to close');
  Readln;
end.  

والمثال التالي نستخدم  عدة  متغيرات لتخزين معلومات شخص معين:

var
  Name: string;
  Address: string;
  ID: string;
  DOB: string;
begin
  Write('Please enter your name : ');
  Readln(Name);
  Write('Please enter your address : ');
  Readln(Address);
  Write('Please enter your ID number : ');
  Readln(ID);
  Write('Please enter your date of birth : ');
  Readln(DOB);
  Writeln;
  Writeln('Card:');
  Writeln('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');
  Writeln('| Name    : ', Name);
  Writeln('| Address : ', Address);
  Writeln('| ID      : ', ID);
  Writeln('| DOB     : ', DOB);
  Writeln('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');

  Writeln('Press enter key to close');
  Readln;
end.



ًل إذا افترضنا أن لدينا ثلث متغيرات مقطعية، أحدها يحمل  يمكن تجميع المقاطع لتكوين مقطع أكبر، مث
ًا: ًا واحد إسم الشخص والثاني يحمل إسم والده والثالث إسم الجد، فيمكن تجميعها لتصبح  مقطع

var
  YourName: string;
  Father: string;
  GrandFather: string;
  FullName: string;
begin
  Write('Please enter your first name : ');
  Readln(YourName);
  Write('Please enter your father name : ');
  Readln(Father);
  Write('Please enter your grand father name : ');
  Readln(GrandFather);
  FullName:= YourName + ' ' + Father + ' ' + GrandFather;
  Writeln('Your full name is: ', FullName);

  Writeln('Press enter key to close');
  Readln;
end.

 بين كل إسم والخر حتى لتلتصق الأسماء وتكون ' ' نلحظ في المثال السابق أننا قمنا بإضافة مسافة
ًا عبارة  عن رمز يمثل وحدة  من وحدات المقطع. غرير مقروءة . وهذه المسافة أيض

 يمكن إجراء عدة  عمليات على المقاطع، مثل البحث فيها عن مقطع معين أو نسخها لمتغير مقطعي
 آخر، أو تحويلها إلى حروف إنجليزية كبيرة  أو صغيرة . وكمثال يمكننا تجربة الدوال الخاصة بالحروف

الكبيرة  والصغيرة  في اللغة الإنجليزية:

ًل. والإضافة هي: يمكننا إضافة سطر للمثال السابق قبل طباعة الإسم كام

  FullName:= UpperCase(FullName);

أو التحويل للحروف الصغيرة :

  FullName:= LowerCase(FullName);

  والتي تقوم Pos في إسم المستخدم ، وذلك بإستخدام  الدالة aفي المثال التالي نريد البحث عن الحرف 
ًل الحرف رقم   ، وهكذا، وإذا لم يكن موجودي فإن هذه الدالة2 أو 1بإرجاع رقم الحرف في المقطع، مث

 وهو يعني عدم  وجودي هذا الحرف في المقطع المعني:0ترجع 

var
  YourName: string;
begin
  Write('Please enter your name : ');
  Readln(YourName);
  If Pos('a', YourName) > 0 then
    Writeln('Your name contains a')
  else
    Writeln('Your name does not contain a letter');



  Writeln('Press enter key to close');
  Readln;
end.   

  الصغيرة a كبيرة  فإنها لتساوي الـ Aنلحظ بعد تشغيل البرنامج السابق أن الإسم إذا إحتوى على 
 ويعتبر البرنامج أنها غرير موجودية ، ولجعل البرنامج يقوم  بتجاهل حالة الحرف يمكننا إضافة الدالة

LowerCase:أثناء البحث 

  If Pos('a', LowerCase(YourName)) > 0 the

كذلك يمكن معرفة موضع الحرف في الإسم وذلك بتعديل البرنامج إلى:

var
  YourName: string;
begin
  Write('Please enter your name : ');
  Readln(YourName);
  If Pos('a', LowerCase(YourName)) > 0 then
  begin
    Writeln('Your name contains a');
    Writeln('a position in your name is: ', 
      Pos('a', LowerCase(YourName)));
  end
  else
    Writeln('Your name does not contain a letter');

  Write('Press enter key to close');
  Readln;
end.

  تقوم  بإرجاع رقم أول ظهور لهذاPos فإن الدالة aنلحظ أن الإسم إذا أحتوى على أكثر من حرف 
الحرف فقط.

 كما في المثال التالي:lengthيمكن معرفة طول المقطع (عددي حروفه) بإستخدام  الدالة 

  Writeln('Your name length is ', Length(YourName), ' letters');

كذلك يمكن معرفة الحرف الأول بإستخدام : 

  Writeln('Your first letter is ', YourName[1]);

والحرف الثاني:

  Writeln('Your second letter is ', YourName[2]);

والحرف الأخير:

  Writeln('Your last letter is ', YourName[Length(YourName)]);



ويمكن كتابة كافة الحروف مفرقة ، كل حرف في سطر كالتي:

  for i:= 1 to Length(YourName) do
    Writeln(YourName[i]);

Copyالدالة 

ًل إذا كان المقطع يحتوى على الكلمة '  ' فيمكنناhello worldيمكن نسخ جزء معين من المقطع مث
' إذا عرفنا موضعها بالضبط في المقطع الأصلي وذلك كما في المثال التالي:worldإستخلص كلمة '

var
  Line: string;
  Part: string;
begin
  Line:= 'Hello world';

  Part:= Copy(Line, 7, 5);

  Writeln(Part);

  Writeln('Press enter key to close');
  Readln;
end.

 بالصيغة التاليةCopyنلحظ أننا إستخدمنا الدالة 

  Part:= Copy(Line, 7, 5);

ولشرحها من الشمال إلى اليمين:

Part=: وهي تعني أن نضع المخرجات النهائية للدالة   Copy في المتغير المقطعي part
Lineوهو المقطع المصدر الذي نريد النسخ منه      

ًء منه، وهو في هذا المثال يمثل الحرف 7 w          وهو رقم الحرف الذي نريد النسخ إبتدا
world          وهو طول المقطع الذي نريد نسخه، وهي في هذه الحالة يمثل طول المقطع الجزئي 5

  ويقوم  البرنامج بإظهار إختصارFebruaryفي المثال التالي نطلب من المستخدم  إديخال إسم شهر مثل 
 وهي تمثل الحروف الثلث الأولى من إسم الشهر:Febالشهر في هذه الصيغة : 

var
  Month: string;
  ShortName: string;
begin
  Write('Please input full month name e.g. January : ');
  Readln(Month);



  ShortName:= Copy(Month, 1, 3);

  Writeln(Month, ' is abbreviated as : ', ShortName);

  Writeln('Press enter key to close');
  Readln;
end.

Insertالإجراء 

  التي تقوم  بإلصاق+ بإضافة مقطع دياخل مقطع آخر، بخلف إستخدام  علمة الجمع insertتقوم  الدالة 
 يمكنها إديخال مقطع أو حرف دياخل مقطع آخر.insertمقطعين في النهاية، فإن الدالة 

Hello Pascal World لتصبح  hello world دياخل عبارة  Pascalفي المثال التالي نقوم  بإديخال كلمة 

var
  Line: string;
begin
  Line:= 'Hello world';

  Insert('Pascal ', Line, 7);

  Writeln(Line);

  Writeln('Press enter key to close');
  Readln;
end.

ومدخلت الإجراء هي كالتالي:

'Pascal وهو المقطع الذي نريد إديخاله على العبارة '
Lineوهو المقطع الأصلي الذي نريد تغييره وإديخال مقطع آخر في وسطه     

line          المكان بالأحرف الذي نريد الإضافة فيه. وهذا المكان أو الفهرس هو موضع في المقطع 7

Delete الإجراء

ًل في المثال التالي نقوم  بحذف حرفيdeleteيستخدم  الإجراء    لحذف حرف أو مقطع من مقطع آخر، مث
 . وللقيام  بهذه العملية يجب معرفة موقع الحرف الذيheo World لتصبح  Hello World في عبارة  Lالـ 

 على التوالي.2 و 3نريد الحذف من عنده وطول الجزء المحذوف. وهي في هذه الحالة 

var
  Line: string;
begin
  Line:= 'Hello world';
  Delete(Line, 3, 2);
  Writeln(Line);
  Writeln('Press enter key to close');
  Readln;
end. 



Trimالدالة 

ًل إذا كان لدينا هذا المقطع Trimتقوم  الدالة   بحذف المسافات فقط في بداية ونهاية مقطع معين. مث
 'hello'  فإنه يصبح ' helloبعد إستخدام  هذه الدالة. لكن ليمكن إظهار المسافة في أمثلتنا الحالية إل ' 

إذا قمنا بكتابة حرف أو رمز قبل وبعد الكلمة المطبوعة:

program MultTableWithForLoop;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

{$IFDEF WINDOWS}{$R first.rc}{$ENDIF}

var
  Line: string;
begin
  Line:= ' Hello ';

  Writeln('<', Line, '>');

  Line:= Trim(Line);

  Writeln('<', Line, '>');

  Writeln('Press enter key to close');
  Readln;
end.  

 وهي التي توجد فيها هذه الدوال.SysUtilsنلحظ أننا قمنا بإضافة الوحدة  أو المكتبة 

  وهي تقوم  بحذف المسافات من جهة واحدة  يمكن تجربتهاTrimRight و TrimLeftتوجد ديوال أخرى هي 
في المثال السابق.

StringReplaceالدالة 

  بتبديل مقاطع أو حروف معينة من مقطع مصدر، ثم وضع النتيجة في مقطعStringReplaceتقوم  الدالة 
جديد، كما في المثال التالي:

program SٍtrReplace;

{$mode objfpc}{$H+}



uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  Line: string;
  Line2: string;
begin
  Line:= 'This is a test for string replacement';
  Line2:= StringReplace(Line, ' ', '­', [rfReplaceAll]);
  Writeln(Line);
  Writeln(Line2);
  Write('Press enter key to close');
  Readln;
end.

مدخلت الدالة هي كالتي:

1.Lineوهو المقطع المصدر الذي نريد إصدار نسخة معدلة منه :
: وهو المقطع الذي نريد البحث عنه وإبداله، وفي هذه الحالة يمثل حرف واحد وهو المسافة' '.2
: المقطع البديل، وهو المقطع الذي سوف يتم تبديله'-'.3
4.]rfReplaceAll.وهي طريقة الإبدال، في هذه الحالة سوف يتم إبدال كلي لكل المسافات :[

 مع الحصول على نفس النتيجة كالتي:Line2 والإستغناء عن المتغير Lineيمكن إستخدام  مقطع واحد 

var
  Line: string;
begin
  Line:= 'This is a test for string replacement';
  Writeln(Line);
  Line:= StringReplace(Line, ' ', '­', [rfReplaceAll]);
  Writeln(Line);
  Write('Press enter key to close');
  Readln;
end.



arraysالمصفوفات 

ًل إذا قلنا أن لدينا مصفوفة  المصفوفة هي عبارة  عن صف أو مجموعة من متغيرات ذات نوع واحد، مث
 عناصر فإن تعريفها يكون كالتالي10من الأعدادي الصحيحة تحتوي على 

  Numbers: array [1 .. 10] of Integer;

ويمكننا وضع قيمة في المتغير الأول في المصفوفة كالتالي:

  Numbers[1]:= 30;

ويمكننا وضع قيمة في المتغير الثاني في المصفوفة كالتالي:

  Numbers[2]:= 315;

 طلب ثم عرضها:10في المثال التالي نقوم  بتخزين ديرجات 

var
  Marks: array [1 .. 10] of Integer;
  i: Integer;
begin
  for i:= 1 to 10 do
  begin
    Write('Input student number ', i, ' mark: ');
    Readln(Marks[i]);
  end;

  for i:= 1 to 10 do
  begin
    Write('Student number ', i, ' mark is : ', Marks[i]);
    if Marks[i] >= 40 then
      Writeln(' Pass')
    else
      Writeln(' Fail');
  end;

  Writeln('Press enter key to close');
  Readln;
end.

  لإديخال وطباعة الدرجات. كذلك قمنا بمقارنة نتيجة كل طالب دياخل الحلقةforنجد أننا إستخدمنا حلقة 
لنعرف الناجح  وغرير الناجح .

يمكن تعديل البرنامج السابق لنعرف أكبر وأصغر ديرجة في الدرجات كالتالي:

var
  Marks: array [1 .. 10] of Integer;
  i: Integer;
  Max, Min: Integer;
begin



  for i:= 1 to 10 do
  begin
    Write('Input student number ', i, ' mark: ');
    Readln(Marks[i]);
  end;

  Max:= Marks[1];
  Min:= Marks[1];

  for i:= 1 to 10 do
  begin
    // Get if current Mark is maximum mark or not
    if Marks[i] > Max then
      Max:= Marks[i];

    // Check if current value is minimum mark or not
    if Marks[i] < Min then
      Min:= Marks[i];

    Write('Student number ', i, ' mark is : ', Marks[i]);
    if Marks[i] >= 40 then
      Writeln(' Pass')
    else
      Writeln(' Fail');
  end;

  Writeln('Max mark is: ', Max);
  Writeln('Min mark is: ', Min);
  Writeln('Press enter key to close');
  Readln;
end.

  كذلك إعتبرنا أنهMaxنلحظ أننا إفترضنا أن الرقم الأول هو أكبر ديرجة لذلك قمنا بوضعه في المتغير 
 إلى أن يثبت العكس في كل الحالتين.Minأصغر ديرجة فوضعناه في المتغير 

  Max:= Marks[1];
  Min:= Marks[1];

  فإذا وجدناmin و maxوفي دياخل الحلقة عند طباعة الأرقام  العشرة ، قمنا بمقارنة كل رقم مع القيم 
.min بقيمة الدرجة الحالية ، ونفعل نفس الشيء مع القيمة max قمنا بإستبدال قيمة maxرقم أكبر من 

نلحظ في المثال السابق أننا إستخدمنا تعليق مثل:

    // Get if current Mark is maximum mark or not

  وهي تعني أن باقي السطر عبارة  عن تعليق لتتم ترجمته، إنما يستفيد منه//وقد بدأنا السطر بالعلمة 
المبرمج كشرح حتى يصبح  البرنامج مقروء له ولمن يريد الإطلع على البرنامج.

 هذه الطريقة تصلح  للتعليق القصير، أما إذا كان التعليق أكثر من سطر يمكس إستخدام  الأقواس
(* *) أو الأقواس والنجمة {}المعكوفة 

ًل: مث



  for i:= 1 to 10 do
  begin
    { Get if current Mark is maximum mark or not
      check if Mark is greater than Max then put
      it in Max }
    if Marks[i] > Max then
      Max:= Marks[i];

    (* Check if current value is minimum mark or not
       if Min is less than Mark then put Mark value in Min
    *)
    if Marks[i] < Min then
      Min:= Marks[i];

    Write('Student number ', i, ' mark is : ', Marks[i]);
    if Marks[i] >= 40 then
      Writeln(' Pass')
    else
      Writeln(' Fail');
  end;

ًا كالتالي: كذلك يمكن الإستفادية  من هذه الخاصية بتعطيل جزء من الكودي مؤقت

  Writeln('Max mark is: ', Max);
 // Writeln('Min mark is: ', Min);
  Writeln('Press enter key to close');
  Readln;

في هذا المثال قمنا بتعطيل إجراء طباعة أصغر ديرجة.



Recordsالسجلت 

 كما لحظنا أن المصفوفات تحتوي على مجموعة متغيرات من نوع واحد، فإن السجلت تجمع بين
ًل إذا افترضنا أننا نريدFieldsمجموعة من أنواع مختلفة تسمى حقول   ، ولكنها تمثل كيان واحد. مث

تسجيل معلومات سيارة ، فنجد أن هذه المعلومات هي:

نوع السيارة : متغير مقطعي.1
سعة المحرك : حددي حقيقي (كسري).2
سنة التصنيع: متغير صحيح .3

فليمكن التعبير عن هذه الأنواع المختلفة كوحدة  واحدة  إل بإستخدام  السجل كما في المثال التالي:

program Cars;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

type
  TCar = record
    ModelName: string;
    Engine: Single;
    ModelYear: Integer;
  end;

var
  Car: TCar;
begin
  Write('Input car Model Name: ');
  Readln(Car.ModelName);
  Write('Input car Engine size: ');
  Readln(Car.Engine);
  Write('Input car Model year: ');
  Readln(Car.ModelYear);

  Writeln;
  Writeln('Car information: ');
  Writeln('Model Name  : ', Car.ModelName);
  Writeln('Engine size : ', Car.Engine);
  Writeln('Model Year  : ', Car.ModelYear);

  Write('Press enter key to close..');
  Readln;
end.



 :typeفي المثال السابق نجد أننا قمنا بتعريف نوع جديد بإستخدام  الكلمة المفتاحية 

type
  TCar = record
    ModelName: string;
    Engine: Single;
    ModelYear: Integer;
  end;

 حتى نفرق بينه وبين المتغيرات.Type هو من كلمة T والحرف TCarوسمينا هذا النوع الجديد 
 يحتوي على ثلث أنواع كما يظهر في المثال.Recordوهذا النوع الجديد الذي يمثل سجل 

 وعندما نريد استخدام  هذا النوع الجديد لبد من تعريف متغير يمثل هذا النوع، حيث ليمكننا استخدام 
ًة  كما ليمكننا إستخدام  النوع TCarالنوع  ًل Integer  مباشر  .Num أو I مباشرة  إل بعد تعريف متغير مث

 في بند المتغيرات:TCar من النوع Carلذلك قمنا بتعريف المتغير 

var
  Car: TCar;

وعندما نريد إديخال قيم للمتغيرات أو طباعتها نستخدم  هذه الطريقة للوصول لمتغير ما في سجل:

  Car.ModelName

 في ديرس الملفات ذات الوصول العشوائي إن شاء الله سوف نستفيد فائدة  مباشرة  من السجلت التي
تمثل ركن أساسي في قواعد البيانات.



filesالملفات 

ًا، فإن بعض إجزاء نظام  التشغيل  الملفات هي من أهم العناصر في نظام  التشغيل والبرامج عموم
 نفسها هي عبارة  عن ملفات. والمعلومات والبيانات هي عبارة  عن ملفات، مثل الصور والكتب والبرامج

 والنصوص البسيطة هي عبارة  عن ملفات. ويجب على نظام  التشغيل توفير إمكانية لإنشاء الملفات،
ولقراءتها وكتابتها وتحريرها وحذفها.

ًء على عدة  أوجه نظر. فيمكن تقسيم الملفات إلى ملفات تنفيذية  تنقسم الملفات إلى عدة  أنواع بنا
 وملفات بيانات، حيث أن الملفات التنفيذية هي التي تمثل البرامج وأجزائه التنفيذية مثل الملف الثنائي
 الذي يحتوي على كودي يفهمه نظام  التشغيل ومثال لها الملفات التنفيذية التي تصدر عن مترجم الفري

  والتي يمكن نقلها في عددي من الأجهزة  التي تحتوي على نفس نظام  التشغيل ثمgccباسكال أو الـ 
 . وكمثال لها برنامج اللة الحاسبة،consoleتشغيلها بالنقر عليها أو بكتابة إسمها في شاشة الطرفية 

محرر النصوص، برامج الألعاب، إلخ.
 أما النوع الثاني فهي ملفات البيانات التي لتحتوى على كودي ولتمتلك إمكانية التشغيل، إنما تحتوي على

 ، first.lprبيانات بسيطة أو معقدة  يمكن قراءتها مثل النصوص والملفات المصدرية للغات البرمجة مثل 
 أو الصور وملفات الصوت التي هي عبارة  عن بيانات يمكن عرضها بإستخدام  برامج معينة مثل برامج

  وملفاتmp3. وكمثال لهذه الملفات ملفات الصوت .multimediaتحرير الصور وبرامج تعددي الوسائط 
pdfالكتب .

 إلى قسمين:البياناتيمكننا كذلك تقسيم الملفات من حيث نوع 

 cat بسيطة يمكن إنشائها وقراءتها عن طريق أديوات بسيطة في نظام  التشغيل مثل ملفات نصية.1
  التي تطبع محتويات ملف نصي في وندوزا، وtypeالتي تعرض محتويات ملف نصي في نظام  لينكس، و 

vi.التي تقوم  بتحرير وإنشاء الملفات النصية في نظام  لينكس 

  وهي يمكن أن تحتوى على رموزا غرير مقروءة  وهذه الملفات لتصلح  لأن يقوم ملفات بيانات ثنائية.2
 المستخدم  بتحريرها أو قراءتها مباشرة ، بل يجب عليه إستخدام  برامج محددية  لهذه العمليات. مثل

ُتفهم، لذلك يجبcatملفات الصور إذا حاول المستخدم  قراءتها بإستخدام   ًل فإن يحصل على رموزا ل   مث
 . كذلك يمكن أن يكون مثال لهذهGimpفتحها بإستخدام  برامج عرض أو تحرير مثل المتصفح  أو برنامج 

ُتخزن فيهاparadox, dBaseالأنواع ملفات قواعد البيانات البسيطة مثل    فهي عبارة  عن ملفات ثنائية 
سجلت تحتوى على معلومات منظمة بطريقة معينة.

 والكتابة، حيث يوجد نوعين من الملفات:طريقة الوصولالنوع الخر من التقسيم للملفات هو 

 : ومثال لها الملفات النصية، وتتميز بأنsequential access filesملفات ذات وصول تسلسلي .1
 طول السجل أو السطر فيها غرير ثابت، لذلك لمكن معرفة موقع سطر معين في الملف. ويجب فتح 

 الملف للقراءة  فقط أو الكتابة فقط، ول يمكن الجمع بين الوضعين (الكتابة والقراءة ) وليمكن تعديلها
 بسهولة إل بقراءتها كاملة في مخزن مؤقت ثم تعديل أسطر معينة فيها ثم مسح  الملف وكتابته من

 جديد. كذلك فإن القراءة  والكتابة تتم بتسلسل وهو من بداية الملف إلى نهايته، حيث ليمكن الرجوع
سطر إلى الوراء.

  وهي ملفات ذات طول سجل ثابت، حين:random access filesملفات ذات وصول عشوائي .2
 أن السجل يمثل أصغر وحدة  يتعامل معها البرنامج في القراءة ، الكتابة والتعديل. ويمكن الجمع بين

ًل يمكن قراءة  السجل الخامس، ثم نسخ محتوياته في السجل  الكتابة والقراءة  في نفس اللحظة، مث
ًل إلى  الأخير. ويمكن الوصول مباشرة  إلى أي سجل ديون التقيد بمكان القراءة  أو الكتابة الحالي. فمث



 في هذا الملف.50 حروف أو بايت، فإن السجل الخامس يبدأ في الموقع رقم 5كان طول السجل هو 

text filesالملفات النصية 

 الملفات النصية كما سبق ذكره هي ملفات بسيطة يمكن قراءتها بسهولة. ومن ناحية الوصول هي
  حيث يجب القراءة  فقط في إتجاه واحد وهو من البداية للنهاية ويجبsequential filesملفات تسلسلية 

الكتابة فيها بنفس الطريقة.
ًل إنشاء ملفات نصية أو البحث عنها ثم وضعها  لعمل برنامج بسيط يقوم  بقراءة  محتويات ملف، علينا أو

في الدليل أو المجلد الذي يوجد فيه البرنامج ثم تنفيذ البرنامج التالي:

نصي    ملف قراءة نصي   برنامج ملف قراءة برنامج
program ReadFile;

{$mode objfpc}{$H+}
uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, sysUtils
  { you can add units after this };

var
  FileName: string;
  F: TextFile;
  Line: string;
begin
  Write('Input a text file name: ');
  Readln(FileName);
  if FileExists(FileName) then
  begin
    // Link file variable (F) with physical file (FileName)
    AssignFile(F, FileName);

    Reset(F); // Put file mode for read, file should exist

    // while file has more lines that does not read yet do the loop
    while not Eof(F) do 
    begin
      Readln(F, Line); // Read a line from text file
      Writeln(Line);   // Display this line in user screen
    end;
    CloseFile(F); // Release F and FileName connection
  end
  else // else if FileExists..
    Writeln('File does not exist');
  Write('Press enter key to close..');
  Readln;
end.



ًل  بعد تنفيذ البرنامج السابق نقوم  بإديخال إسم ملف موجودي في نفس مكان البرنامج أو بكتابة الإسم كام
مثل: 

/etc/resolv.conf
أو

/proc/meminfo
/proc/cpuinfo

 بإستخدام  أسماء الملفين الأخيرين، فإن هذا البرنامج قادير على قراءة  الذاكرة  الكلية للحاسوب والذاكرة 
المستخدمة والذاكرة  المتاحة، كذلك يمكنه قراءة  نوع المعالج المستخدم  في جهازا الحاسوب وسرعته.

نلحظ أننا استخدمنا ديوال وإجراءات وأنواع جديدة  في هذه البرنامج وهي:

1.
  F: TextFile;

  وهو النوع الذي يتعامل مع الملفات النصية. وفي البرنامجTextFile بأن نوعه Fوهي تعريف المتغير 
سوف نستخدم  هذا المتغيير للتعبير عن الملف الفعلي.

2.
  if FileExists(FileName) then

  وهي تختبر وجودي هذا الإسم كملف، فإذا كان موجودي تحققSysUtilsوهي ديالة موجودية  في المكتبة 
ُتنفّذ.elseالشرط وإن لم يكن موجودي فإن عبارة    هي التي سوف 

3.
  AssignFile(F, FileName);

 ، والذي عنF بربط إسم الملف الفعلي بالمتغير AssignFileبعد التأكد من أن الملف موجودي يقوم  إجراء 
طريقه يمكننا التعامل مع الملف من دياخل باسكال.

4.
  Reset(F); // Put file mode for read, file should exist

 وهي العبارة  التي تقوم  بفتح  الملف النصي للقراءة  فقط، وهي تخبر نظام  التشغيل أن هذا الملف
 محجوزا للقراءة ، فإذا حاول برنامج آخر فتح  الملف للكتابة أو حذفه فإن نظام  التشغيل يمنعه برسالة

access deniedمفاديها أن الملف مفتوح بواسطة تطبيق آخر أو أن السماحية غرير متوفرة  

5.
  Readln(F, Line); // Read a line from text file

 Fوهو إجراء القراءة  من الملف، حيث أن هذا الإجراء يقوم  بقراءة  سطر واحد فقط من الملف المفتوح 
.Lineووضع هذا السطر في المتغير 

6.
  while not Eof(F) do 

  فإنه يقوم  بقراءة  سطر واحد فقط، وبالنسبة للملف النصي فإننا ليمكنناReadlnكما ذكرنا في الإجراء 
  والتي تعنيeofمعرفة كم سطر يحتوي هذا الملف قبل الإنتهاء من قراءته، لذلك إستخدمنا هذه الدالة 

  وذلك للستمرار في القراءة  حتى الوصول إلىwhile. إستخدمناها مع الحلقة End Of Fileنهاية المل



.end of fileنهاية الملف وتحقيق شرط 

7.
   CloseFile(F); // Release F and FileName connection

 بعد الفراغ من قراءة  الملف، يجب إغرلقه وذلك لتحريره من جهة نظام  التشغيل حتى يمكن لبرامج
ًل.resetأخرى التعامل معه بحرية، ولبد من تنفيذ هذا الإجراء فقط بعد فتح  الملف بنجاح بواسطة    مث

ًل بواسطة  .CloseFile فليجب إغرلقه بواسطة resetفإذا فشل فتح  الملف أص

في المثال التالي سوف نقوم  بإنشاء ملف نصي جديد والكتابة فيه:

نصي     ملف وكتابة إنشاء نصي    برنامج ملف وكتابة إنشاء برنامج

var
  FileName: string;
  F: TextFile;
  Line: string;
  ReadyToCreate: Boolean;
  Ans: Char;
  i: Integer;
begin
  Write('Input a new file name: ');
  Readln(FileName);

  // Check if file exists, warn user if it is already exist
  if FileExists(FileName) then
  begin
   
    Write('File already exist, did you want to overwrite it? (y/n)');
    Readln(Ans);
    if upcase(Ans) = 'Y' then
      ReadyToCreate:= True
    else
      ReadyToCreate:= False;
  end
  else // File does not exist
    ReadyToCreate:= True;

  if ReadyToCreate then
  begin
    // Link file variable (F) with physical file (FileName)
    AssignFile(F, FileName);

    Rewrite(F); // Create new file for writing
   
    Writeln('Please input file contents line by line, '
     , 'when you finish write % then press enter');
    i:= 1;
    repeat
      Write('Line # ', i, ':');



      Inc(i);
      Readln(Line);
      if Line <> '%' then
        Writeln(F, Line); // Write line into text file
    until Line = '%';

    CloseFile(F); // Release F and FileName connection, flush buffer
  end
  else // file already exist and user does not want to overwrite it
    Writeln('Doing nothing');
  Write('Press enter key to close..');
  Readln;
end.

في البرنامج السابق استخدمنا عدة  أشياء وهي:

1.
  ReadyToCreate: Boolean;

  . وهذه القيم يمكن استخدامهاTrue/False يمكن لمتغيراته أن تحمل إحدى قيمتين فقط: booleanالنوع 
 . حيث أنrepeat وحلقة while كما في المثال، كذلك يمكن إستخدامها مع حلقة if conditionمع عبارة  

الشرط في النهاية يتحول إلى إحدى هتين القيمتين، فكما في أمثلة سابقة إستخدمنا هذه العبارة :

  if Marks[i] > Max then

 . فعندماFalse أو تكون خاطئة فيتحول الشرط إلى Trueفهي إما أن تكون النتيجة فيتحول الشرط إلى 
 يتم تنفيذ الشرط:Trueتتحول إلى 

  if True then

 إن وجد.else لتم تنفيذ إجراء الشرط وإنما يتم تنفيذ إجراء Falseوإذا كانت قيمتها النهائية 

2.
  if upcase(Ans) = 'Y' then

 هذه العبارة  يتم تنفيذها في حالة وجودي الملف، فيقوم  البرنامج بتنبيه المستخدم  بوجودي الملف، وسؤاله
  صغيرة y). فإذا قام  بإديخال الحرف overwriteإذا كان يرغرب في حذف محتوياته والكتابة عليه من جديد (

  تقوم  بتحويل الحرف إلىupCase  كبيرة  فإن الشرط يتم تنفيذه في الحالتين، حيث أن الدالة Yأو 
ًا Yحرف كبير لمقارنته مع الحرف الكبير ' ًا كبير  upCase فإن الدالة Y'. أما إذا اديخل المستخدم  حرف
.Yلتقوم  بأي تغيير وترجع الحرف كما هو 

3.
  Rewrite(F); // Create new file for writing

  يقوم  بإنشاء ملف جديد أو حذف محتويات الملف إذا كان موجودي. كذلك فهو يفتح Rewriteالإجراء 
الملف للكتابة فقط في حالة الملف النصي.

4.
  Writeln(F, Line); // Write line into text file



 . وهيCR/LF في الملف ثم إضافة علمة نهاية السطر وهي Line يقوم  بكتابة المقطع Writeln(Fالإجراء 
عبارة  عن رموزا الواحد منها يمثل بايت وقيمتها هي كالتي :

CR: Carriage Return = 13
LF: Line Feed = 10

 وهذه الرموزا من الأحرف الغير مرئية، حيث لتتم كتابتها في الشاشة، إنما يظهر فقط مفعولها، وهي
الإنتقال إلى سطر جديد.

5.
  Inc(i);

 وهو يعاديل هذه العبارة :i بإضافة واحد إلى قيمة المتغير الصحيح ، في هذه الحالة Incيقوم  الإجراء 

  i:= i + 1;

6.
   CloseFile(F); // Release F and FileName connection, flush buffer

ًا فإن الإجراء    يقوم  بإغرلق الملف وإيقاف عملية الكتابة أو القراءة  من الملف،CloseFileكما ذكرنا سابق
 وتحريره من الحجز بواسطة نظام  التشغيل. إل  أن له وظيفة إضافية في حالة الكتابة. فكما نعلم أن

ًا مقارنة بالكتابة في الذاكرة ، مثل إعطاء قيم  الكتابة على القرص الصلب هي عملية بطيئة نسبي
 للمتغيرات، أو الكتابة في مصفوفة. لذلك من غرير المنطقي كتابة كل سطر على حده في القرص

 الصلب أو أي وسيط تخزين آخر، لذلك يقوم  البرنامج بتخزين عددي معين من السطور في الذاكرة  بعملية
ًاWriteln أو Write ، فكلما إستخدمنا عبارة  Bufferingتسمى الـ    لكتابة سطر فإن البرنامج يقوم  تلقائي

ًا بالكتابةBufferبكتابته في الذاكرة  إلى أن يمتليء هذا الوعاء (  ) في الذاكرة  فيقوم  البرنامج تلقائي
 Flushing)، وهذه العلمية تسمى Bufferالفعلية على القرص الصلب ثم حذف المحتويات من الذاكرة  (

ًل لكتابة سطر واحد ربما  وبهذه الطريقة نضمن السرعة في كتابة الملف، بإعتبار أن التكلفة الزمنية مث
ًا تكلفة كتابة    أسطر في القرص الصلب ديفعة واحدة . وتكمن خطورة  هذه الطريقة في10تساوي تقريب

 إنقطاع الطاقة عن الحاسوب قبل الكتابة الفعلية، فيجد المستخدم  أن الأسطر الأخيرة  التي قام  بكتابتها
 ، كذلكFlushقد ضاعت. ويمكن إجبار البرنامج بالقيام  بالكتابة الفعلية على القرص بإستخدام  الإجراء 

ًا تحدث Flushingفإن عملية الـ  .CloseFileعند إغرلق الملف بإستخدام  إيض



الإضافة إلى ملف نصي

 سوف نقوم  في هذا المثال بفتح  ملف نصي يحتوي على بيانات ثم إضافة أسطر عليه بدون حذف
AppendFileالأسطر القديمة، ويتم ذلك بإستخدام  الإجراء 

: نصي     ملف إلى الإضافة :برنامج نصي     ملف إلى الإضافة برنامج

var
  FileName: string;
  F: TextFile;
  Line: string;
  i: Integer;
begin
  Write('Input an existed file name: ');
  Readln(FileName);
  if FileExists(FileName) then
  begin
    // Link file variable (F) with physical file (FileName)
    AssignFile(F, FileName);

    Append(F); // Open file for appending

    Writeln('Please input file contents line by line', 
      'when you finish write % then press enter');
    i:= 1;
    repeat
      Write('Line # ', i, ' append :');
      Inc(i);
      Readln(Line);
      if Line <> '%' then
        Writeln(F, Line); // Write line into text file
    until Line = '%';
    CloseFile(F); // Release F and FileName connection, flush buffer
  end
  else
    Writeln('File does not exist');
  Write('Press enter key to close..');
  Readln;
end.

 بعد تنفيذ البرنامج يمكننا كتابة إسم ملف نصي موجودي وإضافة بضعة أسطر عليه، بعد ذلك يمكننا أن
  في لينكس، أو عن طريق تصفح  الدليل الذي يوجد فيه البرنامجcatنستعرض الملف عن طريق الأمر 

والضعط عليه بالماوس لفتحه.



Random access filesملفات الوصول العشوائي  

 كما سبق ذكره فإن النوع الثاني من الملفات من حيث الوصول هي ملفات الوصول العشوائي أو كما
ًا بملفات الوصول المباشر. وهي تتميز بطول معروف للسجل ويمكن الكتابة والقراءة  من  تسمى أحيان

 الملف في آن واحد، كذلك يمكن الوصول مباشرة  إلى أي سجل بغض النظر عن موقع القراءة  أو
الكتابة الحالي. 

 توجد طريقتين للتعامل مع الملفات ذات الوصول العشوائي في لغة باسكال: الطريقة الأولى هي
 ، أما الطريقة الثانية فهي إستخدام  الملفات الغير محددية  النوعtyped filesإستخدام  الملفات ذات النوع 

untyped files.

typed fileالملفات ذات النوع 

 في هذه الطريقة يكون الملف المرادي قراءته أو كتابته مرتبط بنوع معين، والنوع المعين يمثله سجل،
ًل يمكن أن يكون ملف من النوع الصحيح    ، في هذه الحال نقول أن السجل هو عبارة  عن عدديByteفمث

 بايت.1 وفي هذه الحالة يكون طول السجلByteصحيح  

المثال التالي يوضح  طريقة كتابة ملف لأعدادي صحيحة:

الطلب     درجات تسجيل الطلب    برنامج درجات تسجيل برنامج

var
  F: file of Byte;
  Mark: Byte;
begin
  AssignFile(F, 'marks.dat');
  Rewrite(F); // Create file
  Writeln('Please input students marks, write 0 to exit');

  repeat
    Write('Input a mark: ');
    Readln(Mark);
    if Mark <> 0 then  // Don't write 0 value
      Write(F, Mark);
  until Mark = 0;
  CloseFile(F);

  Write('Press enter key to close..');
  Readln;
end. 

نلحظ في البرنامج أننا قمنا بتعريف نوع الملف بهذه الطريقة:

    F: file of Byte;

  والذي يحتل في الذاكرة Byte أو أن سجلته عبارة  هي قيم للنوع Byteوهي تعني أن الملف هو من نوع 
.255 إلى 0 بايت، ويمكنه تخزين القيم من 1وفي القرص 

كذلك قمنا بإنشاء الملف وتجهيزه للكتابة بإستخدام  الأمر:



  Rewrite(F); // Create file

 للكتابة في الملف:Writeوقد قمنا بإستخدام  الإجراء 
      Write(F, Mark);

  كما سبقCR/LF لتصلح  لهذا النوع من الملفات لأنها تقوم  بإضافة علمة نهاية السطر Writelnحيث أن 
  فهي تكتب السجل كما هو بدون أي زاياديات، لذلك نجد أننا في المثال السابق يمكنناWriteذكره، أما 

 1 وبما أن طول السجل هو 3 ديرجات فإن عددي السجلت يكون 3معرفة حجم الملف، فإذا قمنا بإديخال 
 بايت. 3بايت فإن حجم الملف يكون 

 خانات، ورؤية حجم الملف الناتج.4 الذي يحتل Integerيمكن تغيير النوع إلى 

 في المثال التالي نقوم  بقراءة  محتويات الملف السابق، فلننسى إرجاع نوع الملف في البرنامج السابق
، لأن القراءة  والكتابة لبد أن تكون لملف من نفس النوع.Byteإلى النوع 

الدرجات    ملف قراءة الدرجات   برنامج ملف قراءة برنامج

program ReadMarks;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  F: file of Byte;
  Mark: Byte;
begin
  AssignFile(F, 'marks.dat');
  if FileExists('marks.dat') then
  begin
    Reset(F); // Open file
    Writeln('Please input students marks, write 0 to exit');

    while not Eof(F) do
    begin
      Read(F, Mark);
      Writeln('Mark: ', Mark);
    end;
    CloseFile(F);
  end
  else
    Writeln('File (marks.dat) not found');

  Write('Press enter key to close..');
  Readln;
end.

 نلحظ أن البرنامج السابق قام  بإظهار الدرجات الموجودية  في الملف. وأظن أن البرنامج واضح  وليحتاج
لشرح.



في البرنامج التالي نقوم  بفتح  الملف السابق ونضيف إليه ديرجات جديدة  بدون حذف الدرجات السابقة:

الطلب    درجات إضافة الطلب   برنامج درجات إضافة برنامج

program AppendMarks;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  F: file of Byte;
  Mark: Byte;
begin
  AssignFile(F, 'marks.dat');
  if FileExists('marks.dat') then
  begin
    FileMode:= 2; // Open file for read/write
    Reset(F); // open file
    Seek(F, FileSize(F)); // Go to after last record
    Writeln('Please input students marks, write 0 to exit');

    repeat
      Write('Input a mark: ');
      Readln(Mark);
      if Mark <> 0 then  // Don't write 0 value
        Write(F, Mark);
    until Mark = 0;
    CloseFile(F);
  end
  else
    Writeln('File marks.dat not found');

  Write('Press enter key to close..');
  Readln;
end.

 بعد تنفيذ البرنامج السابق نقوم  بتشغيل برنامج قراءة  ملف الدرجات وذلك لرؤية أن قيم السجلت
ًا في نفس الملف. السابقة والجديدة  موجودية  مع

ًل من Resetنلحظ في هذا البرنامج أننا إستخدمنا الإجراء   . وفي هذا النوع منRewrite للكتابة بد
  تقوم  بإنشاءRewriteالملفات يمكن إستخدام  كليهما للكتابة، والفرق الرئيسي بينهما يكمن في أن 

  فهي تفترض وجودي الملف، فإذا لمResetالملف إذا لم يكن موجودي وحذف محتوياته إذا كان موجودي، أما 
 :FileMode تقوم  بفتح  الملف حسب قيمة Resetيكون موجودي حدث خطأ. لكن عبارة  

  يفتح  للكتابة فقط، وإذا كانت1 فإن الملف يفتح  للقراءة  فقط، وإذا كانت قيمته  0فإذا كانت قيمتها 
ًا:وهي القيمة الإفتراضية -2قيمته  - فإنه يفتح  للقراءة  والكتابة مع



    FileMode:= 2; // Open file for read/write
    Reset(F); // open file

  تقوم  بوضع مؤشر القراءة  والكتابة في أول سجل، لذلك إذا باشرنا الكتابة فإنResetكذلك فإن الدالة 
 البرنامج السابق يقوم  بالكتابة فوق محتويات السجلت السابقة، لذلك وجب تحريك هذا المؤشر للسجل

  الذي يقوم  بتحريك الموشر، وبهذه الطريقة تمت تسمية هذا النوعSeekالأخير، وذلك بإستخدام  الإجراء 
  يسمح  لنا بالتنقلSeekمن الملفات بالملفات العشوائية أو ملفات الوصول المباشر، حيث أن الإجراء 

ًل إذا قمنا  لأي سجل مباشرة  إذا علمنا رقمه، شريطة أن يكون هذا الرقم موجودي، فسوف يحدث خطأ مث
  في حين أن الملف يحتوي على عددي سجلت أقل من100بمحاولة توجيه المؤشر إلى السجل رقم 

100.
 والتي تقوم  بإرجاع عددي السجلت:FileSizeاستخدمنا الدالة 

        Seek(F, FileSize(F)); // Go to after last record

 نلحظ أن المثال السابق يصلح  فقط في حالة وجودي ملف الدرجات، أما إذا لم يكن موجودي وجب
إستخدام  البرنامج الأول لكتابة الدرجات.

 يمكننا المزج بين الطريقتين، بحيث أن البرنامج يقوم  بفحص وجودي الملف، فإذا كان موجودي يقوم  بفتحه
:Rewrite وإذا لم يكن موجودي يقوم  بإنشائه بإستخدام  Resetللضافة عن طريق 

الطلب     درجات وإضافة إنشاء الطلب    برنامج درجات وإضافة إنشاء برنامج

program ReadWriteMarks;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  F: file of Byte;
  Mark: Byte;
begin
  AssignFile(F, 'marks.dat');
  if FileExists('marks.dat') then
  begin
    FileMode:= 2; // Open file for read/write
    Reset(F); // open file
    Writeln('File already exist, opened for append');
    // Display file records
    while not Eof(F) do
    begin
      Read(F, Mark);
      Writeln('Mark: ', Mark);
    end
  end
  else // File not found, create it
  begin



    Rewrite(F);
    Writeln('File does not exist, created');
  end;

  Writeln('Please input students marks, write 0 to exit');
  Writeln('File pointer position at record # ', FilePos(f));
  repeat
    Write('Input a mark: ');
    Readln(Mark);
    if Mark <> 0 then  // Don't write 0 value
      Write(F, Mark);
  until Mark = 0;
  CloseFile(F);

  Write('Press enter key to close..');
  Readln;
end.

 بعد تشغيل البرنامج نجد أن القيم السابقة تم عرضها في البداية قبل الشروع في إضافة ديرجات جديدة .
  وذلك لأننا قمنا بقراءة  كل محتويات الملف، ومنSeekنلحظ أننا في هذه الحالة لم نستخدم  الإجراء 

 المعروف أن القراءة  تقوم  بتحريك مؤشر الملف إلى الأمام ، لذلك بعد الفراغ من قراءة  كافة سجلته
يكون المؤشر في الخانة الأخيرة  في الملف، لذلك يمكن الإضافة مباشرة .

 التي تقوم  بإرجاع الموقع الحالي لمؤشر الملفات.FilePosاستخدمنا الدالة 

  لتسجيل بيانات سيارة ، نلحظ أننا نقوم  بكتابة وقراءة Recordفي المثال التالي سوف نستخدم  سجل 
السجل كوحدة  واحدة :

السيارات   سجل السيارات  برنامج سجل برنامج

program CarRecords;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

type
  TCar = record
    ModelName: string[20];
    Engine: Single;
    ModelYear: Integer;
  end;

var
  F: file of TCar;
  Car: TCar;
begin
  AssignFile(F, 'cars.dat');



  if FileExists('cars.dat') then
  begin
    FileMode:= 2; // Open file for read/write
    Reset(F); // open file
    Writeln('File already exist, opened for append');
    // Display file records
    while not Eof(F) do
    begin
      Read(F, Car);
      Writeln;
      Writeln('Car # ', FilePos(F), ' ­­­­­­­­­­­­­­­­­­­­­­­­­­');
      Writeln('Model : ', Car.ModelName);
      Writeln('Year  : ', Car.ModelYear);
      Writeln('Engine: ', Car.Engine);
    end
  end
  else // File not found, create it
  begin
    Rewrite(F);
    Writeln('File does not exist, created');
  end;

  Writeln('Please input car informaion, ', 
        'write x in model name to exit');
  Writeln('File pointer position at record # ', FilePos(f));

  repeat
    Writeln('­­­­­­­­­­­­­­­­­­­­­­­­­­');
    Write('Input car Model Name : ');
    Readln(car.ModelName);
    if Car.ModelName <> 'x' then
    begin
      Write('Input car Model Year : ');
      Readln(car.ModelYear);
      Write('Input car Engine size: ');
      Readln(car.Engine);
      Write(F, Car);
    end;
  until Car.ModelName = 'x';
  CloseFile(F);

  Write('Press enter key to close..');
  Readln;
end.

  هو منModelNameفي البرنامج السابق إستخدمنا سجل لمعلومات السيارة ، والحقل الأول في السجل 
 حرف:20 وهو يمثل 20النوع المقطعي، إل أننا في هذه الحالة قمنا بتحديد طول المقطع بالعددي 

  ModelName: string[20];

 وبهذه الطريقة يكون طول المقطع معروف ومحددي ويأخذ مساحة معروفة من القرص، أما طريقة
ًا والذي يسمى stringإستخدام  النوع    فهي طريقة لها تبعاتها في طريقة تخزينها فيAnsiString مطلق

الذاكرة ، وسوف نتكلم عنها في كتاب لحق إن شاء الله.



Files copyنسخ الملفات 

ًء كانت ملفات نصية أو ثنائية، صور ، برامج، أو غريرها فإنها تشترك  في أن  الملفات بكافة أنواعها سوا
 ، حيث أن أي ملف هو عبارة  عن مجموعة من البايتات، تختلفByteالوحدة  الأساسية فيها هي البايت 

 ، لذلك فإذا قرأنا أي ملف255في محتوياتها، لكن البايت يحتوي على أرقام  من القيمة صفر إلى القيمة 
).255 - 0فنجد أن رموزاه لتخرج عن هذه الإحتمالت (

ًا بإستخدام  متغير يحتل بايت ًا حرف  عملية نسخ الملف هي عملية بسيطة، فنحن نقوم  بنسخ الملف حرف
 . في هذه الحالة ليهم نوع الملف، لأن النسخ بهذهChar أو الرمز Byteواحد من الذاكرة  مثل البايت 

الطريقة يكون الملف المنسوخ صورة  طبق الأصل من الملف الأصلي:

البايت      طريق عن الملفات نسخ البايت     برنامج طريق عن الملفات نسخ برنامج

program FilesCopy;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  SourceName, DestName: string;
  SourceF, DestF: file of Byte;
  Block: Byte;
begin
  Writeln('Files copy');
  Write('Input source file name: ');
  Readln(SourceName);

  Write('Input destination file name: ');
  Readln(DestName);
  if FileExists(SourceName) then
  begin
    AssignFile(SourceF, SourceName);
    AssignFile(DestF, DestName);

    FileMode:= 0;   // open for read only
    Reset(SourceF); // open source file
    Rewrite(DestF); // Create destination file

    // Start copy
    Writeln('Copying..');
    while not Eof(SourceF) do
    begin
      Read(SourceF, Block); // Read Byte from source file
      Write(DestF, Block);  // Write this byte into new 
                                            // destination file 



    end;
    CloseFile(SourceF);
    CloseFile(DestF);

  end
  else // Source File not found
    Writeln('Source File does not exist');

  Write('Copy file is finished, press enter key to close..');
  Readln;
end.

ًل مثل في نظام   عند تشغيل هذا البرنامج يجب كتابة إسم الملف المرادي النسخ منه والملف الجديد كام
لينكس نكتب:

Input source file name: /home/motaz/quran/mishari/32.mp3
Input destination file name: /home/motaz/Alsajda.mp3

وفي نظام  وندوزا:

Input source file name: c:\photos\mypphoto.jpg
Input destination file name: c:\temp\copy.jpg

  موجودي في نفس الدليل للملف المصدر والنسخة، فيمكن كتابة إسميFileCopyأما إذا كان البرنامج 
ًل: الملف بدون كتابة إسم الدليل مث

Input source file name: test.pas
Input destination file name: testcopy.pas

 نلحظ أن برنامج نسخ الملفات يأخذ وقت طويل عند نسخ الملفات الكبيرة  مقارنة بنسخها بواسطة
 نظام  التشغيل نفسه، وذلك يعني أن نظام  التشغيل يستخدم  طريقة مختلفة لنسخ الملفات. فهذه

ًا بسبب قراءة  حرف واحد في الدورة  الواحدة  ثم نسخة في الملف الجديد، فلو كان  الطريقة بطيئة جد
حجم الملف مليون بايت فإن الحلقة تدور مليون مرة ، تتم فيها القراءة  مليون مرة  والكتابة مليون مرة .

 وكانت هذه الطريقة للشرح فقط، أما الطريقة المثلى لنسخ الملفات فهي عن طريق استخدام  الملفات
.untyped filesغرير محددية  النوع 



untyped filesالملفات غرير محددية  النوع 

 وهي ملفات ذات وصول عشوائي، إل أنها تختلف عن الملفات محددية  النوع في أنها لترتبط بنوع محددي،
 كذلك فإن القراءة  والكتابة غرير مرتبطة بعددي سجلت معين، حيث أن للمبرمج كامل الحرية في تحديد

عددي السجلت التي يرغرب في كتابتها أو قراءتها في كل مرة .

النوع        محددة غير الملفات بإستخدام الملفات نسخ النوع       برنامج محددة غير الملفات بإستخدام الملفات نسخ برنامج
program FilesCopy2;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  SourceName, DestName: string;
  SourceF, DestF: file;
  Block: array [0 .. 1023] of Byte;
  NumRead: Integer;
begin
  Writeln('Files copy');
  Write('Input source file name: ');
  Readln(SourceName);

  Write('Input destination file name: ');
  Readln(DestName);

  if FileExists(SourceName) then
  begin
    AssignFile(SourceF, SourceName);
    AssignFile(DestF, DestName);

    FileMode:= 0;   // open for read only
    Reset(SourceF, 1); // open source file
    Rewrite(DestF, 1); // Create destination file

    // Start copy
    Writeln('Copying..');
    while not Eof(SourceF) do
    begin
      // Read Byte from source file
      BlockRead(SourceF, Block, SizeOf(Block), NumRead); 
      // Write this byte into new destination file
      BlockWrite(DestF, Block, NumRead);  
    end;
    CloseFile(SourceF);



    CloseFile(DestF);

  end
  else // Source File not found
    Writeln('Source File does not exist');

  Write('Copy file is finished, press enter key to close..');
  Readln;
end.

في المثال السابق نجد أن هناك  أشياء جديدة  وهي:

طريقة تعريف الملفات، وهي بتعريفها أن المتغير مجردي ملف:.1

  SourceF, DestF: file;

المتغير الذي يستخدم  في نسخ البيانات بين الملفين:.2

  Block: array [0 .. 1023] of Byte;

 نجد أنه في هذه الحالة عبارة  مصفوفة من نوع البايت تحتوي على كيلو بايت، ويمكن تغييرها إلى أي
رقم يريده المبرمج.

ًل:.3 طريقة فتح  الملف إختلفت قلي

  Reset(SourceF, 1); // open source file
  Rewrite(DestF, 1); // Create destination file

ًا يحمل القيمة  فقد زاادي مُدخل جديد وهو طول السجل، وفي حالة نسخ ملفات هذه يجب أن يكون ديائم
 واحد وهو يعني أن طول السجل واحد بايت. والسبب يكمن في أن الرقم واحد يقبل القسمة على جميع

ًل يمكن أن يكون حجم الملف   بايت وهكذا.23490 بايت، أو 125قيم حجم الملفات، مث

طريقة القراءة :.4

  BlockRead(SourceF, Block, SizeOf(Block), NumRead); 

  مع الملفات غرير محددية  النوع، حيث يعتبر أن القيمة المرادي قراءتها هيBlockReadيستخدم  الإجراء 
عبارة  عن كومة أو رزامة غرير معلومة المحتويات. والمدخلت لهذا الإجراء هي:

SourceF.وهو متغير الملف المرادي القراءة  منه :

Block.وهو المتغير أو المصفوفة التي يرادي وضع محتويات القراءة  الحالية فيها : 

SizeOf(Block وهو عددي السجلت المرادي قراءتها في هذه اللحظة، ونلحظ أننا استخدمنا الدالة :(SizeOf 
.1024التي ترجع حجم المتغير من حيث عددي البايتات، وفي هذه الحالة هو الرقم 

NumRead  ًا في حالة أن تكون1024: عندما نقول أننا نريد قراءة   بايت ربما ينجح  الإجراء بقراءتها جميع
 هناك  بيانات متوفر في الملف، أما إذا كانت محتويات الملف أقل من هذه القيمة أو أن مؤشر القراءة 

  بايت، وتكون القيمة التي قرأها أقل1024وصل قرب نهاية الملف، ففي هذه الحالة ليستطيع قراءة  
ًل إذا كان حجم المف NumReadمن ذلك وتخزن القيمة في المتغير    بايت، فيقوم  الإجراء1034. فمث



  بايت فقط ويرجع هذه10 بايت في المرة  الأولى، أما في المرة  الثانية فيقوم  بقراءة  1024بقراءة  
.BlockWrite حتى يتسنى إستخدامها مع الإجراء NumReadالقيمة في المتغير 

طريقة الكتابة:.5

  BlockWrite(DestF, Block, NumRead);  

  في هذه المرة  عددي البايتات المرادي كتابتها في الملفNumReadوأظن أنها واضحة، والمتغير الأخيرة  
 . وعند تشغيل هذا الإجراء فإن المتغيرBlockالمنسوخ، وهي تعني عددي البياتات من بداية المصفوفة 

NumRead يحمل قيمة عددي البايتات التي تمت قراءتها عن طريق BlockRead

ًل إذا كان طول الملف  وعند تشغيل البرنامج سوف نلحظ السرعة الكبيرة  في نسخ الملفات، فمث
مليون بايت، فيلزم  حوالي أقل من ألف ديورة  فقط للقراءة  ثم الكتابة في الملف الجديد.

 في المثال التالي، يقوم  البرنامج بإظهار محتويات الملف بالطريقة المستخدمة في التخزين في الذاكرة 
أو الملف، فكما سبق ذكره فإن الملف هو عبارة  عن سلسلة من الحروف أو البايتات.

بالبايت     ملف محتويات عرض بالبايت    برنامج ملف محتويات عرض برنامج

program ReadContents;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  FileName: string;
  F: file;
  Block: array [0 .. 1023] of Byte;
  i, NumRead: Integer;
begin
  Write('Input source file name: ');
  Readln(FileName);

  if FileExists(FileName) then
  begin
    AssignFile(F, FileName);

    FileMode:= 0;   // open for read only
    Reset(F, 1);

    while not Eof(F) do
    begin
      BlockRead(F, Block, SizeOf(Block), NumRead);
      // display contents in screen



      for i:= 0 to NumRead ­ 1 do
        Writeln(Block[i], ':', Chr(Block[i]));
    end;
    CloseFile(F);

  end
  else // File does not exist
    Writeln('Source File does not exist');

  Write('press enter key to close..');
  Readln;
end.

بعد تنفيذ البرنامج يمكن للمستخدم  كتابة إسم ملف نصي حتى نقوم  بعرض محتوياته.
  في نظام  لينكس، أما في نظام 10 وقيمتها بالبايت Line Feed LFنلحظ أن بعد كل سطر نجد علمة الـ 

 ، وهي الفاصل10 و 13 وقيمتهما على التوالي Carriage Return/Line Feed CRLFوندوزا فنجد علمتي 
 الموجودي بين السطور في الملف النصي. يمكن كذلك إستخدام  البرنامج في فتح  ملفات من نوع آخر

لمعرفة طريقة تكوينها.
ًل نجد أن البايت الذي قيمتهChrإستخدمنا في البرنامج الدالة    التي تقوم  بتحويل البايت إلى حرف، مث

Date and Timeالتاريخ والوقت 

 التاريخ والوقت من الأساسيات المهمة في البرامج، فهي من الخصائص المهمة المرتبطة بالبيانات
ًل  والمعلومات، فمعظم المعلومات والبيانات تحتاج لتاريخ يمثل وقت إديراج هذه المعلومة أو تعديلها. فمث
 إذا كانت هناك  معاملة بنكية أو صرف شيك فلبد من تسجيل الوقت الذي حدثت فيه هذه المعاملة نسبة

ًل، أو عندما يقوم  البنك بحصر ًا عندما يطلب العميل كشف الحساب مث  لأهميتها في المراجعة لحق
ًل. المعاملت التي حدثت في شهر ما مث

  وهو عبارة  عن عددي حقيقي (كسري) ذوTDateTimeيتم تخزين التاريخ والوقت في متغير واحد من نوع 
  وهو يحتل ثماني خانات من الذاكرة . العددي الصحيح  منه يمثل الأيام ، والجزءDoubleديقة مضاعفة 

 لمتغير من هذا النوع تمثل الأيام  التي انقضت منذ يوم صفرالكسري يمثل الوقت. والقيمة 
 ميلديية.30/12/1899

 والتي تقوم  بإرجاع قيمة التاريخ والزمن الحاليين:Nowفي البرنامج التالي نستخدم  الدالة 

program DateTime;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes , SysUtils
  { you can add units after this };

begin
  Writeln('Current date and time: ', DateTimeToStr(Now));
  Write('Press enter key to close');
  Readln;
end.

  والتي تقوم  بتفسير قيمةSysUtils الموجودية  في المكتبة DateTimeToStrنلحظ أننا استخدمنا الدالة 
التاريخ والوقت وتحويلهما لمقطع مقروء حسب إعداديات المستخدم  في طريقة إظهار التاريخ.

فإذا لم نستخدمها سوف نحصل على عددي حقيقي ليمكن فهمه:

  Writeln('Current date and time: ', Now);

ًل: كذلك توجد ديوال لإظهار قيمة التاريخ فقط وأخرى لإظهار قيمة الوقت فقط مث

  Writeln('Current date is ', DateToStr(Now));
  Writeln('Current time is ', TimeToStr(Now));

 ,Date لكنهما يرجعان التاريخ فقط والأخرى الزمن فقط وهما Nowكذلك توجد ديالتان مشابهتان للدالة 
Time

  Writeln('Current date is ', DateToStr(Date));
  Writeln('Current time is ', TimeToStr(Time)); 



  ترجع فقط التاريخ وتقوم  بإرجاعDateهذه الدوال تقوم  بحذف الجزء الخر أو تحويله إلى صفر، فالدالة 
ًا.12صفر في الجزء الكسري الذي يمثل الزمن. والقيمة صفر بالنسبة للزمن تمثل الساعة   صباح

  تقوم  بإرجاع الزمن ووضع القيمة صفر في الجزء الصحيح  الذي يمثل التاريخ والذي قيمتهTimeوالدالة 
 . يمكن ملحظة هذه القيم في المثال التالي عند استخدام  الدالة30/12/1899كما ذكرنا 

DateTimeToStr  مع الدوال الجديدة Date, Time

  Writeln('Current date is ', DateTimeToStr(Date));
  Writeln('Current time is ', DateTimeToStr(Time));

  والتي تقوم  بتحويل التاريخFormatDateTimeيمكن أن يتضح  المثال السابق أكثر بإستخدام  الدالة 
والوقت بحسب الشكل الذي يريده المبرمج بغض النظر عن إعداديات المستخدم  في حاسوبه:

  Writeln('Current date is ', 
     FormatDateTime('yyyy­mm­dd hh:nn:ss', Date));
  Writeln('Current time is ', 
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Time));

في المثال التالي سوف نعامل التاريخ كعددي حقيقي فنقوم  بإضافة أو طرح قيم منه.

begin
  Writeln('Current date and time is ',
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Now));
  Writeln('Yesterday time is ', 
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Now ­ 1));
  Writeln('Tomorrow time is ', 
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Now + 1));
  Writeln('Today + 12 hours is ', 
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Now + 1/2));
  Writeln('Today + 6 hours is ', 
    FormatDateTime('yyyy­mm­dd hh:nn:ss', Now + 1/4));
  Write('Press enter key to close');
  Readln;
end.

  فإننا نقوم  بزيادية  يوم  أوNow أو طرحه من قيمة التاريخ الحالي 1نلحظ عندما نقوم  بزيادية  الرقم 
  ساعة، في هذه الحالة أثرنا12 فهي تعني ½بإنقاص يوم  على التوالي. وعندما نقوم  بزيادية  نصف يوم  

على الجزء الكسري في التاريخ الذي يمثل الوقت.

في المثال التالي نقوم  بتكوين تاريخ معين من السنين، الشهور، والأيام :

var
  ADate: TDateTime;
begin
  ADate:= EncodeDate(1975, 11, 16);
  Writeln('My date of birth is: ', FormatDateTime('yyyy­mm­dd', ADate));
  Write('Press enter key to close');
  Readln;
end.

  والتي مدخلتها هي السنة، الشهر ثم اليوم ،EncodeDateنلحظ في المثال السابق أننا استخدمنا الدالة 
.TDateTimeوهي تقوم  بتحويل وتجميع هذه القيم في متغير واحد من النوع 



في المثال التالي سوف نستخدم  ديالة مشابهة لكنها تختص بالوقت:

var
  ATime: TDateTime;
begin
  ATime:= EncodeTime(19, 22, 50, 0);
  Writeln('Almughrib prayer time is: ', FormatDateTime('hh:nn:ss', ATime));
  Write('Press enter key to close');
  Readln;
end. 

مقارنة التواريخ والأوقات

  أكبر من الرقم9.3يمكن مقارنة التواريخ والأوقات بنفس طريقة مقارنة الأعدادي الحقيقية، فالرقم 
5.1. 

 Now + 1فكما علمنا يمثل العددي الصحيح  الأيام  أو التاريخ، ويمثل العددي الكسري الوقت. فبالتأكيد قيمة 
ًا أكبر من  ًا تكون ديائم   والتي تعني بعد ساعة من الن تكون أكبرNow + 1/24  و قيمةNowوالتي تعني غرد

 والتي تعني قبل ساعتين من الن.Now – 2/24من 

 ، اليوم  الأول من شهر يناير، فنقوم  بمقارنته بالتاريخ2012في البرنامج التالي نضع تاريخ معين وهو عام  
الحالي لنعرف هل تجاوزانا هذا التاريخ أم  ليس بعد.

var
  Year2012: TDateTime;
begin
  Year2012:= EncodeDate(2012, 1, 1);

  if Now < Year2012 then
    Writeln('Year 2012 is not coming yet')
  else
    Writeln('Year 2012 is already passed');
  Write('Press enter key to close');
  Readln;
end.  

 يمكن إضافة معلومة جديدة  في البرنامج وهي عددي الأيام  المتبقية لهذا التاريخ أو الأيام  التي إنقضت
بعده في المثال التالي:

var
  Year2012: TDateTime;
  Diff: Double;
begin



  Year2012:= EncodeDate(2012, 1, 1);
  Diff:= Abs(Now ­ Year2012);

  if Now < Year2012 then
    Writeln('Year 2012 is not coming yet, there are ',
      Format('%0.2f', [Diff]), ' days Remaining ')
  else
    Writeln('First day of January 2012 is passed by ',
      Format('%0.2f', [Diff]), ' Days');
  Write('Press enter key to close');
  Readln;
end.  

  قيمة فرق التارخين بالأيام . ونلحظ أننا استخدمنا الدالةDiffفي المثال السابق سوف يحتوي المتغير 
Abs أي حذف السالب إن وجد والتي تقوم  بإرجاع القيمة المطلقة لعددي).(

الأخبار  الأخبار مسجل مسجل
 البرنامج التالي يستخدم  الملفات النصية لتسجيل عناوين الأخبار المكتوبة ويقوم  بإضافة الزمن الذي

 تمت كتابة الخبر عنده بإعتبارة  وقت حدوث الخبر. عند إغرلق البرنامج وتشغيله مرة  أخرى يعرض
البرنامج الأخبار السابقة التي تمت كتابتها مع عرض الزمن الذي كتبت فيه.

program news;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var
  Title: string;
  F: TextFile;
begin
  AssignFile(F, 'news.txt');
  if FileExists('news.txt') then
  begin       // Display old news
    Reset(F);
    while not Eof(F) do
    begin
      Readln(F, Title);
      Writeln(Title);
    end;
    CloseFile(F); // reading is finished from old news
    Append(F);   // open file again for appending
  end
  else
    Rewrite(F);



  Write('Input current hour news title: ');
  Readln(Title);
  Writeln(F, DateTimeToStr(Now), ', ', Title);
  CloseFile(F);

  Write('Press enter to close');
  Readln;
end.



constantsالثوابت 

 الثوابت تشبه المتغيرات من حيث أن كلٍٍ يمثل قيمة معينة مثل قيمة لعددي صحيح ، حقيقي أو مقطع. إل
 أن الثوابت تبقى قيمتها ثابتة لتتغير، بل أن قيمتها محددية  في وقت ترجمة البرنامج. بالتالي نستطيع أن

ًا أثناء قيامه بترجمة البرنامج وتحويله إلى لغة آلة. نقول أن المترجم يعرف قيمة الثابت سلف
مثال لقيمة ثابتة:

  Writeln(5);
  Writeln('Hello');

  في كل الأحوال وليمكن أن تتغير هذه القيمة،5نجد أن البرنامج عند تنفيذ هذا السطر يكتب الرقم 
 ليتغير.Helloكذلك فإن المقطع 

 عندما نتكلم عن الثوابت فإننا نقصد الأرقام  المحددية  أو المتغيرات المقطعية أو أي نوع آخر تمت كتابته
في البرنامج بالطريقة السابقة أو بالطريقة التالية:

الوقود   إستهلك الوقود  برنامج إستهلك برنامج

program PetrolConsumption;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

const GallonPrice = 6.5;

var
  Payment: Integer;
  Consumption: Integer;
  Kilos: Single;
begin
  Write('How much did you pay for your car''s petrol: ');
  Readln(Payment);
  Write('What is the consumption of your car? (Kilos per Gallon): ');
  Readln(Consumption);

  Kilos:= (Payment / GallonPrice) * Consumption;

  Writeln('This petrol will keep your car running for : ',
    Format('%0.1f', [Kilos]), ' Kilometers');
  Write('Press enter');
  Readln;
end.



ًء على عدة  عوامل: البرنامج السابق يحسب عددي الكيلومترات التي سوف تقطعها السيارة  بنا
  حتى نضع فيه عدديConsumptionمعدل إستهلك  السيارة  للوقودي، وقد استخدمنا المتغير .1

الكيلومترات التي تقطعها السيارة  بالجالون الواحد من الوقودي
Paymentقيمة النقودي التي ديفعها سائق السيارة  لشراء الوقودي، وقد استخدمنا المتغير .2
  لتخزين قيمة سعر الجالون والتي تعتبر قيمةGallonPriceسعر الجالون، وقد استخدمنا الثابت .3

ًا قمنا بتثبيتها في البرنامج حتى لنطلب من المستخدم  إديخالها في كل مرة  مع الـ  ثابتة نسبي
payment وال Consumptionفإذا تغير سعر الوقودي ماعلينا إل الذهاب إلى أعلى البرنامج وتغير . 

قيمة هذه الثابت ثم إعادية  ترجمة البرنامج. 

 يفضل استخدام  هذه الطريقة للثوابت عندما نستخدمها لأكثر من مرة  في البرنامج، فتغيير القيمة في
التعريف تغني عن التغيير في كل أجزاء البرنامج.



Ordinal typesالأنواع العدديية 

 الأنواع العدديية هي عبارة  عن نوع يحمل قِيم عدديية صحيحة ذات أسماء، ونعاملها بأسمائها لتسهيل
ًل في البرنامج التالي نريد أن نقوم  بتخزين اللغة التي يريد استخدامها  قراءة  الكودي وفهم البرنامج. مث

ّير  مستخدم  البرنامج، وقد قمنا بتخييره بين اللغة العربية والإنجليزية، وكان من الممكن أن نستخدم  متغ
  للغة الإنجليزية. لكننا استخدمنا الأنواع العدديية ليكون2 للغة العربية و1صحيح  ونضع فيه قيم مثل 

ًا للمبرمج: البرنامج أكثر وضوح

program OrdinaryTypes;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

type
  TLanguageType = (ltArabic, ltEnglish);

var
  Lang: TLanguageType;
  AName: string;
  Selection: Byte;
begin
  Write('Please select Language: 1 (Arabic), 2 (English)');
  Readln(Selection);

  if Selection = 1 then
    Lang:= ltArabic
  else
  if selection = 2 then
    Lang:= ltEnglish
  else
    Writeln('Wrong entry');

  if Lang = ltArabic then
    Write('ماهو اسمك: ')
  else
    if Lang = ltEnglish then
      Write('What is your name: ');

  Readln(AName);

  if Lang = ltArabic then
  begin
    Writeln('مرحباً بك ', AName);
    Write('الرجاء الضغط على مفتاح إدخال لغللق  البرنامج');
  end
  else
  if Lang = ltEnglish then
  begin



    Writeln('Hello ', AName);
    Write('Please press enter key to close');
  end;
  Readln;

end.

 الأرقام  الصحيحة والحروف تعتبر من الأنواع العدديية،أما الأعدادي الكسرية والمقاطع فلتعتبر من الأنواع
العدديية.



setsالمجموعات 

 المجموعات هي عبارة  عن نوع من المتغيرات يستطيع جمع عددي من القيم أو الخصائص، شرط أن
ًل عندما نريد تصنيف برامج من حيث أنظمة التشغيل التي تعمل فيها نلجأ  تكون من الأنواع العدديية، فمث

إلى التي:

TApplicationEnvتعريف نوع معدودي يمثل أنظمة التشغيل، كما في المثال التالي .1
نقوم  بتعريف متغيرات للبرامج المرادي تصنفيها بالشكل التالي:.2

  AppName: set of TApplicationEnv;

نقوم  بإسنادي أنواع أنظمة التشغيل في شكل مجموعة كالتالي:.3

  AppName:= [aeLinux, aeWindows];

والمثال هو:

program Sets;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

type
  TApplicationEnv = (aeLinux, aeMac, aeWindows);

var
  FireFox: set of TApplicationEnv;
  SuperTux: set of TApplicationEnv;
  Delphi: set of TApplicationEnv;
  Lazarus: set of TApplicationEnv;
begin
  FireFox:= [aeLinux, aeWindows];
  SuperTux:= [aeLinux];
  Delphi:= [aeWindows];
  Lazarus:= [aeLinux, aeMac, aeWindows];

  if aeLinux in Lazarus then
    Writeln('There is a version for Lazarus under Linux')
  else
    Writeln('There is no version of Lazarus under linux');

  if aeLinux in SuperTux then
    Writeln('There is a version for SuperTux under Linux')
  else
    Writeln('There is no version of SuperTux under linux');

  if aeMac in SuperTux then
    Writeln('There is a version for SuperTux under Mac')



  else
    Writeln('There is no version of SuperTux under Mac');

  Readln;
end.

يمكن كذلك إستخدام  صيغة المجموعات مع الأنواع العدديية أو الصحيحة كالتي:

Monthمثال لإستخدام  متغير صحيح : 

  if Month in [1, 3, 5, 7, 8, 10, 12] then
    Writeln('This month contains 31 days');

كذلك يمكننا إستخدام  الحروف كلتي:

if Char in [' 'آ, 'إ', 'أ', ا' '] then
    Writeln('هذا حرف الألف');



Exception handlingمعالجة الإعتراضات 

ّير بدون تعريفه فيcompilation errorsيوجد نوعان من الأخطاء: أخطاء أثناء الترجمة    مثل إستخدام  متغ
 ، أو كتابة عبارة  بطريقة خاطئة. فهذه الأخطاء ليمكن تجاوزاها، وتحول ديون ترجمة وربطvarقسم 

.Messagesالبرامج، ويقوم  المترجم بالتنبيه عن وجوديها في نافذة  
ًل إذا طلب البرنامج من  أما النوع الثاني من الأخطاء فهي الإعتراضات التي تحدث بعد تنفيذ البرنامج، مث

 المستتخدم  إديخال رقمين لإجراء القسمة عليهما، وقام  المستخدم  بإديخال صفر في المقسوم  عليه،
  وهي من الأشياءDivision by Zeroففي هذه الحالة يحدث الخطأ المعروف بالقسمة على الصفر

 الممنوعة في الرياضيات. أو مثل محاولة فتح  ملف غرير موجودي، أو محاولة الكتابة في ملف ليس
 للمستخدم  صلحية عليه (من ناحية نظام  التشغيل). فمثل هذه الأخطاء تسمى أخطاء في وقت التنفيذ

run-time errorsُتركت على  . وليس من السهولة أن يتوقعها المترجم، لذلك فمعالجة هذه الإستثناءات 
  يستطيعreliableعاتق المبرمج، حيث أن على المبرمج أن يقوم  بكتابة برنامج أو كودي شديد الإعتماديية 

التغلب على معظم الأخطاء التشغيلية التي يمكن أن تحدث. 
ُطرق منها حماية جزء معين من الكودي بإستخدام  عبارة    try أو try exceptمعالجة الإعتراضات تتم بعدة  

finally

try exceptعبارة  

تكون بهذه الطريقة:

try
  // Start of protected code
  CallProc1;
  CallProc2;
  // End of protected code

except
on e: exception do // Exception handling
begin
  Writeln('Error: ' + e.message);
end;
end;

   
:exceptمثال لبرنامج قسمة عدديين بإستخدام  

program ExceptionHandling;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, sysutils
  { you can add units after this };

var
  x, y: Integer;
  Res: Double;
begin



  try
    Write('Input x: ');
    Readln(x);
    Write('Input y: ');
    Readln(y);
    Res:= x / y;
    Writeln('x / y = ', Res);

  except
  on e: exception do
  begin
    Writeln('An error occured: ', e.message);
  end;
  end;
  Write('Press enter key to close');
  Readln;
end.

  وهو الكودي المحمي، والكوديtry .. exceptنلحظ أن هناك  قسمين للكودي: الكودي الموجودي بين عبارتي 
  وهو يمثل قسم معالجة الإستثناءات، ويتم نتفيذه فقط عند حدوثexcept .. endالموجودي بين عبارتي 

خطأ.

  يتم تنفيذه.except فإن الكودي في قسم معالجة الإستثناءات y للمتغير صفرعندما نقوم  بإديخال القيمة 
 ومؤشر التنفيذ في البرنامج عند حدوث خطأ يقفز إلى قسم معالجة الإستثناءات ويترك  تنفيذ باقي الكودي

 yالمحمي، ففي هذه الحالة الكودي التالي ليتم تنفيذه عند حدوث خطأ أثناء القسمة، مثل أن تكون قيمة 
ًا: صفر

Writeln('x / y = ', Res);

 أما في حالة عدم  حدوث خطأ ، أي في الحالت العاديية يتم تنفيذ كل الكودي المحمى وليتم تنفيذ الكودي
في قسم معالجة الأخطاء باللون الأحمر.

try finallyعبارة  
try
  // Start of protected code
  CallProc1;
  CallProc2;
  // End of protected code

finally
  Writeln('This line will be printed in screen for sure');
end;

   

:finallyمثال لبرنامج قسمة عدديين بإستخدام  

program ExceptionHandling;



{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

var
  x, y: Integer;
  Res: Double;
begin
  try
    Write('Input x: ');
    Readln(x);
    Write('Input y: ');
    Readln(y);
    Res:= x / y;
    Writeln('x / y = ', Res);

  finally
    Write('Press enter key to close');
    Readln;
  end;
end.

  باللون الأخضر الفاتح ، يتم تنفيذه في كافة الأحوال، في حالfinallyفي هذه المرة  فإن الكودي في قسم 
ّفة الأحوال. حدوث خطأ أو في حال عدم  حدوث خطأ. وبهذه الطريقة نضمن تنفيذ كودي معين في كا

raise  an exceptionرفع الإستثناءات 

ًل إذا طلبنا من المستخدم  الإلتزام  بمدى  يمكن للمبرمج رفع إستثناء، وذلك عند حدوث شرط معين، فمث
ٍذ يمكن إفتعال إسثناء مصحوب برسالة خطأ  معين لمُدخل ما، وقام  المستخدم  بتجاوزا هذا المدى، حينئ

ًل البرنامج التالي: معين. مث

program RaiseExcept;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

var



  x: Integer;
begin
  Write('Please input a number from 1 to 10: ');
  Readln(X);
  try

    if (x < 1) or (x > 10) then  // rais exception
      raise exception.Create('X is out of range');
    Writeln(' X * 10 = ', x * 10);

  except
  on e: exception do  // Catch my exception
  begin
    Writeln('Error: ' + e.Message);
  end;
  end;
  Write('Press enter to close');
  Readln;

end.

 ، فإن الخطأ التالي سوف10 إلى 1نلحظ أنه إذا قام  المستخدم  بتجاوزا المدى والذي هو محددي من 
 try except وبما أننا في نفس البرنامج قد قمنا بحماية الكودي بعبارة  x is out of rangeيحدث في البرنامج 

فإن هذا الخطأ سوف يظهر للمستخدم  بطريقة لتؤثر على إستمرارية البرنامج.



الثاني  الثاني الفصل الفصل

الهيكلية  الهيكلية البرمجة البرمجة
  Structured ProgrammingStructured Programming



مقدمة

 البرمجة الهيكلية هي طريقة للبرمجة ظهرت الحاجة لها بعد التوسع الذي حدث في البرامج، حيث أصبح 
من الصعب كتابة وتنقيح  وتطوير البرامج الكبيرة  المكتوبة في شكل ملف واحد ككتلة واحدة .

 البرمجة الهيكلية هي إتباع إسلوب تقسيم البرامج إلى وحدات أو مكتبات وإجراءات وديوال في شكل
 منظم ومجزاء يمكن الإستفادية  منها أكثر من مرة  مع سهولة القراءة  وتتبع الأخطاء في الأجزاء التي بها

مشاكل في البرامج. فنجد أن من أهم فوائد البرمجة الهيكلية هي:

ُتسّهل فهم البرامج.1 تقسيم البرنامج إلى وحدات برمجية أصغر 
 إمكانية إستخدام  (نداء) هذه الإجراءات في أكثر من موضع في البرنامج الواحد أو عددي من البرامج.2

code re-usabilityبإستخدام  الوحدات. وهذه الخاصية تسمى بإعادية  إستخدام  الكودي 
 إمكانية مشاركة تصميم وكتابة برنامج واحد لأكثر من مبرمج، حيث يمكن لأي مبرمج كتابة وحدة .3

منفصلة أو إجراء منفصل ثم يتم تجميعها في برنامج واحد.
 سهولة صيانة البرنامج، حيث تسهل معرفة الأخطاء وتصحيحها بالتركيز على الدالة أو الإجراء الذي.4

ينتج عنه الخطأ.

proceduresالإجراءات 

 لعلنا استخدمنا البرمجة الهيكلية من قبل، فقد مر علينا كثير من الدوال والإجراءات والوحدات التي هي
 عبارة  عن أجزاء من مكتبات فري باسكال، لكن الفرق الن أن المبرمج في هذه المرحلة سوف يقوم 

بكتابة إجراءات وديوال خاصة به تمثل خصوصية البرنامج الذي يقوم  بكتابته.

SayGoodbye و SayHelloفي المثال التالي قمنا بكتابة إجرائين: 

program Structured;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

procedure SayHello;
begin
  Writeln('Hello there');
end;

procedure SayGoodbye;
begin
  Writeln('Good bye');
end;

begin  // Here main application starts
  Writeln('This is the main application started');
  SayHello;



  Writeln('This is structured application');
  SayGoodbye;
  Write('Press enter key to close');
  Readln;
end.

 أظن أن الفكرة  واضحة. ففي حالة منادياتنا لهذه الإجراءات يقوم  البرنامج بتنفيذ الكودي المكتوب في
ّغر. الإجراء والذي كأنه برنامج مص

Parametersالمدخلت 

 . والمدخلت هي قيم تمرر عندparametersفي البرنامج التالي نقوم  بعمل إجراء يحتوي على مدخلت 
نداء الإجراء:

procedure WriteSumm(x, y: Integer);
begin
  Writeln('The summation of ', x, ' + ', y, ' = ', x + y)
end;

begin
  WriteSumm(2, 7);
  Write('Press enter key to close');
  Readln;
end. 

  والذي يقوم  بإستقبال هذه القيم فيWriteSumm مررتا عند نداء الإجراء 7 و 2نلحظ أن القيمتان 
 ليكتب حاصل جمعهما.x, yالمتغيرات 

في المثال التالي نقوم  بإعادية  كتابة برنامج المطعم بطريقة الإجراءات:

الإجراءات    بإستخدام المطعم الإجراءات   برنامج بإستخدام المطعم برنامج
procedure Menu;
begin
  Writeln('Welcome to Pascal Restaurant. Please select your order');
  Writeln('1 ­ Chicken      (10 Geneh)');
  Writeln('2 ­ Fish         (7 Geneh)');
  Writeln('3 ­ Meat         (8 Geneh)');
  Writeln('4 – Salad        (2 Geneh)');
  Writeln('5 ­ Orange Juice (1 Geneh)');
  Writeln('6 ­ Milk         (1 Geneh)');
  Writeln;
end;

procedure GetOrder(AName: string; Minutes: Integer);
begin
  Writeln('You have ordered : ', AName, ', this will take ', 
    Minutes, ' minutes');



end;

// Main application

var
  Meal: Byte;
begin

  Menu;
  Write('Please enter your selection: ');
  Readln(Meal);

  case Meal of
    1: GetOrder('Chicken', 15);
    2: GetOrder('Fish', 12);
    3: GetOrder('Meat', 18);
    4: GetOrder('Salad', 5);
    5: GetOrder('Orange juice', 2);
    6: GetOrder('Milk', 1);
  else
    Writeln('Wrong entry');
  end;
  Write('Press enter key to close');
  Readln;
end.

  وMenuنلحظ أن البرنامج الرئيسي أصبح  أوضح  وسهل الكتابة، والأجزاء الأخرى مثل القائمة الرئيسية 
  أصبح  أصغر وتمMainالطلب منها أصبحتا إجراءات منفصلة. كذلك فإن الجزء الرئيسي في البرنامج 

التخلص من أجزاء كبيرة  من الكودي  لتصبح  إجراءات منفصلة.
  تم ندائه عددي من المرات وفي كل مرة  بمدخلت مختلفة، وهنا تحقق إعادية GetOrderنلحظ أن الإجراء 

إستخدام  الكودي.



functionsالدوال 

 الدوال شبيهة بالإجراءات إل أنها تزيد عليها بأنه تقوم  بإرجاع قيمة. فقد استخدمنا ديوال من قبل مثل
Abs التي ترجع القيمة المطلقة، و UpperCaseالتي تستقبل المقطع كمدخل وتقوم  بإرجاع مقطع جديد  

محول إلى الحروف الإنجليزية الكبيرة .

  والتي تستقبل مدخلين من الأعدادي الصحيحة فتقوم GetSummفي البرنامج التالي قمنا بكتابة ديالة 
بإرجاع قيمة حاصل جمعهما فقط ديون كتابته:

function GetSumm(x, y: Integer): Integer;
begin
  Result:= x + y;
end;

var
  Sum: Integer;
begin
  Sum:= GetSumm(2, 7);
  Writeln('Summation of 2 + 7 = ', Sum);
  Write('Press enter key to close');
  Readln;
end.

  والتي تعبرResultفي هذه الحالة. كذلك استخدمنا الكلمة  Integerنلحظ هذه المرة  أن للدالة نوع وهو  
  لإستقبالSumعن القيمة التي سوف ترجع عند منادياة  هذه الدالة. وعند ندائها قمنا بإستخدام  المتغير 

 Writelnقيمة حاصل الجمع. لكن يمكن أن نستغنى عن هذا المتغير ونقوم  بإديخال نداء الدالة في الإجراء 
وهو فرق آخر للدوال عن الإجراءات، حيث ليمكن نداء إجراء بهذه الطريقة، فقط الدوال:

function GetSumm(x, y: Integer): Integer;
begin
  Result:= x + y;
end;

begin
  Writeln('Summation of 2 + 7 = ', GetSumm(2, 7));
  Write('Press enter key to close');
  Readln;
end.

في البرنامج التالي قمنا بإعادية  كتابة برنامج المطعم بإستخدام  الدوال:



وحلقة      الدوال بإستخدام المطعم وحلقة     برنامج الدوال بإستخدام المطعم repeatrepeatبرنامج

procedure Menu;
begin
  Writeln('Welcome to Pascal Resturant. Please select your order');
  Writeln('1 ­ Chicken      (10 Geneh)');
  Writeln('2 ­ Fish         (7 Geneh)');
  Writeln('3 ­ Meat         (8 Geneh)');
  Writeln('4 – Salad        (2 Geneh)');
  Writeln('5 ­ Orange Juice (1 Geneh)');
  Writeln('6 ­ Milk         (1 Geneh)');
  Writeln('X ­ nothing');
  Writeln;
end;

function GetOrder(AName: string; Minutes, Price: Integer): Integer;
begin
  Writeln('You have ordered: ', AName, ' this will take ',
    Minutes, ' minutes');
  Result:= Price;
end;

var
  Selection: Char;
  Price: Integer;
  Total: Integer;
begin
  Total:= 0;
  repeat
    Menu;
    Write('Please enter your selection: ');
    Readln(Selection);

    case Selection of
      '1': Price:= GetOrder('Chicken', 15, 10);
      '2': Price:= GetOrder('Fish', 12, 7);
      '3': Price:= GetOrder('Meat', 18, 8);
      '4': Price:= GetOrder('Salad', 5, 2);
      '5': Price:= GetOrder('Orange juice', 2, 1);
      '6': Price:= GetOrder('Milk', 1, 1);
      'x', 'X': Writeln('Thanks');
      else
      begin
        Writeln('Wrong entry');
        Price:= 0;
      end;
    end;

    Total:= Total + Price;

  until (Selection = 'x') or (Selection = 'X');
  Writeln('Total price            = ', Total);
  Write('Press enter key to close');
  Readln;
end. 



Local Variablesالمتغيرات المحلية 

ًا في الإجراء أو  يمكن تعريف متغيارت محلية دياخل الإجراء أو الدالة بحيث يكون استخدامها فقط محلي
ًل: الدالة وليمكن الوصول لهذه المتغيرات من البرنامج الرئيسي. مث

procedure Loop(Counter: Integer);
var
  i: Integer;
  Sum: Integer;
begin
  Sum:= 0;
  for i:= 1 to Counter do
    Sum:= Sum + i;
  Writeln('Summation of ', Counter, ' numbers is: ', Sum);
end;

begin   // Main program section
  Loop;
  Write('Press enter key to close');
  Readln;
end.

ًاi, Sum به تعريفات لمتغيرات محلية وهي Loopنلحظ أن الإجراء    يتم حجز جزء من الذاكرة  لهما مؤقت
 ) في المثالi, Sum وهي ذاكرة  للمتغيرات المؤقتة مثل المتغيرات المحلية(Stackفي مايعرف بالمكدسة 

  في هذا المثال. وتتميز المتغيرات التيCounterالسابق والمدخلت للجراءات والدوال مثل المدخل 
 تخزن في هذا النوع من الذاكرة  بقصر عمرها، إذ أن المتغيرات تكون صالحة فقط في الفترة  التي يقوم 

 البرنامج بنداء وتنفيذ هذه الدوال والإجراءات، فبعد نهاية تنفيذ الدالة والفراغ منها، كما في المثال
السابق عند الوصول  لهذه النقطة في البرنامج الرئيسي:

  Write('Press enter key to close');

تكون هذه المتغيرات قد إنقضت وليمكن الوصول إليها أو لقيمها مرة  أخرى بالطرق العاديية.
 وهذه بخلف المتغيرات التي يقوم  المبرمج بتعريفها للستخدام  في البرنامج الرئيسي كما اعتدنا

ًا حيث أنها تكون صالحة للستخدام  منذ بداية تشغيل البرنامج  استخدمناها سابقا، فهي تكون أكثر عمر
 إلى نهايته، فإذا كان البرنامج يحتوي على حلقة فإنها تعمّر فترة  طويلة، وإذا تم تشغيل البرنامج لمدة 

ساعة، كان عمر هذه المتغيرت ساعة، وهكذا.

ًا هي إحدى طرق تحقيق البرمجة الهيكلية، حيث أنها توفر حماية  هذه الطريقة لتعريف المتغيرات محلي
 وخصوصية لهذا الجزء القائم بذاته من البرنامج (الإجراء أو الدالة) وتجعله غرير مرتبط بمتغيرات أخرى

 Loopفي البرنامج مما يسهّل نقله لبرامج أخرى أو الإستفادية  منه عدة  مرات. فمن أرادي نداء هذه الدالة 
ًا من المتغيرات التي تعينه علىCounterماعليه إل مدها بقيمة    ثم يقوم  الإجراء بتلبية إحتياجاته محلي

تنفيذ هذا الكودي مما يحقق إعادية  إستخدام  الكودي وسهولة الصيانة.



الأخبار    بيانات قاعدة الأخبار   برنامج بيانات قاعدة برنامج

 البرنامج التالي فيه ثلث إجراءات وديالة واحدة ، وفيه عدة  إمكانات وهي : إضافي عنوان خبر جديد،
عرض كافة الأخبار والبحث عن خبر بإستخدام  كلمة مفتاحية:

program news;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this };

type
  TNews = record
    ATime: TDateTime;
    Title: string[100];
  end;

procedure AddTitle;
var
  F: file of TNews;
  News: TNews;
begin
  AssignFile(F, 'news.dat');
  Write('Input current news title: ');
  Readln(News.Title);
  News.ATime:= Now;
  if FileExists('news.dat') then
  begin
    FileMode:= 2; // Read/Write
    Reset(F);
    Seek(F, System.FileSize(F)); // Go to last record to append
  end
  else
    Rewrite(F);
  Write(F, News);
  CloseFile(F);
end;

procedure ReadAllNews;
var
  F: file of TNews;
  News: TNews;
begin
  AssignFile(F, 'news.dat');
  if FileExists('news.dat') then
  begin
    Reset(F);
    while not Eof(F) do
    begin
      Read(F, News);



      Writeln('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');
      Writeln('Title: ', News.Title);
      Writeln('Time : ', DateTimeToStr(News.ATime));
    end;
    CloseFile(F);
  end
  else
    Writeln('Empty database');
end;

procedure Search;
var
  F: file of TNews;
  News: TNews;
  Keyword: string;
  Found: Boolean;
begin
  AssignFile(F, 'news.dat');
  Write('Input keyword to search for: ');
  Readln(Keyword);
  Found:= False;
  if FileExists('news.dat') then
  begin
    Reset(F);
    while not Eof(F) do
    begin
      Read(F, News);
      if Pos(LowerCase(Keyword), LowerCase(News.Title)) > 0 then
      begin
        Found:= True;
        Writeln('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');
        Writeln('Title: ', News.Title);
        Writeln('Time : ', DateTimeToStr(News.ATime));
      end;
    end;
    CloseFile(F);
    if not Found then
      Writeln(Keyword, ' Not found');
  end
  else
    Writeln('Empty database');
end;

function Menu: char;
begin
  Writeln;
  Writeln('...........News database..........');
  Writeln('1. Add news title');
  Writeln('2. Display all news');
  Writeln('3. Search');
  Writeln('x. Exit');
  Write('Please input a selection : ');
  Readln(Result);
end;



// Main application

var
  Selection: Char;

begin 

  repeat
    Selection:= Menu;
    case Selection of
      '1': AddTitle;
      '2': ReadAllNews;
      '3': Search;
    end;
  until LowerCase(Selection) = 'x';
end.

ًا ومقسم إلى أقسام  قائمة بذاتها، حيث يمكن  نجد أن برنامج الأخبار هذه المرة ، أصبح  أكثر وضوح
 للمبرمج حتى ولو لم يكن هو الذي كتب البرنامج أن يفهمه من أول وهلة وتعدليه أو تطويره أو زايادية 

 ميزات أخرى (زايادية  إجراءات أو ديوال) بكل سهولة. كذلك يمكن الإستفادية  من بعض أجزائه في برامج
أخرى.



الدالة كمُدخل

 الفرق الثاني والمهم بين الدوال والإجراءات هو أن الدالة يمكن ندائها من دياخل إجراء أو ديالة أخرى
 كمُدخل ، وهذه الميزة  ناتجة عن الخاصية الأولى للدوال وهي أنها ترجع قيمة، أي يكن معاملتها كأنها

ًل: قيمة ثابتة أو متغير نريد قراءة  قيمته، مث

function DoubleNumber(x: Integer): Integer;
begin
  Result:= x * 2;
end;

// Main

begin
  Writeln('The double of 5 is : ', DoubleNumber(5));
  Readln;
end.

  ، وكان يمكن أن نستخدم  الطريقةWriteln دياخل الإجراء DoubleNumberنلحظ أننا قمنا بتمرير الدالة 
الطويلة كالتي:

function DoubleNumber(x: Integer): Integer;
begin
  Result:= x * 2;
end;

// Main

var
  MyNum: Integer;
begin
  MyNum:= DoubleNumber(5);
  Writeln('The double of 5 is : ', MyNum);
  Readln;
end.

كذلك يمكن نداء الدوال دياخل الشروط وهذه الميزات لتتحقق في الإجراءات:

function DoubleNumber(x: Integer): Integer;
begin
  Result:= x * 2;
end;

// Main

begin
  if DoubleNumber(5) > 10 then
    Writeln('This number is larger than 10')
  else
    Writeln('This number is equal or less than 10);
  Readln;
end.



مخرجات الإجراءات والدوال

 لحظنا من الأمثلة السابقة أننا نستطيع جعل الدالة ترجع قيمة واحدة ، لكن ماذا لو أردينا إرجاع قيمتين أو
أكثر، ففي هذه الحالة ليصلح  أن نستخدم  فقط قيمة ماترجعه الدالة.

 يكمن الحل لهذه المشكلة في إستخدام  المدخلت للجراءات والدوال. حيث نجد أن المدخلت نفسها
 يمكن أن تكون مخرجات بتغيير بسيط في طريقة التعريف. لكن لشرح هذا المفهوم  ديعنا نقوم  بهذه

التجربة:

procedure DoSomething(x: Integer);
begin
  x:= x * 2;
  Writeln('From inside procedure: x = ', x);
end;

//  main

var
  MyNumber: Integer;
begin
  MyNumber:= 5;

  DoSomething(MyNumber);
  Writeln('From main program, MyNumber = ', MyNumber);
  Writeln('Press enter key to close');
  Readln;
end.

     
  ثم يقوم x يقوم  بإستقبال مدخل في المتغير DoSomethingنجد في التجربة السابقة أن الإجراء 

 MyNumberبمضاعفة قيمة هذا الرقم وإظهاره في الشاشة. وفي البرنامج الرئيسي قمنا بتعريف متغير 
  كمدخل لهذاMyNumber واستخدمنا المتغير DoSomething ثم قمنا بنداء الإجراء 5ووضعنا فيه القيمة 

  في الإجراء السابق الذكر. فيx إلى المتغير MyNumberالإجراء. في هذه الحالة تنتقل قيمة المتغير 
  ثم يقوم  بإظهار10 لتصبح  5 والتي بدورها تحتوي على القيمة xهذه الحالة يقوم  الإجراء بمضاعفة قيمة 

 هذه القيمة. بعد الفراغ من تنفيذ هذا الإجراء يعودي مسار البرنامج إلى الجزء الرئيسي من البرنامج
  لم تؤثرx، أي أن مضاعفة قيمة 5 الذي نجده يحتفظ بالقيمة MyNumberونقوم  بإظهار قيمة المتغير 

 ، حيث أننا نسخنا قيمة المدخلcalling by valueعليه. هذه الطريقة من النداء تسمى النداء بالقيمة 
MyNumberإلى متغير x.جديد مؤقت ليؤثر على المتغير في البرنامج الرئيسي 

calling by referenceالنداء بإستخدام  المرجع 

 إلى الإجراء السابق سوف تتغير الموازاين فتصبح  النتيجة مختلفة هذه المرة :varإذا قمنا بإضافة الكلمة 

procedure DoSomething(var x: Integer);
begin
  x:= x * 2;
  Writeln('From inside procedure: x = ', x);
end;

//  main



var
  MyNumber: Integer;
begin
  MyNumber:= 5;

  DoSomething(MyNumber);
  Writeln('From main program, MyNumber = ', MyNumber);
  Writeln('Press enter key to close');
  Readln;

end.

  حيث أنهما يتشاركانMyNumber أثرت مباشرة  على قيمة xفي هذه الحالة نجد أن مضاعفة قيمة 
  فكان لكل متغير منها حيز منفصل فيvarالموقع أو العنوان في الذاكرة ، أما في الحالة السابقة بدون 

الذاكرة .
 لكن الشرط هذه المرة  أن نمرر مدخل في شكل متغير من نفس النوع وليصلح  أن يكون ثابت، حيث

ليمكن أن نقوم  بنداء الدالة بهذه الطريقة:

  DoSomething(5);

 في الحالة الأولى كان يمكن أن يكون هذا الإجراء صحيح  لأن الطريقة كانت نداء بوساطة قيمة وفي
.5هذه الحالة القيمة فقط هي المهمة وهي 

Swapفي المثال التالي قمنا بكتابة إجراء يقوم  بإستبدال قيمتين ببعضهما أو مايسمى بالـ 

procedure SwapNumbers(var x, y: Integer);
var
  Temp: Integer;
begin
  Temp:= x;
  x:= y;
  y:= Temp;
end;

//  main

var
  A, B: Integer;
begin

  Write('Please input value for A: ');
  Readln(A);

  Write('Please input value for B: ');
  Readln(B);

  SwapNumbers(A, B);
  Writeln('A = ', A, ', and B = ', B);
  Writeln('Press enter key to close');
  Readln;
end.  



unitsالوحدات 

 الوحدات هي عبارة  عن مكتبات للجراءات والدوال التي تستخدم  بكثرة  بالإضافة إلى الثوابت، و
 المتغيرات والأنواع المعرفة بواسطة المستخدم . ويمكن استخدام  هذه الإجراءات والدوال الموجودية  في

 هذه الوحدات ونقلها لعدة  برامج مما يحقق إحدى أهداف البرمجة الهيكلية وهي إعادية  الإستفادية  من
 الكودي. كذلك فهي تقلل من إزاديحام  البرنامج الرئيسي حيث يتم وضع الإجراءات والدوال التي تلبي حل

مشكلة معينة في وحدة  خاصة حتى تسهل صيانة وقراءة  الكودي.

 فنجد شكل وحدة  جديدة  فيها هذا الكودي:File/New Unitلإنشاء وحدة  جديدة  ماعلينا إلى الذهاب إلى 

unit Unit1;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils; 

implementation

end.

ًل إذا اخترنا اسم    للوحدة  بدلً منTestبعد ذلك نقوم  بإعطائها إسم وحفظها كملف بنفس الإسم، مث
Unit1 لبد من حفظها كملف بإسم Test.pasبعد ذلك نقوم  بكتابة الإجراءات والدوال التي نريد . 

ًا: استخدامها لحق

unit Test;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils; 

const
  GallonPrice = 6.5;

  function GetKilometers(Payment, Consumption: Integer): Single;

implementation

function GetKilometers(Payment, Consumption: Integer): Single;
begin
  Result:= (Payment / GallonPrice) * Consumption;
end;

end.



 ليتم استخدامهما في أي برنامج.GetKilometers وديالة GallonPriceقمنا بكتابة ثابت 
  أما كودي الدالة فهو مكتوب في قسمInterfaceنلحظ أننا قمنا بإعادية  كتابة رأس الدالة في قسم الـ 

Implementation وضرورة  إعادية  كتابة رأس الدالة أو الإجراء في قسم الـ .interfaceّكن باقي البرامج ُيم   
 من رؤية هذه الدالة ومن ثم منادياتها، فالدوال الغير مكتوبة في هذا القسم ليمكن ندائها من خارج

الوحدة ، بل يمكن فقط ندائها دياخل هذه الوحدة .

قمنا بحفظ هذا الملف في نفس المجلد الذي يوجد فيه البرنامج التالي:

program PetrolConsumption;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils
  { you can add units after this }, Test;

var
  Payment: Integer;
  Consumption: Integer;
  Kilos: Single;
begin
  Write('How much did you pay for your car''s petrol: ');
  Readln(Payment);
  Write('What is the consumption of your car (Kilos per one Gallon) ');
  Readln(Consumption);

  Kilos:= GetKilometers(Payment , Consumption);

  Writeln('This petrol will keep your car running for: ',
    Format('%0.1f', [Kilos]), ' Kilometers');
  Write('Press enter');
  Readln;
end.

  في هذاGetKilometers وقمنا بنداء الدالة Uses في قسم الـ Testنلحظ أننا قمنا بإضافة إسم الوحدة  
البرنامج.

لإستخدام  وحدةٍ  ما في برنامج يجب تحقق واحد من هذه الشروط:

يكون ملف الوحدة  موجودي في نفس المجلد الذي يوجد فيه البرنامج.1
Project/Add Editor File to projectفتح  ملف الوحدة  المعينة ثم إضافتها للمشروع بواسطة .2
Project/Compiler Options/Paths/Other Unit Filesإضافة مسار أو مجلد الوحدة  في .3



الوحدات وبنية لزااراس وفري باسكال

 نجد أن الوحدات في مترجم فري باسكال وأدياة  التطوير لزااراس تمثل البنية الأساسية للغة البرمجة،
 حيث نجدهما يحتويان على عددي كبير من الوحدات التي تمثل مكتبة لغة أوبجكت باسكال والتي بدورها

ًا من الدوال والإجراءات والأنواع وغريرها الكثير من الأشياء التي جعلت لغة  تحتوي على عددي كبير جد
 أوبجكت باسكال لغة عنية ثرية بكافة الأديوات التي يحتاجها المبرمج لبناء النظم والبرامج الكبيرة  بسرعة
 وبدون جهد كبير. والسبب في ذلك أن هذه اللغة أخذت وقت كافي من التطوير والإستخدام  وقد صاحب
 هذا الإستخدام  بناء وحدات تمثل الخبرة  التي قام  بجمعها المبرمجين من كافة أرجاء العالم ليصبوها في

شكل مكتبات مُجربة يمكن الإستفادية  منها في معظم البرامج.
 وغريرها.Classes, SysUtilsهذه الوحدات التي يحتويها لزااراس هي وحدات عامة مثل 

الوحدات التي يكتبها المبرمج

  أما الوحدات التي يحتاج أن يكتبها المبرمج فهي تعتبر وحدات خاصة تلبي حاجة برنامجه الذي يقوم 
ًل اذا افترضنا أن المبرمج يقوم  بكتابة برنامج لإديارة  صيانة  بتصميمه وكتابته أو البرامج المشابة، مث

 السيارات، فإن الوحدات التي يكتبها يمكن أن تحتوي على إجراءات خاصة بالسيارات وبرنامج الصيانة،
مثل إجراء تسجيل سيارة  جديدة ، أو البحث عن سيارة  بدللة رقم اللوحة، أو إضافة سجل صيانة، إلخ.

 ففي هذه الحالة يقوم  المبرمج بكتابة وحدة  أو وحدات تحتوي على ديوال وإجراءات تمكنه من بناء قاعدة 
ًل إذا قام  المبرمج بكتابة  من الأديوات التي تسهل له بناء البرنامج الرئيسي بسرعة وبوضوح. فمث

 إجراءات تمكنه من القيام  بكافة مطلبات البرنامج، فإن كتابة البرنامج الرئيسي يكون عبارة  عن نداء
ًا.  ًا وأقل تعقيد لتلك الدوال والإجراءات بطريقة تجعل البرنامج الرئيسي أكثر وضوح

 كذلك يمكن لعددي من المبرمجين أن ينفرديوا بكتابة وحدات تمثل جزئية معينة من النظام ، وعند تجميعها
 نكون قد تحصلنا على برنامج كبير قام  بكتابته عددي من المبرمجين ديون أن يكون هناك  تعارض في

أجزائه.



وحدة  التاريخ الهجري

 التاريخ الهجري له أهمية كبيرة  للمسملين حيث أن كل المعاملت الشرعية المرتبطة بتاريخ أو زامن
ًل زاكاة  المال مرتبطة بتمام  الحول  معين فإن هذا الزمن أو الفترة  تكون محسوبة بالتاريخ الهجري، مث

وهو عام  هجري.
  شهر، كانت تستخدم  من زامن الجاهلية،12التاريخ الهجري مبني على السنة القمرية وهي تحتوي على 

 لكن عمر بن الخطاب رضى الله عنه اعتمد هجرة  الرسول صلى الله عليه وسلم إلى المدينة هي بداية
ًا له، وهو يوم    ميلديية.622 يوليو 16لعهد جديد يبدأ المسلمون بحساب السنوات إستنادي

 يوم .29.530588 يوم ، والشهر القمري فيه 354.367056السنة القمرية فيها 
ًء على هذه الثوابت. ويمكن أن يكون  وقد قمت بكتابة وحدة  للتحويل بين السنة الهجرية والميلديية بنا

ًا. ًة  أو نقصان الخطأ في التحويل هو يوم  واحد فقط زايادي
وكودي الوحدة  هو:

{
********************************************************************************

  HejriUtils:   Hejri Calnder converter, for FreePascal and Delphi
  Author:       Motaz Abdel Azeem
  email:        motaz@code.sd
  Home page:    http://motaz.freevar.com/
  License:      LGPL
  Created on:   26.Sept.2009
  Last modifie: 26.Sept.2009

*******************************************************************************
}

unit HejriUtils;

{$IFDEF FPC}
{$mode objfpc}{$H+}
{$ENDIF}

interface

uses
  Classes, SysUtils; 

const
  HejriMonthsEn: array [1 .. 12] of string = ('Moharram', 'Safar', 'Rabie Awal',
    'Rabie Thani', 'Jumada Awal', 'Jumada Thani', 'Rajab', 'Shaban', 'Ramadan',
    'Shawal', 'Thi­Alqaida', 'Thi­Elhajah');

  HejriMonthsAr: array [1 .. 12] of string = (' 'ربيع أول, 'صفر', محرم'  ',
    ' 'رمضان, 'شعبان', 'رجب', 'جمادى الآخر', 'جمادى الأول', ربيع ثاني' ',
    ' 'ذي الحجة,'ذي القعدة', شوال' ');

  HejriYearDays = 354.367056;
  HejriMonthDays = 29.530588;

  procedure DateToHejri(ADateTime: TDateTime; var Year, Month, Day: Word);
  function HejriToDate(Year, Month, Day: Word): TDateTime;



  procedure HejriDifference(Year1, Month1, Day1, Year2, Month2, Day2: Word; 
    var YearD, MonthD, DayD: Word);

implementation

var
  HejriStart : TDateTime;

procedure DateToHejri(ADateTime: TDateTime; var Year, Month, Day: Word);
var
  HejriY: Double;
  Days: Double;
  HejriMonth: Double;
  RDay: Double;
begin
  HejriY:= ((Trunc(ADateTime) ­ HejriStart ­ 1) / HejriYearDays);
  Days:= Frac(HejriY);
  Year:= Trunc(HejriY) + 1;
  HejriMonth:= ((Days * HejriYearDays) / HejriMonthDays);
  Month:= Trunc(HejriMonth) + 1;
  RDay:= (Frac(HejriMonth) * HejriMonthDays) + 1;
  Day:= Trunc(RDay);
end;

function HejriToDate(Year, Month, Day: Word): TDateTime;
begin
  Result:= (Year ­ 1) * HejriYearDays + (HejriStart ­ 0) + 
    (Month ­ 1) * HejriMonthDays + Day + 1;
end;

procedure HejriDifference(Year1, Month1, Day1, Year2, Month2, Day2: Word; var    
  YearD, MonthD, DayD: Word);
var
  Days1: Double;
  Days2: Double;
  DDays: Double;
  RYear, RMonth: Double;
begin
  Days1:= Year1 * HejriYearDays + (Month1 ­ 1) * HejriMonthDays + Day1 ­ 1;
  Days2:= Year2 * HejriYearDays + (Month2 ­ 1) * HejriMonthDays + Day2 ­ 1;
  DDays:= Abs(Days2 ­ Days1);

  RYear:= DDays / HejriYearDays;
  RMonth:= (Frac(RYear) * HejriYearDays) / HejriMonthDays;
  DayD:= Round(Frac(RMonth) * HejriMonthDays);

  YearD:= Trunc(RYear);
  MonthD:= Trunc(RMonth);

end;

initialization

  HejriStart:= EncodeDate(622, 7, 16);

end.



وتحتوي هذه الوحدة  على الدوال والإجراءات التالية:

1.DateToHejri:ًل : وتستخدم  لتحويل تاريخ ميلديي لما يقابله من التاريخ الهجري، مث

program Project1;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes  , HejriUtils, SysUtils
  { you can add units after this };

var
  Year, Month, Day: Word;
begin
  DateToHejri(Now, Year, Month, Day);
  Writeln('Today in Hejri: ', Day, '­', HejriMonthsEn[Month], 
    '­', Year);
  Readln;
end.

        
2.HejriToDate.ويستخدم  للتحويل من تاريخ هجري لمايقابله بالميلديي :
3.HejriDifference.ويستخدم  لمعرفة الفارق الزمني بين تاريخين هجريين :

 يوجد مثال آخر للتحويل بين التاريخ الهجري والميلديي في فصل الواجهة الرسومية، واسم المثال هو
ّول الهجري. المح



 Procedure and functionتحميل الإجراءات والدوال 
Overloading

 المقصودي بتحميل الإجراءات والدوال هو أن تكون هناك  أكثر من ديالة أو إجراء مشترك  في الإسم لكن
ًل لو أردينا أن نكتب ديالة تقوم  بجمع عدديين صحيحين أو  مع تغيير المدخلت من حيث العددي أو النوع. مث

  مع أي من نوع المتغيرات السابقة يقوم  البرنامجSumحقيقيين، بحيث عندما يقوم  المبرمج بنداء الدالة 
بإختيار الدالة المناسبة:

program sum;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

function Sum(x, y: Integer): Integer; overload;
begin
  Result:= x + y;
end;

function Sum(x, y: Double): Double; overload;
begin
  Result:= x + y;
end;

var
  j, i: Integer;
  h, g: Double;
begin
  j:= 15;
  i:= 20;
  Writeln(J, ' + ', I, ' = ', Sum(j, i));

  h:= 2.5;
  g:= 7.12;
  Writeln(H, ' + ', G, ' = ', Sum(h, g));

  Write('Press enter key to close');
  Readln;

end.

  لكتابة ديالتين بنفس الإسم مع إختلف نوع المتغيرoverloadنلحظ أننا قمنا بكتابة الكلمة المحجوزاة  
وناتج الدالة.



default parametersالقيم الإفتراضية للمدخلت 

 يمكن أن نضع قيم إفتراضية في مدخلت الدوال أو الإجراءات. وفي هذه الحالة يمكن تجاهل هذه
المدخلت عند النداء أو إستخدامها كما في المثال التالي:

program defaultparam;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

function SumAll(a, b: Integer; c: Integer = 0; 
  d: Integer = 0): Integer;
begin
  result:= a + b + c + d;
end;

begin
  Writeln(SumAll(1, 2));
  Writeln(SumAll(1, 2, 3));
  Writeln(SumAll(1, 2, 3, 4));
  Write('Press enter key to close');
  Readln;
end.

 نجد أن كل النداءات السابقة صحيحة، وفي حالة عدم  إرسال متغير معين يتم إسنادي القيمة الإفتراضية
.d و cله، وفي هذه الحالة القيمة صفر للمتغيرات 

ّيد بالترتيب، حيث يمكن تجاهل   c فقط، أو تجاهل dوفي حالة تجاهل المتغيرات وعدم  إرسالها، يجب التق
ًاd لبد من تجاهل c. فإذا تجاهلنا إرسال قيمة d وإرسال c، لكن ليمكن تجاهل المغير أو المُدخل dو    تبع

 لذلك. لذلك على المبرمج أن يضع المدخلت الغير مهمة في نهاية تعريف الدالة والأكثر أهمية في
.dبدايتها. أي أن الأقل أهمية في المثال السابق كان المتغير 



sortingترتيب البيانات 

 ، وقد قمنا بذكره في هذاData Structureترتيب البيانات هو موضوع يقع ضمن مفهوم  هيكلية البيانات 
الفصل كمثال للبرامج التي تعتمد البرمجة الهيكلية والإجراءات لتحقيق ترتيب البيانات.

ًل إذا افترضنا أن لدينا مصفوفة فيها أرقام ، ونريد ترتيب هذه  ترتيب البيانات نحتاجه مع المصفوفات، فمث
ًا، في هذه الحالة يمكننا استخدام  عدة  طرق لتحقيق هذا الهدف. ًا أو تنازالي الأرقام  تصاعدي

bubble sortخوارزامية ترتيب الفُقّاعة 

 وهي من أسهل طرق الترتيب، حيث يقوم  إجراء الترتيب بمقارنة العُنصر الأول في المصفوفة مع
 العنصر الثاني، ففي حالة الترتيب التصاعدي ،إذا وُجد أن العنصر الأول ذو قيمة أكبر من العنصر الثاني،

 قام  البرنامج بتبديل العناصر، ثم يقوم  بمقارنة العنصر الثاني مع الثالث إلى نهاية المصفوفة. ثم يتم
تكرار هذه العملية عدة  مرات حتى يتم الترتيب الكامل للمصفوفة.

 عناصر في المصفوفة تم إديخالها بهذا الشكل:6إذا افترضنا أن لدينا 

7
10
2
5
6
3

 نجد أن الإجراء يحتاج لأكثر من ديورة  لإتمام  العملية. حيث أن في كل ديورة  تتم مقارنة العنصر الأول مع
 الثاني وإبدالهما إذا كان الأول أكبر من الثاني، ثم الثاني مع الثالث، والثالث مع الرابع، ثم الرابع مع

 الخامس، ثم الخامس مع الساديس. فإذا لم يتم الترتيب الكامل فسوف نحتاج لدورة  أخرى، ثم ثالثة
ورابعة إلى أن يتم الترتيب الكامل.

في الدورة  الأولى تصبح  المتغيرات في المصفوفة كالتي:

7
2
5
6
3
10

وفي الدورة  الثانية:

2
5
6
3
7
10



وفي الدورة  الثالثة:

2
5
3
6
7
10

وفي الدورة  الرابعة:

2
3
5
6
7
10

 نلحظ أنه بنهاية الدورة  الرابعة تم الترتيب الكامل للرقام ، لكن ليس من السهولة معرفة أن الترتيب
 قد تم إل بإضافة ديورة  خامسة، وفي هذه الدورة  نجد أن الإجراء لم يقم بإي تبديل، وبهذه الطريقة

نعرف أن البيانات قد تم ترتيبها.
ّقاعي بسبب أن القيم الأقل تطفو للعلى في كل ديورة . ُف وجاء إسم الترتيب ال

كودي برنامج ترتيب الفقاعة:

program BubbleSortProj;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes;

procedure BubbleSort(var X: array of Integer);
var
  Temp: Integer;
  i: Integer;
  Changed: Boolean;
begin
  repeat  // Outer loop
    Changed:= False;
    for i:= 0 to High(X) ­ 1 do // Inner loop
      if X[i] > X[i + 1] then
      begin
        Temp:= X[i];
        X[i]:= X[i + 1];
        X[i + 1]:= Temp;
        Changed:= True;
      end;
  until not Changed;
end;



var
  Numbers: array [0 .. 9] of Integer;
  i: Integer;
begin
  Writeln('Please input 10 random numbers');
  for i:= 0 to High(Numbers) do
  begin
    Write('#', i + 1, ': ');
    Readln(Numbers[i]);
  end;

  BubbleSort(Numbers);
  Writeln;
  Writeln('Numbers after sort: ');

  for i:= 0 to High(Numbers) do
  begin
    Writeln(Numbers[i]);
  end;
  Write('Press enter key to close');
  Readln;
end.

ًل ثم الأصغر) وذلك بتحويل معامل المقارنة  يمكن تحويل الترتيب التنازالي إلى ترتيب تصاعدي (الأكبر أو
 كالتالي:أصغر من إلى أكبر منمن 

   if X[i] < X[i + 1] then

وفي المثال التالي نقوم  بالترتيب التصاعدي لدرجات طلب لنعرف الأول ثم الثاني إلى الأخير.
وفي هذا المثال استخدمنا مصفوفة سجلت لتسجيل إسم الطالب وديرجته:

الطلب    : درجات ترتيب الطلب    :برنامج درجات ترتيب برنامج
program smSort;   // Stuents mark sort

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

type
  TStudent = record
    StudentName: string;
    Mark: Byte;
  end;



procedure BubbleSort(var X: array of TStudent);
var
  Temp: TStudent;
  i: Integer;
  Changed: Boolean;
begin
  repeat
    Changed:= False;
    for i:= 0 to High(X) ­ 1 do
      if X[i].Mark < X[i + 1].Mark then
      begin
        Temp:= X[i];
        X[i]:= X[i + 1];
        X[i + 1]:= Temp;
        Changed:= True;
      end;
  until not Changed;
end;

var
  Students: array [0 .. 9] of TStudent;
  i: Integer;

begin
  Writeln('Please input 10 student names and marks');
  for i:= 0 to High(Students) do
  begin
    Write('Student #', i + 1, ' name : ');
    Readln(Students[i].StudentName);

    Write('Student #', i + 1, ' mark : ');
    Readln(Students[i].Mark);
    Writeln;
  end;

  BubbleSort(Students);
  Writeln;
  Writeln('Marks after Buble sort: ');
  Writeln('­­­­­­­­­­­­­­­­­­­­­­');

  for i:= 0 to High(Students) do
  begin
    Writeln('# ', i + 1, '  ', Students[i].StudentName, 
      ' with mark (', Students[i].Mark, ')');
  end;
  Write('Press enter key to close');
  Readln;
end.

 طريقة ترتيب الفقاعة تمتازا بالبساطة، حيث أن المبرمج يمكن أن يحفظها ويكتب الكودي كل مرة  بدون
 الرجوع لأي مرجع. وهي مناسبة للمصفوفات الصغيرة  والشبه مرتبة، أما إذا كانت المصفوفة تحتوي
ًا ففي هذه الحالة سوف يستغرق هذا ًل) وكانت عشوائية تمام  على عناصر كثيرة  (عشرات اللف مث

ًا ويجب اللجوء إلى نوع آخر من الترتيب كما في الأمثلة اللحقة. النوع من الترتيب وقت

  وعددي لفاتهاrepeat untilنلحظ أن خوارزامية ترتيب الفقاعة تعتمد على ديورتين، ديورة  خارجية تستخدم  
 غرير محددية ، معتمدة  على شكل البيانات، فإذا كانت البيانات مرتبة فإن عددي اللفات تكون مرة  واحدة 



ًا فإن عددي اللفات تكون بعددي عناصر المصفوفة. والدورة  الداخلية  فقط، أما إذا كانت عشوائية تمام
ًا بعددي عناصر المصفوفة ناقص واحد.for loopبإستخدام    تكون لفاتها ديائم

  عنصر في المصفوفة وكانت المصفوفة مُرتبة فإن عددي اللفات الكلي هو1000فإذا إفترضنا أن لدينا 
 لفة. 999، أي الحصيلة 999لفة واحدة  خارجية مضروبة في 

ًا فإن عددي اللفات الخارجية تكون    لفة999 مضروبة في 1000وإذا كانت نفس المصفوفة عشوائية تمام
  لفة. فكلما زاادي عددي عناصر المصفوفة وزااديت عشوائيتها فإن كفاءة 999000دياخلية، فتكون الحصيلة 

هذا النوع من الترتيب يسوء.

 Selection Sortخوارزامية الترتيب الإختياري 

ّقاعة، إل أنها أسرع مع البيانات الكبيرة ، حيث أنها تحتوي على ديورتين، ُف  هذه الطريقة أشبه بطريقة ال
  في المثال السابق، وهي تدور بعددي عناصر المصفوفة ناقصrepeat untilديورة  خارجية مشابهة لدورة  

واحد، أما الدورة  الداخلية فتقل في كل ديورة  بمقدار واحد، إلى أن تصل إلى ديورتين.

program SelectionSort;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes
  { you can add units after this };

procedure SelectSort(var X: array of Integer);
var
  i: Integer;
  j: Integer;
  SmallPos: Integer;
  Smallest: Integer;
begin
  for i:= 0 to High(X) ­1 do // Outer loop
  begin
    SmallPos:= i;
    Smallest:= X[SmallPos];
    for j:= i + 1 to High(X) do // Inner loop
      if X[j] < Smallest then
      begin
        SmallPos:= j;
        Smallest:= X[SmallPos];
      end;
      X[SmallPos]:= X[i];
      X[i]:= Smallest;
  end;
end;



// Main application

var
  Numbers: array [0 .. 9] of Integer;
  i: Integer;

begin
  Writeln('Please input 10 random numbers');
  for i:= 0 to High(Numbers) do
  begin
    Write('#', i + 1, ': ');
    Readln(Numbers[i]);
  end;

  SelectSort(Numbers);
  Writeln;
  Writeln('Numbers after Selection sort: ');

  for i:= 0 to High(Numbers) do
  begin
    Writeln(Numbers[i]);
  end;
  Write('Press enter key to close');
  Readln;
end.

 بالرغرم من أن خوارزامية الترتيب الإختياري أسرع في حالة البيانات الكبيرة ، إل أنها لترتبط بمدى ترتيب
ًا تبديل لنفس  أو عشوائية المصفوفة، ففي كل الحالت نجد أن سرعتها شبه ثاتبة، حيث يحدث أحيان

العنصر مع نفسه بعد فراغ الحلقة الداخلية.

Shellخوارزامية الترتيب 

ّقاعة في ُف  تمتازا هذه الخوارزامية بالسرعة العالية مع البيانات الكبيرة ، وسلوكها مشابهة لخوارزامية ال
ًا من الخوارزاميتين السابقتين. حالة البيانات المرتبة أو الشبه مرتبة، إل أنها أكثر تعقيد

.Donald Shellوسميت بهذا الإسم نسبة لمخترعها 

program ShellSort;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes;

procedure ShellS(var X: array of Integer);
var



  Done: Boolean;
  Jump, j, i: Integer;
  Temp: Integer;
begin
  Jump:= High(X);
  while (Jump > 0) do // Outer loop
  begin
    Jump:= Jump div 2;
    repeat         // Intermediate loop
      Done:= True;
      for j:= 0 to High(X) ­ Jump do // Inner loop
      begin
        i:= j + Jump;
        if X[j] > X[i] then // Swap
        begin
          Temp:= X[i];
          X[i]:= X[j];
          X[j]:= Temp;
          Done:= False;
        end;

      end; // end of inner loop
    until Done; // end of intermediate loop
  end; // end of outer loop 
end;

var
  Numbers: array [0 .. 9] of Integer;
  i: Integer;

begin
  Writeln('Please input 10 random numbers');
  for i:= 0 to High(Numbers) do
  begin
    Write('#', i + 1, ': ');
    Readln(Numbers[i]);
  end;

  ShellS(Numbers);
  Writeln;
  Writeln('Numbers after Shell sort: ');

  for i:= 0 to High(Numbers) do
  begin
    Writeln(Numbers[i]);
  end;
  Write('Press enter key to close');
  Readln;
end.



ترتيب المقاطع

ًل  يمكن ترتيب المقاطع بنفس طريقة ترتيب الأسماء، حيث أن المقاطع يتم ترتيبها حسب الأحرف، فمث
 65 في الذاكرة  هو عبارة  عن الرقم A حيث أن قيمة الحرف .B قيمته أقل من الحرف Aالحرف 

 . في المثال التالي أعدنا كتابة برنامج ترتيب ديرجات الطلب، لكن هذه المرة  يتم66 قيمته Bوالحرف 
ًا حسب الإسم وليس الدرجة: الترتيب هجائي

بالأسماء    الطلب ترتيب بالأسماء   برنامج الطلب ترتيب برنامج

program smSort;   // Stuents mark sort by name

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes;

type
  TStudent = record
    StudentName: string;
    Mark: Byte;
  end;

procedure BubbleSort(var X: array of TStudent);
var
  Temp: TStudent;
  i: Integer;
  Changed: Boolean;
begin
  repeat
    Changed:= False;
    for i:= 0 to High(X) ­ 1 do
      if X[i].StudentName > X[i + 1].StudentName then
      begin
        Temp:= X[i];
        X[i]:= X[i + 1];
        X[i + 1]:= Temp;
        Changed:= True;
      end;
  until not Changed;
end;

var
  Students: array [0 .. 9] of TStudent;
  i: Integer;

begin
  Writeln('Please input 10 student names and marks');
  for i:= 0 to High(Students) do



  begin
    Write('Student #', i + 1, ' name : ');
    Readln(Students[i].StudentName);

    Write('Student #', i + 1, ' mark : ');
    Readln(Students[i].Mark);
    Writeln;
  end;

  BubbleSort(Students);
  Writeln;
  Writeln('Marks after Bubble sort: ');
  Writeln('­­­­­­­­­­­­­­­­­­­­­­');

  for i:= 0 to High(Students) do
  begin
    Writeln('# ', i + 1, '  ', Students[i].StudentName, 
     ' with mark (', Students[i].Mark, ')');
  end;
  Write('Press enterkey to close');
  Readln;
end.

مقارنة خوارزاميات الترتيب

 في هذا البرنامج سوف نقوم  بإديخال أرقام  عشوائية في مصفوفة كبيرة  ونجري عليها أنواع الترتيب
الثلث السابقة ونقارن الزمن الذي تأخذه كل خوارزامية:

program SortComparison;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Classes, SysUtils;

//  Bubble sort

procedure BubbleSort(var X: array of Integer);
var
  Temp: Integer;
  i: Integer;
  Changed: Boolean;
begin
  repeat
    Changed:= False;
    for i:= 0 to High(X) ­ 1 do
      if X[i] > X[i + 1] then
      begin
        Temp:= X[i];



        X[i]:= X[i + 1];
        X[i + 1]:= Temp;
        Changed:= True;
      end;
  until not Changed;
end;

// Selection Sort

procedure SelectionSort(var X: array of Integer);
var
  i: Integer;
  j: Integer;
  SmallPos: Integer;
  Smallest: Integer;
begin
  for i:= 0 to High(X) ­1 do // Outer loop
  begin
    SmallPos:= i;
    Smallest:= X[SmallPos];
    for j:= i + 1 to High(X) do // Inner loop
      if X[j] < Smallest then
      begin
        SmallPos:= j;
        Smallest:= X[SmallPos];
      end;
      X[SmallPos]:= X[i];
      X[i]:= Smallest;
  end;
end;

// Shell Sort

procedure ShellSort(var X: array of Integer);
var
  Done: Boolean;
  Jump, j, i: Integer;
  Temp: Integer;
begin
  Jump:= High(X);
  while (Jump > 0) do // Outer loop
  begin
    Jump:= Jump div 2;
    repeat         // Intermediate loop
      Done:= True;
      for j:= 0 to High(X) ­ Jump do // Inner loop
      begin
        i:= j + Jump;
        if X[j] > X[i] then // Swap
        begin
          Temp:= X[i];
          X[i]:= X[j];
          X[j]:= Temp;
          Done:= False;
        end;



      end; // inner loop
    until Done; // innermediate loop end
  end; // outer loop end
end;

// Randomize Data

procedure RandomizeData(var X: array of Integer);
var
  i: Integer;
begin
  Randomize;
  for i:= 0 to High(X) do
    X[i]:= Random(10000000);
end;

var
  Numbers: array [0 .. 59999] of Integer;
  i: Integer;
  StartTime: TDateTime;

begin

  Writeln('­­­­­­­­­­­­­­­­­ Full random data');
  RandomizeData(Numbers);
  StartTime:= Now;
  Writeln('Sorting.. Bubble');
  BubbleSort(Numbers);
  Writeln('Bubble sort tooks (mm:ss): ', 
    FormatDateTime('nn:ss', Now ­ StartTime));
  Writeln;

  RandomizeData(Numbers);
  Writeln('Sorting.. Selection');
  StartTime:= Now;
  SelectionSort(Numbers);
  Writeln('Selection sort tooks (mm:ss): ', 
    FormatDateTime('nn:ss', Now ­ StartTime));
  Writeln;

  RandomizeData(Numbers);
  Writeln('Sorting.. Shell');
  StartTime:= Now;
  ShellSort(Numbers);
  Writeln('Shell sort tooks (mm:ss): ', 
   FormatDateTime('nn:ss', Now ­ StartTime));

  Writeln;
  Writeln('­­­­­­­­­­­­­­­­­ Nearly sorted data');
  Numbers[0]:= Random(10000);
  Numbers[High(Numbers)]:= Random(10000);
  StartTime:= Now;
  Writeln('Sorting.. Bubble');
  BubbleSort(Numbers);
  Writeln('Bubble sort tooks (mm:ss): ', 
    FormatDateTime('nn:ss', Now ­ StartTime));
  Writeln;



  Numbers[0]:= Random(10000);
  Numbers[High(Numbers)]:= Random(10000);
  Writeln('Sorting.. Selection');
  StartTime:= Now;
  SelectionSort(Numbers);
  Writeln('Selection sort tooks (mm:ss): ', 
    FormatDateTime('nn:ss', Now ­ StartTime));
  Writeln;

  Numbers[0]:= Random(10000);
  Numbers[High(Numbers)]:= Random(10000);
  Writeln('Sorting.. Shell');
  StartTime:= Now;
  ShellSort(Numbers);
  Writeln('Shell sort tooks (mm:ss): ', 
    FormatDateTime('nn:ss', Now ­ StartTime));

  Write('Press enterkey to close');
  Readln;
end.



الثالث  الثالث الفصل الفصل

الرسومية  الرسومية الواجهة الواجهة
  Graphical User InterfaceGraphical User Interface



مقدمة

 ،command lineالواجهة الرسومية أو واجهة المستخدم  الرسومية هي بديل حديث لواجهة سطر الأوامر 
ُتستخدم  فيها إمكانات جديدة  مثلtext interfaceأو الواجهة النصية   ، وهي أكثر سهولة للمستخدم ، و

 الماوس، الأزارار، مربعات النصوص، وإلى ماذلك مما اعتاديه المستخدم  الن. وكمثال للبرامج ذات
  ، الألعابOfficeالواجهة الرسومية في نظام  لينكس: متصفح  الإنترنت فيرفوكس، برامج المكتب 

 وغريرها. فبرامج الواجهة الرسومية أصبحت تشكل البنية الأساسية في كل أنظمة الحاسوب، مثل
 الأنظمة المحاسبية ، وبرامج التصميم والألعاب التي تحتوي على مكتبات متقدمة ومعقدة  للواجهة

  هي عبارة  عن مجموعةKDE و Gnomeالرسومية،كذلك فإن أجزاء أنظمة التشغيل مثل سطح  المكتب 
من برامج ذات واجهات رسومية.

 لكتابة برامج ذات واجهة رسومية يجب استخدام  احدى المكتبات المشهورة  المتوفرة  في هذا المجال
مثل:

GTK+ library    المستخدمة في نظام  لينكس وبعض واجهات الهواتف النقالة 
QT library المستخدمة في نظام  لينكس ووندوزا وماكنتوش
Win32/64 GDI المستخدمة في نظام  وندوزا

 أو يمكن استخدام  أدياة  تطوير متكاملة تستخدم  احدى هذه المكتبات، مثل الزااراس. حيث نجد أن
  في نظام  ويندوزا. وليحتاج المبرمج أنWin32/64 GDI في نظام  لينكس، و GTKاللزااراس تسخدم  

 يدخل في تفاصيل هذه المكتبات حيث أن لزااراس توفر طريقة ذات شفافية عالية في تصميم البرامج
 الرسومية، فماعلى المبرمج إل تصميم هذه البرامج في احدى بيئات التشغيل ثم إعادية  ترجمتها وربطها

 في بيئات التشغيل الأخرى ليحصل على نفس النسخة من البرنامج الأصلي يعمل في كافة أنظمة
ًا يحتاج المبرمج أن يقوم  بإنزال مكتبة  ًل في حال فشل البرنامج ذو الواجهةGTKالتشغيل. لكن احيان   مث

ًل. الرسومية من أن يعمل في حال عدم  توفر هذه المكتبة مث



البرنامج ذو الواجهة الرسومية الأول

لعمل برنامج جديد ذو واجهة رسومية علينا إختيار :

Project/New Project/Application

بعد ذلك نقوم  بحفظه المشروع بواسطة:

File/Save  All

ًل   لنقوم  فيه بحفظ المشروع بكل متعلقاته.firstguiثم نقوم  بإنشاء مجلد جديد مث
main.pasبعد ذلك نقوم  بحفظ الوحدة  الرئيسية في المجلد السابق، ونسميه 

firstgui.lpiثم نقوم  بحفظ ملف المشروع في نفس المجلد ونسميه 

 ليظهر لنا الفورم  التالي:F12في الوحدة  الرئيسية يمكننا الضغط على المفتاح 

إذا قمنا بتشغيل البرنامج سوف نحصل على الشكل التالي:



بعد ذلك نقوم  بإغرلق البرنامج ونرجع لحالة التصميم.
 من صفحة المكونات التالية:Buttonثم نقوم  بإضافة زار 

نقوم  بوضع هذا الزر في وسط الفورم  السابق كالتالي:

 object. فإذا لم تكن الـ Object Inspectorبعد ذلك نقوم  نذهب إلى صفحة الخصائص الموجودية  في الـ 
Inspector موجودية  يمكن إظهارها عن طريق Window/Object Inspectorفي القائمة الرئيسية، أو  

 لتظهر لنا الخصائص التالية:F11الوقوف على الفورم  أو المكون ثم الضغط على المفتاح 



 Objectلبد من التأكد من أننا نؤشر على الزر، ويمكن معرفة ذلك بقراءة  القيمة المظللة في الـ 
Inspector:والتي يجب أن تحتوي على 

Button1: TButton

.Click إلى كلمة Button1 من Caption نقوم  بتغيير قيمة الـ Propertiesفي صفحة الخصائص 
  ونقوم  بالنقر المزديوج علىObject Inspector في نفس الـ Eventsبعد ذلك نذهب إلى صفحة الأحداث 

 فنحصل على الكودي التالي:OnClickقيمة الحدث 

procedure TForm1.Button1Click(Sender: TObject);
begin

end; 

 ونقوم  بتشغيل البرنامج:begin endثم نقوم  بكتابة السطر التالي بين 

procedure TForm1.Button1Click(Sender: TObject);
begin
  ShowMessage('Hello world, this is my first GUI application');
end;

 فنحصل على الرسالة أعله.Clickبعد تنفيذ البرنامج نقوم  بالضغط على الزر المكتوب عليه 
بهذا نكون قد كتبنا أول برنامج ذو واجهة رسومية.

  إذا كنا في بيئةfirstgui.exe إذا كنا في لينكس أو firstgui سوف نجد الملف التنفيذي firstguiفي الدليل 



وندوزا، وهي ملفات تنفيذية يمكن نقلها وتشغيلها في أي حاسوب آخر به نفس نظام  التشغيل.

في البرنامج السابق نريد الوقوف على عددي من النقاط المهمة وهي:

 ، وهو الملف الرئيسيfirstgui.lpi: وهو الملف الذي قمنا بحفظه بالإسم: البرنامج الرئيسي.1
الذي يمثل البرنامج. ويمكن فتحه عن طريق:

Project/Source

ًا:firstgui ثم اختيار CTRL-F12أو بالضغط على  . وسوف نجد هذا الكودي الذي قام  لزااراس بكتابته تلقائي

program firstgui;

{$mode objfpc}{$H+}

uses
  {$IFDEF UNIX}{$IFDEF UseCThreads}
  cthreads,
  {$ENDIF}{$ENDIF}
  Interfaces, // this includes the LCL widgetset
  Forms
  { you can add units after this }, main;

begin
  Application.Initialize;
  Application.CreateForm(TForm1, Form1);
  Application.Run;
end.

 ومن النادير أن يقوم  المبرمج بتعديل هذا الملف، فمعظم تصميم وكتابة الكودي تكون في الوحدات
ًا لهذاmainالإضافية المصاحبة لهذا المشروع. كما ونلحظ أن الوحدة  الرئيسية    قد تم إضافتها تلقائي

الملف.

 وهي تحتوي على الفورم  الرئيسي الذي يظهر عند تشغيل main unitالوحدة الرئيسية .2
 البرنامج والكودي المصاحب له. ولعمل برنامج ذو واجهة رسومية لبد من وجودي وحدة  بهذا الشكل

تحتوي على فورم .

الكودي الذي تحتويه هذه الوحدة  بعد الإضافات التي قمنا بها بعد إضافة الزر هو كالتي:

unit main;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics, Dialogs,
  StdCtrls;

type



  { TForm1 }

  TForm1 = class(TForm)
    Button1: TButton;
    procedure Button1Click(Sender: TObject);
  private
    { private declarations }
  public
    { public declarations }
  end; 

var
  Form1: TForm1; 

implementation

{ TForm1 }

procedure TForm1.Button1Click(Sender: TObject);
begin
  ShowMessage('Hello world, this is my first GUI application');
end;

initialization
  {$I main.lrs}

end.

  وهو أشبه بالسجل الذيClass على أنه من النوع TForm1نجد أنه في بداية هذه الوحدة  تم تعريف 
 ذكرناه في ديروس سابقة. وهو مايعرف بالبرمجة الكائنية والتي تعتبر ديرس متقدم  سوف نذكره في

.Button1 واسمه TButtonالفصل القاديم  إن شاء الله. ودياخل هذا الفورم  يوجد زار من النوع 
.main ثم اختيار اسمها وهي في هذا المثال CTRL-F12يمكن فتح  هذه الوحدة  بواسطة 

  وفي هذه الصفحة تظهر خصائص الكائنات:Object Inspector/Propertiesصفحة الخصائص .3
ًل نجد أن للزر بعض الخصائص التي يمكن للمبرمج تغييرها  أو المكونات مثل الأزارار والفورم . فمث

 . كذلكCaption, Top, Left, Width, Heightوتؤثر مباشرة  على شكل الزر أو بعض سلوكه مثل: 
  إلخ. وهي كأنها حقول ومتغيرات دياخل السجلColor, Visibleالفورم  له بعض الخصائص مثل: 

Form1 أو Button1 وهي في الحقيقة كائنات .Objects وليس سجلت Recordsأنما شبهناها  
بالسجلت لتسهيل الفهم.

 وتحتوي هذه الصفحة على الأحداث التي يمكن: Object Inspector/Eventsصفحة الحداث .4
ًل الزر يمكن أن يستقبل بعض الأحداث مثل:   ,OnClickاستقبالها على الكائن المعين، مث

OnMouseMove :أما الفورم  فيستقبل أحداث مثل ،OnCreate, OnClose, OnActivateومن  
ًل الحدث  ًا متى يتم نداء هذا الحدث، فمث   يتم نداءه عندما يمرOnMouseMoveأسمائها يظهر جلي

 الماوس فوق هذا الكائن. وعند النقر المزديوج على هذه الأحداث تظهر لنا صفحة الكودي التي
ًل الحدث الذي قمنا  نقوم  بكتابة ماسوف يقوم  البرنامج بتنفيذه عندما يحدث هذا الحدث، مث

باختياره عند النقر في الزر هو:



procedure TForm1.Button1Click(Sender: TObject);
begin
  ShowMessage('Hello world, this is my first GUI application');
end;

ًا عند النقرEvent Handlerبإجراء الحدث وهذا الكودي أو الإجراء يعرف    ويتم استدعاء هذا الإجراء تلقائي
في هذا الزر في حالة التشغيل.

البرنامج الثاني: برنامج إديخال الإسم

 نقوم  بإنشاء برنامج جديد بنفس الطريقة السابقة، ونقوم  بإنشاء مجلد جديد لهذا المشروع نسميه
inputform حيث نحفظ فيه الوحدة  الرئيسية بإسم main والبرنامج بإسم inputform.

ثم نقوم  بإضافة مكونات أو كائنات وهي :

2 Lables
Edit box
Button

ونقوم  بتغيير الخصائص التالية في هذه المكونات:

Form1:
Name: fmMain
Caption: Input form application

Label1:
Caption: Please enter your name

Label2:
Name: laYourName

Edit1:
Name: edName
Text:

Button1:
Caption: Click

فيصبح  لدينا الشكل التالي



 للزر نضع هذا الكودي:OnClickفي الحدث 

procedure TfmMain.Button1Click(Sender: TObject);
begin
  laYourName.Caption:= 'Hello ' + edName.Text;
end;

  ونشاهد تنفيذClickبعد ذلك نقوم  بتشغيل البرنامج وكاتبة الإسم في مربع النص، ثم الضعط على الزر 
البرنامج.

  لقراءة  ما أديخلهedName الموجودي في الكائن Textفي البرنامج السابق نلحظ أننا استخدمنا الحقل 
  لكتابة إسمlaYourName الموجودي في المكون Captionالمستخدم  من بيانات. وكذلك استخدمنا الحقل 

المستخدم .



ListBoxبرنامج الـ 

  كما في الشكل أديناه. ونقوم ListBoxنقوم  بإنشاء برنامج جديد، ونضع فيه أربع أزارار ومربع نص و 
 بالنسبة للزارار كما هو موضح ، كذلك نقوم  بتغيير أسماء هذه الإزارار إلى:Captionبتغيير الـ 

btAdd, btClear, btDelete, btClose

الكودي المصاحب لهذه الأزارار هو:

procedure TForm1.btAddClick(Sender: TObject);
begin
  ListBox1.Items.Add(Edit1.Text);
end;

procedure TForm1.btClearClick(Sender: TObject);
begin
  ListBox1.Clear;
end;

procedure TForm1.btDeleteClick(Sender: TObject);
var
  Index: Integer;
begin
  Index:= ListBox1.ItemIndex;
  if Index <> ­1 then
    ListBox1.Items.Delete(Index);
end;

procedure TForm1.btCloseClick(Sender: TObject);
begin
  Close;
end;



Text Editorبرنامج محرر النصور 

نقوم  بإنشاء برنامج جديد عليه المكونات التالية:

TMainMenuقائمة رئيسية –
ssBoth إلى ScrollBars و الـ alClient إلى Align ونقوم  بتغيير قيمة الـ TMemoمذكرة  –
–TOpenDialog من صفحة Dialogs
–TSaveDialog

  وتحتها قائمةFile ونقوم  بإضافة القائمة MainMenu1ثم نقوم  بالنقر المزديوج على القائمة الرئيسية 
Open File, Save File, Close فيها الخيارات التالية: SubMenuفرعية 

فيكون شكل البرنامج كالتي:

نقوم  بكتابة الكودي التالي:

Open Fileفي الخيار 
  if OpenDialog1.Execute then
    Memo1.Lines.LoadFromFile(OpenDialog1.FileName);

Save Fileوفي 
  if SaveDialog1.Execute then
    Memo1.Lines.SaveToFile(SaveDialog1.FileName);

Closeوفي 
  Close;

بعد تشغيل البرنامج يمكن كتابة نص ثم حفظه، أو فتح  ملف نصي موجودي على القرص.



ًل يمكن فتح  ملف ينتهي بالإمتدادي .  فهو يعتبر ملف نصي.pasمث

وهذا هو شكل البرنامج بعد التنفيذ:

برنامج الأخبار

هذه المرة  نقوم  بكتابة برنامج لتسجيل عناوين الأخبار يكون له واجهة رسومية:

gnewsقم بإنشاء برنامج جديد وسمه –
TButtonقم بإضافة زارين من نوع –
TEditقم بإضافة مربع نص –
TMemoقم بإضافة مذكرة  –

وخواصهم هي كالتالي:

Button1
Caption: Add Title

Button2
Caption: Save
Anchors: Left=False, Right=True

Edit1:
Text=



Memo1
ScrollBars: ssBoth
ReadOnly: True
Anchors: Top=True, Left=Ture, Right=True, Bottom=Ture

ويكون بالشكل التالي:

 فهو:Add Titleأما كودي

  Memo1.Lines.Insert(0,
    FormatDateTime('yyyy­mm­dd hh:nn', Now) + ': ' + Edit1.Text);

Saveوكودي  

  Memo1.Lines.SaveToFile('news.txt');

Tform OnClose eventوالكودي بالنسبة للفورم  في حالة إغرلقه: 

  Memo1.Lines.SaveToFile('news.txt');

Tform OnCreate eventوكودي الفورم  في حالة إنشائه: 

  if FileExists('news.txt') then
    Memo1.Lines.LoadFromFile('news.txt');



برنامج الفورم  الثاني

 كما يظهر لنا في معظم البرامج ذات الواجهة الرسومية فهي تحتوي على أكثر من فورم . ولعمل ذلك
في لزااراس نقوم  بالتي:

secondformإنشاء برنامج جديد نحفظه في ديليل نقوم  بتمسيته .1
  ونقوم fmMain وتسمية الفورم  الرئيسي بالإسم main.pasنقوم  بحفظ الوحدة  الرئيسية بإسم .2

secondform.lpiبتسمية المشروع بإسم 
  ونسميsecond.pas ونقوم  بحفظ الوحدة  بإسم File/ New Formنقوم  بإضافة فورم  آخر بواسطة .3

fmSecondالفورم  
  بخط كبير. يمكن تكبير الخط في خاصية الـSecond Form نتكب فيه Labelنقوم  بإضافة .4

Font.Size الموجودية  في الـ Label
 ونقوم  بوضع زار فيه.mainنرجع للفورم  الرئيسي .5
:implementation تحت قسم mainنقوم  بكتابة هذا السطر في كودي الوحدة  الرئيسية .6

uses second;

  نقوم  بكتابة الكوديfmMain بالنسبة للزر الموجودي في الفورم  الرئيسي OnClickفي الحدث .7
التالي:

  fmSecond.Show;

نقوم  بتشغيل البرنامج ونضغط الزر في الفور الأول ليظهر لنا الفورم  الثاني. 

برنامج المحول الهجري

  الموجودية  في الفصل السابق، ويمكن الرجوع إليها في الأمثلةHejriUtilsهذا البرنامج يستخدم  الوحدة  
المصاحبة للكتاب.

نقوم  بإنشاء برنامج جديد ونقوم  بإديراج الكائنات التالية في الفورم :

2 Tlabel
TCalnder
2 TCompoBox
2 Tbutton
Tmemo
TEdit

ونقوم  بتصميم الفورم  كالشكل التالي:



وكودي الوحدة  المصاحبة للفورم  الرئيسي هو:

unit main;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics, Dialogs,
  StdCtrls, Calendar, HejriUtils;

type

  { TfmMain }

  TfmMain = class(TForm)
    Button1: TButton;
    Button2: TButton;
    Calendar1: TCalendar;



    cbDay: TComboBox;
    cbMonth: TComboBox;
    edYear: TEdit;
    Label1: TLabel;
    Label2: TLabel;
    Memo1: TMemo;
    procedure Button1Click(Sender: TObject);
    procedure Button2Click(Sender: TObject);
    procedure FormCreate(Sender: TObject);
  private
    { private declarations }
  public
    { public declarations }
  end;

var
  fmMain: TfmMain;

implementation

{ TfmMain }

procedure TfmMain.Button1Click(Sender: TObject);
var
  Year, Month, Day: Word;
begin
  DateToHejri(Calendar1.DateTime, Year, Month, Day);
  Memo1.Lines.Add(DateToStr(Calendar1.DateTime) + '  =  ' +
   Format('%d.%s.%d', [Day, HejriMonthsAr[Month], Year]));
  edYear.Text:= IntToStr(Year);
  cbMonth.ItemIndex:= Month ­ 1;
  cbDay.ItemIndex:= Day ­ 1;
end;

procedure TfmMain.Button2Click(Sender: TObject);
var
  ADate: TDateTime;
  CYear, CMonth, CDay: Word;
  YearD, MonthD, DayD: Word;
begin
  ADate:= HejriToDate(StrToInt(edYear.Text), cbMonth.ItemIndex + 1,
    cbDay.ItemIndex + 1);
  Memo1.Lines.Add(FormatDateTime('yyyy­mm­dd ddd', ADate));

  DateToHejri(Now, CYear, CMonth, CDay);
  HejriDifference(StrToInt(edYear.Text), cbMonth.ItemIndex + 1,
   cbDay.ItemIndex + 1, CYear, CMonth,
    CDay, YearD, MonthD, DayD);
  Memo1.Lines.Add('Difference = ' + 
    Format('%d year, %d month, %d day ', [YearD, MonthD, DayD]));
  Calendar1.DateTime:= ADate;

end;



procedure TfmMain.FormCreate(Sender: TObject);
var
  i: Integer;
  Year, Month, Day: Word;
begin
  Calendar1.DateTime:= Now;
  cbMonth.Clear;
  for i:= 1 to 12 do
    cbMonth.Items.Add(HejriMonthsAr[i]);
  DateToHejri(Now, Year, Month, Day);
  cbDay.ItemIndex:= Day ­ 1;
  cbMonth.ItemIndex:= Month ­ 1;
  edYear.Text:= IntToStr(Year);

end;

initialization
  {$I main.lrs}

end.



الرابع  الرابع الفصل الفصل

المنحى   الكائنية المنحى  البرمجة الكائنية البرمجة

Object Oriented ProgrammingObject Oriented Programming



مقدمة

 ترتكز فكرة  البرمجة الكائنية على أنها تقوم  بتقسيم البرنامج ووحداته ومكوناته إلى كائنات. فالكائن
Object:هو عبارة  عن 
التي يمكن أن نعتبرها متغيرات تمثل حالة الكائن. Propertiesمجموعة خصائص .1
يتم تنفيذها فتؤثر على خصائص الكائن. Methods تسمى إجراءات ودوال.2
 تحدث له مثل نقرة  بالماوس أو إظهار له في الشاشة وغريرها. Eventsأحداث .3
 مثل الإجراء الذي يتم تنفيذه عند نقرة  الماوس.Event Handlersإجراءات الحداث .4

ّونوتتكامل هذه الخصائص والإجراءات   كائن. وتقوم  هذه الإجراءات بالتأثير مباشرة  على الخصائص. لتك

بيانات + كود = كائن

 وكمثال لكائن هو مااستخدمناه في الفصل السابق من برمجة الواجهة الرسومية، فهو يعكس الإستخدام 
 اليومي للمبرمج للبرمجة الكائنية المنحى. فنجد أن الفورم  هو عبارة  عن كائن به بعض الخصائص مثل

 ,Close, Show وبه بعض الإجراءات التي تؤثر عليه تأثير مباشر مثل Caption, Width, Heightالـ 
ShowModal, Hide وغريرها. كذلك فإن له أحداث مثل OnClick, OnCreate, OnCloseوله إجراءات  

 مصاحبة لهذه الأحداث، وهي في هذه الحالة الكودي الذي يكتبه المبرمج ليتم تنفيذه إستجابة لحدث
معين.

والوقت     التاريخ برنامج الأول، والوقت    المثال التاريخ برنامج الأول، المثال

 نفرض أننا نريد أن نقوم  بكتابة كودي لكائن يحتوي على تاريخ ووقت، وبعض الإجراءات التي تخص التاريخ
والوقت.

 ،TmyDateTime وفيها كائن يسمى DateTimeUnitقمنا بإنشاء برنامج جديد، ثم وحدة  جديدة  أسميناها 
وهذا هو الكودي الذي قمنا بكتابته في هذه الوحدة :

unit DateTimeUnit;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils; 

type
  { TMyDateTime }

  TMyDateTime = class
  private
    fDateTime: TDateTime;
  public
    function GetDateTime: TDateTime;
    procedure SetDateTime(ADateTime: TDateTime);
    procedure AddDays(Days: Integer);



    procedure AddHours(Hours: Single);
    function GetDateTimeAsString: string;
    function GetTimeAsString: string;
    function GetDateAsString: string;
    constructor Create(ADateTime: TDateTime);
    destructor Destroy; override;
  end;

implementation

{ TMyDateTime }

function TMyDateTime.GetDateTime: TDateTime;
begin
  Result:= fDateTime;
end;

procedure TMyDateTime.SetDateTime(ADateTime: TDateTime);
begin
  fDateTime:= ADateTime;
end;

procedure TMyDateTime.AddDays(Days: Integer);
begin
  fDateTime:= fDateTime + Days;
end;

procedure TMyDateTime.AddHours(Hours: Single);
begin
  fDateTime:= fDateTime + Hours / 24;
end;

function TMyDateTime.GetDateTimeAsString: string;
begin
  Result:= DateTimeToStr(fDateTime);
end;

function TMyDateTime.GetTimeAsString: string;
begin
  Result:= TimeToStr(fDateTime);
end;

function TMyDateTime.GetDateAsString: string;
begin
  Result:= DateToStr(fDateTime);
end;

constructor TMyDateTime.Create(ADateTime: TDateTime);
begin
  fDateTime:= ADateTime;
end;

destructor TMyDateTime.Destroy;
begin
  inherited Destroy;
end;



end.

 أزارار كما في الشكل التالي:5 قمنا بإضافة mainوفي الوحدة  الرئيسية للبرنامج 

وقمنا بكتابة الكودي التالي في الأزارار:

unit main;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics, Dialogs,
  StdCtrls, DateTimeUnit;

type

  { TForm1 }

  TForm1 = class(TForm)
    Button1: TButton;
    Button2: TButton;
    Button3: TButton;
    Button4: TButton;
    Button5: TButton;
    procedure Button1Click(Sender: TObject);
    procedure Button2Click(Sender: TObject);
    procedure Button3Click(Sender: TObject);
    procedure Button4Click(Sender: TObject);
    procedure Button5Click(Sender: TObject);



    procedure FormCreate(Sender: TObject);
  private
    { private declarations }
  public
    MyDT: TMyDateTime;
    { public declarations }
  end; 

var
  Form1: TForm1; 

implementation

{ TForm1 }

procedure TForm1.FormCreate(Sender: TObject);
begin
  MyDT:= TMyDateTime.Create(Now);
end;

procedure TForm1.Button1Click(Sender: TObject);
begin
  ShowMessage(MyDT.GetDateTimeAsString);
end;

procedure TForm1.Button2Click(Sender: TObject);
begin
  ShowMessage(MyDT.GetDateAsString);
end;

procedure TForm1.Button3Click(Sender: TObject);
begin
  ShowMessage(MyDT.GetTimeAsString);
end;

procedure TForm1.Button4Click(Sender: TObject);
begin
  MyDT.AddHours(1);
end;

procedure TForm1.Button5Click(Sender: TObject);
begin
  MyDT.AddDays(1);
end;

initialization
  {$I main.lrs}

end.



نلحظ في هذا البرنامج النقاط التالية:

DateTimeUnitفي الوحدة  

  وهذه هي طريقة تعريف كائن والذي يمكنClass على أنه من النوع TmyDateTimeقمنا بتعريف .1
أن يحتوي على إجراءات، ديوال ومتغيرات كما في هذا المثال

  ، وهو إجراء من نوع خاصConstructor نوع هذا الإجراء هو: Createقمنا بتعريف إجراء اسمه .2
يستخدم  لإنشاء وتفعيل وتهيئة متغيرات كائن ما في الذاكرة  حتى يمكن إستخدامه. 

  ، وهو إجراء من نوع خاص تتم منادياته عندDestructor نوعه Destroyقمنا بتعريف إجراء اسمه .3
  سوف نتطرقoverrideالإنتهاء من العمل بهذا الكائن وإرادية  حذفه من الذاكرة . وفي نهايته كلمة 

لها في كتاب قاديم  إن شاء الله.
  , ومتغيراته وديواله وإجراءاته ليمكن الوصول لهاprivateنلحظ أن هناك  جزئين، الجزء الأول هو .4

من وحدة  أخرى عند استخدام  هذا الكائن.
  وهو ذو متغيرات، ديوال وإجراءات يمكن الوصول لها من خارج وحدة  هذاpublicالجزء الثاني هو .5

 بالنسبة للوحدات.Interfacesالكائن، وهي تمثل الواجهة المرئية لهذا الكائن مثل 

 فنجد التي:mainأما في الوحدة  الرئيسية للبرنامج 

Uses مع باقي الوحدات في الجزء DateTimeUnitقمنا بإضافة إسم الوحدة  .1

 دياخل كائن الفورم :MyDTقمنا بتعريف الكائن .2
  MyDT: TMyDateTime;

:OnFormCreateقمنا بإنشاء وتهيئة الكائن عند الحدث .3
  MyDT:= TMyDateTime.Create(Now);

وهذه هي طريقة إنشاء وتهيئة الكائن في لغة أوبجكت باسكال.

Encapsulationالكبسلة 

 من فوائد البرمجة الكائنية الكبسلة، وهي أن نقوم  بجمع البيانات والإجراءات وحمايتهما من العالم
ُيسمح  بالوصول إلى مابداخلها إل ُتمثل قطعة واحدة ، ول   الخارجي لتصبح  في شكل كبسولة محمية و

 بواسطة بوابات معينة. وهذه البوابات تتمثل في الإجراءات المتاحة، حيث يمكن إضافة إجراء متاح
  ل يستطيع المستخدم  (المبرمج)protected أو محمية private وإجراءات مخفية  publicللمستخدم  

ُيمكن أن تكون مصفوفة رقمية أو ًل هذه البيانات   الوصول إليها. كذلك فإن البيانات تكون محمية. مث
ُنريد للمستخدم  أن يتسبب في تلف  مقطعية بها بيانات يتم إستخدامها بواسطة إجراءات الكائن، فل 

ًل عن ذلك نقوم  بتوفير إجراءات مثل إضافة  البيانات أو حذفها بالوصول إلى هذه المصفوفة مباشرة . بد
ًا. ُنريدها ول تؤثر على البيانات سلب لتلك المصفوفة، أو حذف عنصر بالطريقة التي 

 يمكن تشبيه عملية الكبسلة بجهازا موبايل يحتوي على بطارية وديوائر إلكترونية وشريحة، فيقوم  الصانع
ًل عن ًل أن يصل إلى أسلك  البطارية لمحاولة شحنها أو تفريغها، بد ُيسمح  للمستخدم  مث  بتغيلفها، فل 
 ذلك يقوم  الصانع بتوفير منفذ لشحن الموبايل، وإذا تم الشحن يقوم  هذا المنفذ أو الدوائر التي خلفه

 بإيقاف عملية الشحن حتى ل تتلف البطارية. كذلك فإن الموبايل ل يسمح  بتفريغ البطارية، إل عن
 طريق استهلكها في المكالمات أو الإستخدام  العاديي. كذلك فإن الدوائر الإلكترونية مخفية ول تظهر، ول

 ُيمكن أن يقوم  المستخدم  بالوصول لها مباشرة  ومحاولة توصيل أسلك  أو كهرباء مباشرة  لها. بهذه
 الطريقة تضمن لنا الكبسلة حماية المحتويات الداخلة للموبايل أو البيانات بالنسبة للكائن وتضمن لنا

ُيريده الصانع لهذا الكائن أو الجهازا الإلكتروني. الإستخدام  الأمثل الذي 



كائنية    بطريقة الأخبار كائنية   برنامج بطريقة الأخبار برنامج

 في المثال التالي قمنا بإعادية  كتابة برنامج الأخبار بطريقة كائنية. وفي هذه المرة  قمنا بتصنيف الأخبار،
فكل صنف يتم تسجيله في ملف بيانات منفصل من نوع ملفات الوصول المباشر.

 ، وقمنا بإضافة وحدة  جديدة  كتبنا فيها كوديoonewsقمنا بإنشاء برنامج جديد ذو واجهة رسومية وأسميناه 
 كالتي:TNewsالكائن 

unit news;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils; 

type
  TNewsRec = record
    ATime: TDateTime;
    Title: string[100];
  end;

  { TNews }

  TNews = class
    private
      F: file of TNewsRec;
      fFileName: string;
    public
      constructor Create(FileName: string);
      destructor Destroy; override;
      procedure Add(ATitle: string);
      procedure ReadAll(var NewsList: TStringList);
      function Find(Keyword: string; 
        var ResultList: TStringList): Boolean;
  end;

implementation

{ TNews }

constructor TNews.Create(FileName: string);
begin
  fFileName:= FileName;
end;

destructor TNews.Destroy;
begin
  inherited Destroy;
end;

procedure TNews.Add(ATitle: string);
var
  Rec: TNewsRec;



begin
  AssignFile(F, fFileName);
  if FileExists(fFileName) then
  begin
    FileMode:= 2; // Read/write access
    Reset(F);
    Seek(F, FileSize(F));
  end

  else
    Rewrite(F);

  Rec.ATime:= Now;
  Rec.Title:= ATitle;
  Write(F, Rec);
  CloseFile(F);

end;

procedure TNews.ReadAll(var NewsList: TStringList);
var
  Rec: TNewsRec;
begin
  NewsList.Clear;
  AssignFile(F, fFileName);
  if FileExists(fFileName) then
  begin
    Reset(F);
    while not Eof(F) do
    begin
      Read(F, Rec);
      NewsList.Add(DateTimeToStr(Rec.ATime) + ' : ' + Rec.Title);
    end;
    CloseFile(F);
  end;

end;

function TNews.Find(Keyword: string; var ResultList: TStringList): Boolean;
var
  Rec: TNewsRec;
begin
  ResultList.Clear;
  Result:= False;
  AssignFile(F, fFileName);
  if FileExists(fFileName) then
  begin
    Reset(F);
    while not Eof(F) do
    begin
      Read(F, Rec);
      if Pos(LowerCase(Keyword), LowerCase(Rec.Title)) > 0 then
      begin
        ResultList.Add(DateTimeToStr(Rec.ATime) + ' : ' + Rec.Title);
        Result:= True;
      end;
    end;



    CloseFile(F);
  end;

end;

end.

وفي البرنامج الرئيسي قمنا بإضافة هذه المكونات:
Edit Box, ComboBox, 3 Buttons, Memo, 2 labels

فأصبح  شكل الفورم  كالتي:

وقمنا بكتابة الكودي التالي للوحدة  المصاحبة للفورم  الرئيسي:

unit main;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics,
    Dialogs, News, StdCtrls;

type



  { TForm1 }

  TForm1 = class(TForm)
    btAdd: TButton;
    btShowAll: TButton;
    btSearch: TButton;
    cbType: TComboBox;
    edTitle: TEdit;
    Label1: TLabel;
    Label2: TLabel;
    Memo1: TMemo;
    procedure btAddClick(Sender: TObject);
    procedure btSearchClick(Sender: TObject);
    procedure btShowAllClick(Sender: TObject);
    procedure FormCreate(Sender: TObject);
  private
    { private declarations }
  public
    NewsObj: array of TNews;
    { public declarations }
  end; 

var
  Form1: TForm1; 

implementation

{ TForm1 }

procedure TForm1.FormCreate(Sender: TObject);
var
  i: Integer;
begin
  SetLength(NewsObj, cbType.Items.Count);
  for i:= 0 to High(NewsObj) do
    NewsObj[i]:= TNews.Create(cbType.Items[i] + '.news');

end;

procedure TForm1.btAddClick(Sender: TObject);
begin
  NewsObj[cbType.ItemIndex].Add(edTitle.Text);
end;

procedure TForm1.btSearchClick(Sender: TObject);
var
  SearchStr: string;
  ResultList: TStringList;
begin
  ResultList:= TStringList.Create;
  if InputQuery('Search News', 'Please input keyword', SearchStr) then
    if NewsObj[cbType.ItemIndex].Find(SearchStr, ResultList) then
    begin
      Memo1.Lines.Clear;
      Memo1.Lines.Add(cbType.Text + ' News');
      Memo1.Lines.Add('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');



      Memo1.Lines.Add(ResultList.Text);
    end
    else
      Memo1.Lines.Text:= SearchStr + ' not found in ' + 
        cbType.Text + ' news';
  ResultList.Free;
end;

procedure TForm1.btShowAllClick(Sender: TObject);
var
  List: TStringList;
begin
  List:= TStringList.Create;
  NewsObj[cbType.ItemIndex].ReadAll(List);
  Memo1.Lines.Clear;
  Memo1.Lines.Add(cbType.Text + ' News');
  Memo1.Lines.Add('­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­­');
  Memo1.Lines.Add(List.Text);
  List.Free;
end;

procedure TForm1.FormClose(Sender: TObject; var CloseAction: TCloseAction);
var
  i: Integer;
begin
  for i:= 0 to High(NewsObj) do
    NewsObj[i].Free;

  NewsObj:= nil;

end;

initialization
  {$I main.lrs}

end.

نلحظ في البرنامج السابق التي:

 Dynamic بالمصفوفة المرنةأننا استخدمنا نوع جديد من تعريف المصفوفات وهو مايعرف .1
Arrayًا  ، وهي مصفوفة غرير محددية  الطول في وقت كتابة الكودي، لكن طولها يتغير زايادية  أو نقصان

أثناء تشغيل البرنامج حسب الحاجة، وطريقة التعريف هي كالتي:

  NewsObj: array of TNews;

 وفي أثناء تنفيذ البرنامج وقبل إستخدامها يجب أن نقوم  بحجز مساحة لها في الذاكرة  بإستخدام  الإجراء
SetLength:ًل إذا قمنا بكتابة الكودي التالي ، مث

  SetLength(NewsObj, 10);

فهذا يعني أننا قمنا بحجز عشر خانات ، فهي تماثل في هذه الحالة هذا التعريف:
  NewsObj: array [0 .. 9] of Tnews;

:Combo Boxوفي حالة برنامج الأخبار قمنا بحجز خانات تمثل عددي أنواع الأخبار الموجودية  في الـ 



  SetLength(NewsObj, cbType.Items.Count);

وبهذه الطريقة تجعل عددي الكائنات معتمدة  على عددي أنواع الأخبار المكتوبة في الكائن 
ComboBox.Items.ًا بدون الحاجة لإعادية  تغيير الطول  فكلما قام  المبرمج بزياديتها، كلما زااديت تلقائي

  وهي تمثل النوع، وليمكن استخدامه مباشرة  إل بتعريفClass هو عبارة  عن TNewsنوع الكائن .2
 Integer. بنفس الطريقة التي نعتبر فيها أن النوع الصحيح  Objectsمتغيرات منه تسمى كائنات 

.I, j, xليمكن إستخدامه مباشرة  بل يجب استخدام  متغيرات من نوعه، مثل 

ًل ثم قمنا بتحرير وحذف.3  في نهاية البرنامج قمنا بتحرير الكائنات في المصفوفة من الذاكرة  أو
ًا بواسطة الكودي التالي : المصفوفة المرنة ثاني

  for i:= 0 to High(NewsObj) do
    NewsObj[i].Free;

  NewsObj:= nil;



الصفوف  الصفوف برنامج برنامج
 ، وهي تستخدم  كحلول لمعاجة الصف أوData structureالصفوف هي إحدى ديروس هيكلية البيانات 

ًل إذا لم تكن هناك  أولوية. في هذه الحال تكون ًل يخرج أو  الطابور. ويتميز الصف بأن من يدخله أو
الأولوية هي زامن الدخول في الصف.
  يمكن استخدامه لإضافةTQueue تحتوي على نوع الكائن Queueفي البرنامج التالي قمنا بكتابة وحدة  

ًل أسماء) إلى صف ثم إستخراج هذه البيانات من الصف. والصف يختلف عن المصفوفة في  بيانات (مث
ًل إذا كان هناك  صف يحتوي على   10أن القراءة  (الإستخراج) منه تؤثر على البيانات وطول الصف، فمث

 عناصر، ثم قمنا بقراءة  والحصول على ثلث عناصر منها فإن عددي من في الصف ينقص إلى سبعة. وإذا
ًا. ًا  أصبح  الصفُ فارغر قمنا بقراءة  العناصر جميع

كودي وحدة  الصف:
unit queue;
// This unit contains TQueue class, 
// which is suitable for any string queue

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils; 

type

  { TQueue }

  TQueue = class
    private
      fArray: array of string;
      fTop: Integer;
    public
      constructor Create;
      destructor Destroy; override;
      function Put(AValue: string): Integer;
      function Get(var AValue: string): Boolean;
      function Count: Integer;
      function ReOrganize: Boolean;
  end;

implementation

{ TQueue }

constructor TQueue.create;
begin
  fTop:= 0;
end;

destructor TQueue.destroy;
begin
  SetLength(fArray, 0); // Erase queue array from memory
  inherited destroy;
end;



function TQueue.Put(AValue: string): Integer;
begin
  if fTop >= 100 then
    ReOrganize;

  SetLength(fArray, Length(fArray) + 1);
  fArray[High(fArray)]:= AValue;
  Result:= High(fArray) ­ fTop;
end;

function TQueue.Get(var AValue: string): Boolean;
begin
  AValue:= '';
  if fTop <= High(fArray) then
  begin
    AValue:= fArray[fTop];
    Inc(fTop);
    Result:= True;
  end
  else // empty
  begin
    Result:= False;

    // Erase array
    SetLength(fArray, 0);
    fTop:= 0;
  end;
end;

function TQueue.Count: Integer;
begin
  Result:= Length(fArray) ­ fTop;
end;

function TQueue.ReOrganize: Boolean;
var
  i: Integer;
  PCount: Integer;
begin
  if fTop > 0 then
  begin
    PCount:= Count;
    for i:= fTop to High(fArray) do
      fArray[i ­ fTop]:= fArray[i];

    // Truncate unused data
    setLength(fArray, PCount);
    fTop:= 0;
    Result:= True; // Re Organize is done

  end
  else
    Result:= False; // nothing done
  end;
end;
end.



الفورم  الرئيسي لبرنامج الصف:

كودي الوحدة  الرئيسية للفورم :

unit main;

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics,
 Dialogs, Queue, StdCtrls;

type

  { TfmMain }

  TfmMain = class(TForm)
    bbAdd: TButton;
    bbCount: TButton;
    bbGet: TButton;
    edCustomer: TEdit;
    Label1: TLabel;
    Memo1: TMemo;
    procedure bbAddClick(Sender: TObject);
    procedure bbCountClick(Sender: TObject);
    procedure bbGetClick(Sender: TObject);
    procedure FormClose(Sender: TObject; var CloseAction: TCloseAction);
    procedure FormCreate(Sender: TObject);
  private



    { private declarations }
  public
    MyQueue: TQueue;
    { public declarations }
  end; 

var
  fmMain: TfmMain;

implementation

{ TfmMain }

procedure TfmMain.FormCreate(Sender: TObject);
begin
  MyQueue:= TQueue.Create;
end;

procedure TfmMain.FormClose(Sender: TObject; var CloseAction: TCloseAction);
begin
  MyQueue.Free;
end;

procedure TfmMain.bbCountClick(Sender: TObject);
begin
  Memo1.Lines.Add('Queue length is: ' + IntToStr(MyQueue.Count));
end;

procedure TfmMain.bbAddClick(Sender: TObject);
var
  APosition: Integer;
begin
  APosition:= MyQueue.Put(edCustomer.Text);
  Memo1.Lines.Add(edCustomer.Text + ' has been added as # ' +
    IntToStr(APosition + 1));
end;

procedure TfmMain.bbGetClick(Sender: TObject);
var
  ACustomer: string;
begin
  if MyQueue.Get(ACustomer) then
  begin
    Memo1.Lines.Add('Got: ' + ACustomer + ' from the queue');
  end
  else
    Memo1.Lines.Add('Queue is empty');
end;

initialization
  {$I main.lrs}

end.



  يتم زايادية  سعة المصفوفة المرنة لإستيعاب العنصر الجديد. وعندPutفي البرنامج السابق في الإجراء 
  الذي يؤشر على أول العانصر بالصف.fTop يتم تحريك المؤشر Getإستخراج عنصر عن طريق الدالة 

 وبعد هذا الإجراء ليمكن حذف المصفوفة من البداية، حيث أن المصفوفة المرنة فقط يمكن تقليصها أو
ًا في  توسيعها من النهاية، لذلك اكتفينا فقط بتحريك المؤشر مع بقاء العناصر القديمة تحتل مكان

  لتفريغ الصف من العناصر الغيرReOrganizeالمصفوفة. ولمعالجة هذه المشكلة قمنا بعمل الإجراء 
مستخدمة والتي تم إستخراجها من قبل حتى لنحتل ذاكرة  بدون فائدة .

  ببساطة يعمل كلما يصل عددي المستخرجين من الصف حوالي مائة، حيث يتم نداءReOrganizeالإجراء 
ًل حتى ليتسبب هذا الإجراء في إبطاءPutهذا الإجراء من الدالة   . ويمكن زايادية  هذا الرقم إلى ألف مث

الإضافة، فقط يعمل كلما يصل طول الصف الفعلي في الذاكرة  إلى ألف.
ًا،  في هذا الإجراء نقوم  بنقل العناصر الموجودية  في الصف إلى مكان العناصر القديمة المستخرجة مسبق

بهذه الطريقة:

    for i:= fTop to High(fArray) do
      fArray[i ­ fTop]:= fArray[i];

 ثم نقوم  بعد ذلك بقص المصفوفة عند آخر عنصر دياخل الصف ونضع مؤشر الصف في بداية المصفوفة
كالتالي:

    // Truncate unused data
    setLength(fArray, PCount);
    fTop:= 0;

  بواسطةInformation hiding f من البرنامج السابق نجد أن البرمجة الكائنية وفرت لنا حماية البيانات
ًا. حيث أن البيانات الحساسة التي ربما تسبب في سلوك  غرير معلوم  إذا  الكبسلة كما تكلنا عنها سابق

 أتحنا للمستخدم  الوصول المباشر لها. لذلك قمنا بإخفاء هذه المتغيرات المهمة والتي ليحتاج وليفترض
أن يقوم  المستخدم  بالوصول لها مباشرة  وهي:

    private
      fArray: array of string;
      fTop: Integer;

 حيث أن المبرمج الذي يستخدم  هذا الكائن ليتسطيع الوصول لها من خلل برنامجه الرئيسي ولو وصل
ًل نفرض أنه قام  بتغيير قيمة    في حين1000 إلى fTopلها لتسبب في ضياع البيانات أو حدوث أخطاء. مث

 عناصر في الصف، فهذا يتسبب بخطأ أثناء التشغيل.10وجودي فقط 
 .Put, Getوكبديل سمحنا له بالوصول لإجراءات وديوال آمنة طبيعية تتناسب وطبيعة تحقيق الهدف، مثل 

 فمهما إستخدمها المبرمج لنتوقع منها حدوث خطأ أو شيء غرير منطقي. وهذه الطريقة اشبه بإستخدام 
 بوابات معروفة للوصول للبيانات. وفي هذه الحالة البوابات هي الإجراءات والدوال الموجودية  في هذا

الكائن، حيث ليمكن إستخدام  إجراءات أخرى غرير الموجودية  في هذا الكائن للوصول للبيانات. 

 ومما سبق نجد أن الكائن بالنسبة للمُستخدم  (المبرمج الذي يستخدمه في برامجه) هو عبارة  عن
إجراءات وديوال تقبع خلفها بيانات تخص هذا الكائن.



Object Oriented Fileالملف الكائني 

 رأينا في الفصل الأول كيفية التعامل مع الملفات بأنواعها المختلفة. وقد كانت طريقة التعامل معتمدة 
 على نوع البرمجة الهيكلية، أي أنها تتعامل مع ديوال وإجراءات ومتغيرات فقط، أما هذه المرة  نريد

 .TFileStreamالوصول للملفات عن طريق الكائن الموجودي في مكتبات فري باسكال ولزااراس وهو 
 ،Untyped File يتعامل مع الملفات بطريقة أشبه بنوع الملف غرير محددي النوع TFileStreamوالكائن  

والذي بدوره يصلح  لكافة أنواع الملفات.
ّية تمتازا بكل ماتمتازا  تتميز طريقة إستخدام  الملف الكائني بأنها تحتوي على إجراءات وديوال وخصائص غرن

 به البرمجة الكائنية من سهولة الإستخدام  والمنطقية في التعامل والقياسية وتوقع المبرمج لطريقة
الإستخدام . 

بواسطة     الملفات نسخ بواسطة    برنامج الملفات نسخ TFileStreamTFileStreamبرنامج

 في هذا المثال نريد نسخ ملف بإستخدام  هذا النوع للوصول للملفات، فنقوم  بإنشاء برنامج جديد ذو
واجهة رسومية ونقوم  بإديراج هذه المكونات:

TButton, TOpenDialog, TSaveDialog
 بالنسبة للزر:OnClickثم نقوم  بكتابة الكودي التالي في الحدث 

procedure TfmMain.Button1Click(Sender: TObject);
var
  SourceF, DestF: TFileStream;
  Buf: array [0 .. 1023] of Byte;
  NumRead: Integer;
begin
  if OpenDialog1.Execute and SaveDialog1.Execute then
  begin
    SourceF:= TFileStream.Create(OpenDialog1.FileName, fmOpenRead);
    DestF:= TFileStream.Create(SaveDialog1.FileName, fmCreate);
    while SourceF.Position < SourceF.Size do
    begin
      NumRead:= SourceF.Read(Buf, SizeOf(Buf));
      DestF.Write(Buf, NumRead);
    end;
    SourceF.Free;
    DestF.Free;
    ShowMessage('Copy finished');
  end;
end;

 وهي طريقة مشابهة لطريقة نسخ الملفات بإستخدام  الملفات غرير محددية  النوع التي استخدمناها في
الفصل الأول. 

ًا نسخ الملف بطريقة مبسطة وهي كالتي: ويمكن أيض

procedure TfmMain.Button1Click(Sender: TObject);
var
  SourceF, DestF: TFileStream;
begin
  if OpenDialog1.Execute and SaveDialog1.Execute then
  begin



    SourceF:= TFileStream.Create(OpenDialog1.FileName, fmOpenRead);
    DestF:= TFileStream.Create(SaveDialog1.FileName, fmCreate);
    DestF.CopyFrom(SourceF, SourceF.Size);
    SourceF.Free;
    DestF.Free;
    ShowMessage('Copy finished');
  end;
end;

ًل لأننا حددينا حجم الجزء المراديCopyFromحيث أن الإجراء    يقوم  بعملية النسخ لمحتويات الملف كام
 الذي يمثل الحجم الكامل للملف بالبايت.SourceF.Sizeنسخه وهو 

Inheritanceالوراثة 

ًا. وهو يعني أن الكائن الجديد  الوراثة في البرمجة الكائنية تعني إنشاء كائن جديد من كائن موجودي مسبق
يرث صفات الكائن القديم ويمكن أن يزيد عليه بعض الخصائص والأجراءات والدوال.

ًل من كتابته من الصفر يمكن الإعتمادي على  كمثال للوراثة نريد عمل كائن لصف من النوع الصحيح ، فبد
ًا والذي يستخدم  المقاطع. ولوراثة كائن ما نقوم  بإنشاء وحدة  جديدة   كائن الصف الذي قمنا بكتابته سابق

ونقوم  بإستخدام  الوحدة  القديمة المحتوية على الكائن القديم. ونقوم  بتعريف الكائن الجديد كالتي:

  TIntQueue = class(TQueue)

. PutInt, GetInt. وقمنا بإضافة فقط ديالتين في الكائن الجديد IntQueueوقد أسمينا الوحدة  الجديدة  
ًل هو: وكودي الوحدة  الجديدة  كام

unit IntQueue;

// This unit contains TIntQueue class, which is inherits TQueue 
// class and adds PutInt, GetInt methods to be used with 
// Integer queue

{$mode objfpc}{$H+}

interface

uses
  Classes, SysUtils, Queue;

type

  { TIntQueue }

  TIntQueue = class(TQueue)

  public
    function PutInt(AValue: Integer): Integer;
    function GetInt(var AValue: Integer): Boolean;



  end;

implementation

{ TIntQueue }

function TIntQueue.PutInt(AValue: Integer): Integer;
begin
  Result:= Put(IntToStr(AValue));
end;

function TIntQueue.GetInt(var AValue: Integer): Boolean;
var
  StrValue: string;
begin
  Result:= Get(StrValue);
  if Result then
    AValue:= StrToInt(StrValue);

end;

end.

.TQueue لأنها موروثة من الكائن Count, Create, Destroyنلحظ أننا لم نقم بتكرار بعض العمليات مثل 

لإستخدام  الوحدة  الجديدة  قمنا بإنشاء الفورم  التالي:

وكودي الفورم  السابق هو:

unit main;

{$mode objfpc}{$H+}



interface

uses
  Classes, SysUtils, FileUtil, LResources, Forms, Controls, Graphics,
    Dialogs, IntQueue, StdCtrls;

type

  { TfmMain }

  TfmMain = class(TForm)
    bbAdd: TButton;
    bbCount: TButton;
    bbGet: TButton;
    edCustomerID: TEdit;
    Label1: TLabel;
    Memo1: TMemo;
    procedure bbAddClick(Sender: TObject);
    procedure bbCountClick(Sender: TObject);
    procedure bbGetClick(Sender: TObject);
    procedure FormClose(Sender: TObject; 
       var CloseAction: TCloseAction);
    procedure FormCreate(Sender: TObject);
  private
    { private declarations }
  public
    MyQueue: TIntQueue;
    { public declarations }
  end; 

var
  fmMain: TfmMain;

implementation

{ TfmMain }

procedure TfmMain.FormCreate(Sender: TObject);
begin
  MyQueue:= TIntQueue.Create;
end;

procedure TfmMain.FormClose(Sender: TObject; var CloseAction: TCloseAction);
begin
  MyQueue.Free;
end;

procedure TfmMain.bbCountClick(Sender: TObject);
begin
  Memo1.Lines.Add('Queue length is: ' + IntToStr(MyQueue.Count));
end;

procedure TfmMain.bbAddClick(Sender: TObject);
var
  APosition: Integer;



begin
  APosition:= MyQueue.PutInt(StrToInt(edCustomerID.Text));
  Memo1.Lines.Add(edCustomerID.Text + ' has been added as # ' 
   + IntToStr(APosition + 1));
end;

procedure TfmMain.bbGetClick(Sender: TObject);
var
  ACustomerID: Integer;
begin
  if MyQueue.GetInt(ACustomerID) then
  begin
    Memo1.Lines.Add('Got: Customer ID : ' + IntToStr(ACustomerID) + 
    ' from the queue');
  end
  else
    Memo1.Lines.Add('Queue is empty');
end;

initialization
  {$I main.lrs}

end.

  والدوال الجديدة  الموجودية  فقطTQueueنلحظ أننا قمنا بنداء الإجراءات والدوال الموجودية  في الكائن 
.TIntQueueفي 

ُنسمي الكائن   . ونسمي الكائنancestor أو السلف base class الكائن الأساس TQueueفي هذه الحالة 
 أو الوريث.descender المُنحدر TIntQueueالجديد 

ملحوظة:
 كان من الممكن عدم  استخدام  الوراثة في المثال السابق وإضافة الدوال الجديدة  مباشرة  في الكائن

 ، لكن لجأنا لذلك لشرح الوراثة، وهناك  سبب آخر، هو أننا ربما لنمتلك الكودي المصدرTQueueالأساسي 
، ففي هذه الحالة يتعذر تعديلها، ويكون السبيل الوحيد هي الوراثة منها ثم تعديلها.Queueللوحدة  

أو كما في فري باسكال ppuويمكن للمبرمج أن يقوم  بتوزايع الوحدات في شكل ملفات ذات الإمتدادي 
dcuكما في ديلفي، في هذه الحال ليمكن الإطلع على الكودي المصدري، حيث يمكن الإطلع على الكودي  

 .pasفقط في حالة الحصول على الملف .

ُينال بهذا الكتاب فائدة  المسلمين. وفي الختام  نتمنى أن 

 الخطوة  الثانية مع أوبجكت باسكال –و بعد إنتهاء الدارس من هذا الكتاب، يمكنه قراءة  الكتاب الخر "
"صناعة البرمجيات

